
  السادسالفصل  6

 

  

  
  الفئات الأساسیة

  
من  تقوم بالعديد على مئات الفئات التي NET Framework.حتوي مكتبة فئات إطار عمل ت

ريعك من مراحل بناء برامجك ومشايتجزأ ، واستخدامها يعتبر جزء لا  والمتنوعةمختلفةال الوظائف
 إلى وقت طويل جدا من عمرك يحتاجالفئات هذه جميع ولكن الإلمام ب. Visual Basic .NETبـ 

 حتى لحظة كتابة الصفحة الأخيرة من هذا الكتاب مازلت أني أخبرتك إن سرا أذيعالبرمجي، ولا 
  . الفئات وأعضائهاهاكتشف المزيد والمزيد من هذ

 ،NET Framework. لإطـار عمـل   لأساسـية اإلى الفئـات  مدخلاً هذا الفصل يقدم لك 
باقي الفئات الأخرى بالتدريج في الفصول اللاحقة حتى نهاية هذا          معظم  وسنستمر في التوغل في     

منـي  ، فلا تتوقـع  MSDN هنا بان هذا الكتاب ليس مرجع من مراجع    أذكرك أنالكتاب، وعلي   
لذلك، عليك البحث عـن  .  وأعضائها بنفسيNET Framework. بشرح جميع مكتبة فئات القيام

  NET Documentation.كافة تفاصيل الفئات وأعضائها في مستندات 

  
  System.Objectالفئة 

– وجميع التركيبات والأنواع الأخرى مشتقة وراثيا NET Framework.جميع فئات إطار عمل 
لمعرفـة  ، مما يعني ان جميع الأعضاء اSystem.Object من الفئة -بشكل مباشر او غير مباشر 

المزيد أيـضا،   .  ستكون موجودة أيضا في أنواع البيانات التي تنشئها        System.Objectفي الفئة   
 لـذلك  Base Classفهي الفئة القاعدية  (Objectيمكنك إسناد أي قيمة إلى المؤشرات من النوع 

لفصل  إلى فئة قاعدية كما وضحت لك في اDerived Classتستطيع إسناد أي قيمة من فئة مشتقة 
  ): الوراثةالرابع

 
   يمكنك استخدام الصیغة المختصرة '
 Dim X As Object أیضا '
Dim X As System.Object 
 
X = New AnyClass 

 NET Framework  232.إطار عمل : الجزء الثاني
 ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  
 هي الواجهات NET. في عالم System.Object الوحيد الذي لا يشتق وراثيا من الفئة ءالشي

Interfaces تفويض، والمواصفاتالواجهات، ال والتي ناقشناها في الفصل السابق.  
  

  Objectطرق الفئة 
جميع الطرق التابعة لهذه الفئة ب الإلمام، فمن الجيد Objectبما ان جميع البيانات مشتقة من الفئة 

  .تكون موجودة مع جميع البيانات الأخرىسحيث انها 
  

  :()Equalsالطريقة 
 ان كان المؤشر المرسل Trueقيمة بال تعود و Overridable القيادة لإعادةهذه الطريقة قابلة 

  :يشير إلى نفس الكائن الحالي
 
Dim x As New TestClass() 
Dim y As Object = x 
 
ArabicConsole.WriteLine(y.Equals(x)) ' True 

  
 NET Framework.معظم الفئات والأنواع الأخرى من البيانات في إطار عمل فإن مع ذلك، 

 فمثلا، المتغيرات .)وهذا ما ستقوم به انت في العادة( الطريقة  هذهOverridesتقوم باعادة قيادة 
 قيادة هذه الطريقة بحيث تعود بالقيمة بإعادة تقوم Value Type Variables القيمة ذومن النوع 

Trueان كانت قيمة المتغير المرسل تساوي قيمة المتغير الحالي :  
 
Dim X As Integer = 10 
 
ArabicConsole.WriteLine(X.Equals(10)) ' True 

  
  :()GetTypeالطريقة 

  . تمثل نوع الكائن الحاليSystem.Type من النوع تعود هذه الطريقة بقيمة
  

    انظر أيضا
  

 لاحقا في الفـصل الثـاني   System.Typeسأتحدث عن الفئات من النوع     
  .Reflection Classesفئات الانعكاس عشر 

  :()ToStringالطريقة 
تمثل ) Stringمن النوع ( وهي تعود بقيمة حرفية Overridable القيادة لإعادةقابلة هذه الطريقة 

  :)Namespacesبما في ذلك مجالات الأسماء  ( منها الكائن الحاليأنشئالاسم الكامل للفئة التي 
 
Dim obj As New TestClass() 
 
ArabicConsole.WriteLine(obj.ToString) ' MyNamespace.TestClass 

  
 قيادة هذه الطريقة بحيث تعود بإعادة تقوم NET Framework.معظم فئات إطار عمل فمع ذلك، 

  :بالقيمة التي يحتويها الكائن
 
Dim X As Integer = 10 
Dim Y As String = "xxx" 
 
ArabicConsole.WriteLine(X.ToString) ' 10 
ArabicConsole.WriteLine(Y.ToString) ' xxx 

  
  :()ReferenceEqualsالطريقة 

شيران إلى نفس ي ن المرسلاكان المؤشران ان True بالقيمة ()ReferenceEqualsتعود الطريقة 
  :Is عمل المعامل ()ReferenceEquals تعمل الطريقة وبعبارة أخرى،الكائن، 

 
Dim Turki As New Person() 
Dim Khaled As Person = Turki 
 
ArabicConsole.WriteLine(Turki Is Khaled) ' True 
ArabicConsole.WriteLine(Person.ReferenceEquals(Turki, Khaled)) ' True 

  
، الا انها تختلف المذكورة سابقاً ()Equals مشابهة للطريقة ()ReferenceEqualsالطريقة 
 Shared Method طريقة مشتركة ()ReferenceEquals ان الطريقة الأولى: عنها في نقطتين

   .Not Overridableطة الثانية انها غير قابلة لإعادة القيادة و النق
  

  :()MemberwiseCloneالطريقة 
 وانشاء نسخة جديدة طبق الاصل منه، وقد   الكائن لنسخ   ()MemberwiseCloneتستخدم الطريقة   

 شرحبالتحديد عند و الواجهات، التفويض، والمواصفاتذكرت مثالا يستخدمها في الفصل الخامس    
ــة الو ــة    . ICloneableاجه ــول للطريق ــدد الوص ــا ان مح ــر هن ــم ان اذك ــن المه م

 ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  
MemberwiseClone()   هو Protected  ،  ،فلن تتمكن من الوصول لها إلا من داخل الفئة     وعليه

  ).System.Objectفهي مشتقة تلقائيا من الفئة (نفسها 
  

    انظر أيضا
  

ت الوصـول    ومحـددا  Protectedلمزيد من التفاصیل حول محـدد الوصـول         
  .الوراثةالأخرى، راجع الفصل الرابع 

  
  

  :()Finalizeالطريقة 
 أيضا Protected وهي طريقة محمية Overridable قابلة لإعادة القيادة ()Finalizeالطريقة 

حالها كحال الطريقة السابقة –ولن تتمكن من استدعائها إلا من داخل الفئة 
MemberwiseClone() .الفئات والكائناتقة سابقا في الفصل الثالث تحدثت عن هذه الطري ،
  .Destructors المهدماتوذكرت انها تعمل عمل 

  
  البیانات المرجعیة والبیانات ذات القیمة مرة أخرى

 تنقسم إلى قسمين هما NET Framework. سابقا على ان جميع البيانات في إطار عمل اتفقنا
، واتفقنا أيضا على ان Value Type قيمة وبيانات ذات Reference Type بيانات مرجعية

، وذكرت Managed Heapالبيانات المرجعية يتم حفظها في قسم خاص لها من الذاكرة يسمى 
  .الفئات والكائناتفي الفصل الرابع سابقا كل التفاصيل المتعلقة بالبيانات المرجعية 

، Booleans، القـيم المنطقيـة      )سواء كانت صـحيحة او عـشرية       (Numbersالاعداد  
، Value Type القيمة ذو جميعها بيانات من النوع Enums او Structuresالتركيبات من النوع 

 -في هذا الفصل في التحديد    – Booleanوعندما اطلق عليها كلمة فئات عددية او فئات من النوع           
لفئـات  ، حيث ان اClass … End Classفلا اعني انها فئات حقيقية كما تعرفها بداخل التركيب 

، بينما الفئات السابق ذكرها مـن  Reference Typeالتي تعرفها بنفسك هي من النوع المرجعي 
  .Value Type القيمة ذوالنوع 

ــوع  ــن الن ــات م ــا ذوالفئ ــشتقة وراثي ــة م ــائيو– القيم ــشكل تلق ــة -ب ــن الفئ  م
System.ValueType        مثـل الطريقـة    ( الاداء   متـشابهة فـي   ، حيث انها تحتوي على طـرق

Equals()      والتي تم إعادة قيادتها Overrides         بحيث تقارن القيم التي تحملها المتغيرات ولـيس 
  .)الكائنات التي تشير لها المؤشرات

، إلى Reference Type من النوع المرجعي NET Framework.معظم فئات إطار عمل 
نما الفئـات او    بي. Strings، والحروف   Arraysجانب الفئات التي تعرفها بنفسك، والمصفوفات       

 جميعها مـن    Enums او   Structures، التركيبات من النوع     Numbersجميع البيانات العددية    
المزيد أيضا، يمكنك معرفة ما اذا كانت فئة معينة من النـوع  . Value Typeالأنواع ذات القيمة 

ــي  ــة Reference Typeالمرجع ــستندات Value Type او ذو القيم ــراءة م  NET. بق

Documentation او الانتقال إلى نافذة مستعرض الكائنات  Object Browser )  9-1شـكل 
  . حيث تعطي رموز مختلفة للفئات والتركيبات)23صفحة 

 أسرع بكثير من المتغيرات المرجعيـة  Value Typeبصفة عامة، المتغيرات ذات القيمة 
Reference Typeلمتغيرات ذات القيمة تـصل  الأول ان بيانات ا:  واقل استهلاكا للذاكرة لسببين

اليها مباشرة عن طريق المتغير نفسه بينما بيانات المتغيرات المرجعية فالوصـول اليهـا يكـون         
لتصل إلى بيانات الكائن الموجود فـي القـسم         ) المتمثل في المتغير نفسه   (بالمرور أولا بالمؤشر    

Managed Heapالبيانات من الـذاكرة لا  السبب الثاني ان عملية تفريغ محتويات.  من الذاكرة 
 أي للبيانات Managed Heap فهي مخصصة للقسم Garbage Collectionتتطلب استخدام الـ 

 في برامجك، والطريقة Value Typeلذلك، يفضل دائما استخدام المتغيرات ذات القيمة . المرجعية
 Visual Basic .NET جديدة في Value Typeالوحيدة التي تمكنك من تعريف أنواع ذات قيمة 

 مع العلم بأنك لن تـستطيع تطبيـق مبـدأ الوراثـة             Structuresهي بتعريف تركيبات من نوع      
Inheritanceمع هذه التركيبات .  

مع ذلك، توجد حالات معينة تجعل المتغيرات المرجعية أسرع من المتغيرات ذات القيمـة،       
  :اليمثلا عند عملية إسناد القيمة، فلو كان لدينا التركيب الت

 
Structure Person 
    Public Name As String 
    Public Age As Integer 
End Structure 

  
فان عملية إسناد القيم بين المتغيرات من نوع هذا التركيب ستكون أبطأ بكثير فيما لو كان التركيب 

إسناد ، حيث ان عملية )Class … End Classأي عرف باستخدام (السابق من النوع المرجعي 
 التركيب التابع له المتغير، جميع محتويات وعناصرالقيم بين المتغيرات ذات القيمة تؤدي إلى نسخ 

حجمها لا يتجاوز ( فقط المؤشراتبينما نجد في المتغيرات المرجعية ان إسناد القيم يؤدي إلى نسخ 
  ). بايت4

 ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  
  الصندقة واللاصندقة

 Boxingبالصندقة ن المتغيرات ذات القيمة تعرف حالة أخرى تجعل المتغيرات المرجعية أسرع م
 Valueتحدث الصندقة نتيجة إسناد قيمة لمتغير من نوع ذو قيمة . Unboxingواللاصندقة 

Type إلى متغير من النوع Object او ارسال متغير من النوع ذو القيمة إلى اجراء يستقبل ،
  :Objectوسيطة من النوع 

 
Dim X As Integer = 10 
 
' Boxing الصندقة 
Dim Y As Object = X 

  
، Reference Typeعملية الصندقة تؤدي إلى إنشاء متغير مخفي من النـوع المرجعـي   

وان .  السابق في كل مرة تود الوصول إلى القيمة الفعلية التي يحتويها المتغير      Yيستخدمه المتغير   
، )Value Typeغير ذات قيمـة   إلى متObjectإسناد قيمة متغير من النوع (قمت بتطبيق العكس 

  :Unboxingفانك ستحدث ما يعرف باللاصندقة 
 
 بافتراض ان العبارة Option Strict On مفعلة '
' CInt لذلك استخدمت الدالة 
X = CInt(Y) 

  
الصندقة واللاصندقة تستهلك وقت إضافي في عملية إسناد القيم، او حتى إرسال القيم إلـى               

 التـابع للكـائن   ()WriteLineكالطريقـة  (لـى الإجـراءات    إObjectوسيطات مـن النـوع      
ArabicConsole .( خلاصة القول، استخدم البيانات ذات القيمةValue Type دائما ان كنت لا 

ترغب في تطوير النوع باستخدام الوراثة ولا تتوقع عمليات إسناد قيم متعددة او حدوث أي عمليات 
  .صندقة او لاصندقة أخرى

  
  فیةالفئات الحر

 عليك إسناد قيمة له قبل الوصـول إلـى     Stringعند التصريح عن متغير حرفي جديد من النوع         
  :أعضائه، فالشيفرة التالية ستظهر رسالة خطأ

 
Dim x As String 
 
 رسالة خطأ '
ArabicConsole.WriteLine(x.Length) 

لتابعة له، ولانشاء نسخة     ا Length قبل استخدام الخاصية     xوذلك لاننا لم ننشئ نسخة من الكائن        
 او اسند قيمة ابتدائيـة للمتغيـر لحظـة    Newجديدة من الكائن الحرفي استخدم الكلمة المحجوزة  

  :التصريح عنه
 
' String ولیس Char  تستقبل قیمة من النوعالأولىالوسیطة  
Dim X As New String("A"c, 5) 
 
Dim Y As String = "AAAAA" 

  
ك التي قد تسببه لك فكرة تحول المتغيرات الحرفيـة مـن سلاسـل              اعلم جيدا مدى الارتبا   

 إلى كائنات منشأة من فئات، ولكن عليك من الآن وصـاعداً          -في لغات البرمجة السابقة   –حروف  
وبما أن المتغيرات الحرفية من . التأقلم مع هذا الأمر حتى لا تسبب في ايجاد الشوائب في برنامجك

، فان عملية إسناد القيم بين المتغيرات الحرفية لا تـؤدي إلا  Reference Typeالنوع المرجعي 
  :لنسخ مؤشرات الكائنات فقط

 
Dim X As String = "قیمة حرفیة" 
Dim Y As String = X 
 
' Y و X اثبات ان كلا المؤشرین 
  نفس الكائنإلىیشيران  '
ArabicConsole.WriteLine(X Is Y) ' True 

  
التي (إن زدتك من الشعر بيت وأخبرتك انه حتى الثوابت الحرفية           قد اسبب لك إرباكا أكثر      

 علـى انهـا   Visual Basic .NETيتعامل معها ") و"تكتبها داخل علامات التنصيص المزدوجة 
 تظهر IntelliSenseجرب كتابة نقطة بعد أي ثابت حرفي لترى قائمة . Stringكائنات من النوع 

  .) في الصفحة التالية1-6شكل ( Stringلك خصائص وطرق الفئة 

  .String مع الثوابت الحرفية من النوع IntelliSense ظهور قائمة :1-6شكل 

  
  . من الذاكرةManaged Heapتحفظ في القسم ) والتي هي كائنات(لا تنسى ان الثوابت الحرفية 

  لاVisual Basic .NETدعني اضيف شيء اخر مهم حول الثوابت الحرفية، مترجم اللغة 
ينشئ إلا نسخة واحدة من كل ثابت حرفي، مما يعني انه حتى لو وجد اكثر من ثابت حرفي يحمل         

  :نفس القيمة، فلن يتم الاحتفاظ إلا بثابت واحد فقط، ففي السطر التالي
 
Dim X As String = "ثابت حرفي" 
Dim Y As String = "ثابت حرفي" 
 
ArabicConsole.WriteLine(X Is Y) ' True 

  
 يشيران إلى كائن حرفي واحد فقط بالرغم من اننا عرفنا ثابتين Y و Xتشف ان المؤشران نك

 فسينتج عنه نفس قيمة &ليس هذا فقط، بل حتى عند استخدام معامل الدمج . حرفيين مختلفين
  :Compilingالثابت الحرفي اثناء الترجمة 

 
Dim X As String = "ثابت حرفي" 
Dim Y As String = " ابتث  "حرفي" & " " & "
 
ArabicConsole.WriteLine(X Is Y) ' True 

  الخصائص والطرق
، الخاصـية  Chars و Length إلا على خاصتين فقط همـا     Stringلا تحتوي الفئات من النوع      

 تعود بقيمة مـن  Charsالأولى تعود بعدد الحروف الموجودة في المتغير الحرفي، بينما الخاصية     
  ):يبدأ الترقيم عادة من الصفر(الحرف الذي تحدد رقمه في السلسلة الحرفية  تمثل Charالنوع 

 
Dim X As String = "ABCDE" 
 
ArabicConsole.WriteLine(X.Chars(0))   ' A 
ArabicConsole.WriteLine(X.Chars(X.Length - 1)) ' E 

  
  : داخل محرر الشيفرةلا تنسى ان الحروف العربية تحفظ بترتيب معاكس للترتيب الذي تظهر به

 
Dim x As String = " ث-ت-ب-ا " 
Dim y As String = "A-B-C-D" 
 
 
ArabicConsole.WriteLine(x.Chars(0)) ' ا 
ArabicConsole.WriteLine(x.Chars(6)) ' ث 
 
ArabicConsole.WriteLine(y.Chars(0)) ' A 
ArabicConsole.WriteLine(y.Chars(6)) ' D 
 

  
 المخصصة للتعامـل  Methods على العديد من الطرق    Stringsن النوع   تحتوي الفئات م  

، اما هنا فلـن  NET Documentation.المباشر مع الحروف، تجد شرحا مفصلا في مستندات 
وقبل ان اعرض لك هذه الطرق، عليـك        . أقدم لك إلا عرض سريع ومختصر لبعض هذه الطرق        

بقيمة  حيث ان استخدام هذه الطرق تؤدي إلى العودة       أبدا، لا تتأثر معرفة ان قيمة المتغير الحرفي      
  . ولا تؤثر بأي حال على القيمة الأصلية التي يحملها المتغير الحرفيحرفية جديدة

 التي تمكنك من حذف جميع المسافات التي قد تكون موجودة فـي             ()Trimسأبدأ بالطريقة   
  :بداية ونهاية المتغير الحرفي

 
Dim MyString As String = "  حذف المسافات  " 
 
ArabicConsole.WriteLine(MyString.Trim) ' حذف المسافات 

  
وهـي غيـر محـدودة العـدد        (Char من النوع    ()Trimيمكنك إرسال وسيطة أخرى للطريقة      

ParamArray (بحيث تحدد فيها الحروف التي تود حذفها في بداية ونهاية المتغير الحرفي:  

Dim MyString As String = " ** حذف المسافات وعلامات النجمة ** " 
 
ArabicConsole.WriteLine(MyString.Trim(" "c, "*"c)) 

  
 وهي لحذف ()TrimEnd و ()TrimStart السابقة، توجد الطريقتين Trimبالإضافة إلى الطريقة 

  .المسافات قبل وبعد المتغير الحرفي
  

    انظر أيضا
  

 في الفـصل الثـاني   ParamArrayناقشت الوسیطات غیر محدودة العدد  
  .لغة البرمجة

  
  

 تمكنك من اختبار الحروف التي يبدأ او ينتهي       ()EndsWith و   ()StartsWithالطريقتان  
  :بها المتغير الحرفي

 
Dim MyName As String = "تركي العسيري" 
 
 تحقق من الاسم الأول '
If MyName.StartsWith("تركي") Then 
 
  الاخيرتحقق من الاسم '    
    If MyName.EndsWith("العسيري") Then 
        … 
        … 
    End If 
 
End If 

  
قيمة حرفية إلى قيمة المتغير الحرفي في الموقـع         ) حشر( بإضافة   ()Insertتقوم الطريقة   

 فهي تحذف عدداً معينا من الحروف من السلسلة الحرفيـة           ()Removeالذي تحدده، إما الطريقة     
  : والموقع الذي تحدده عبر وسيطات هذه الطريقةبالطول

 
Dim MyName As String = "تركي العسيري" 
 
ArabicConsole.WriteLine(MyName.Insert(4, " ابراهیم"))' تركي ابراهیم العسيري 
ArabicConsole.WriteLine(MyName.Remove(4, 8))  ' تركي 

    ملاحظة

  
  .لحرفي يبدأ بصفرترقیم مواقع الحروف في الكائن ا

  
  

 هو استبدال قيمة حرفيـة بـأخرى فـي المتغيـر الحرفـي،              ()Replaceوظيفة الطريقة   
  :محمد صلى االله عليه وسلم بـ محمدفالاستدعاء التالي سيستبدل جميع الكلمات 

 
Dim Story As String 
 
… 
… 
ArabicConsole.WriteLine (Story.Replace ("محمد", "  ((" وسلممحمد صلى االله علیه

  
 تقوم بتقسيم الجملة الحرفية إلى مصفوفة ويتم توزيعها استنادا إلى نـوع             ()Splitالطريقة  

القيمة التي تحددها كوسيطة ترسلها لهذه الطريقة، ففي الشيفرة التالية سيتم فصل كلمات الجملـة               
  : مع الوسيطةc" "وذلك لأني أرسلت الحرف 

 
Dim MyString As String = " ت هذه الجملةفصل كلما " 
 
Dim Words As String() = MyString.Split(" "c) 
Dim counter As Integer 
 
For counter = 0 To UBound(Words) 
    ArabicConsole.WriteLine(Words(counter)) 
Next 

  
  :مخرجات الشيفرة السابقة ستكون

 
 فصل
 كلمات
 هذه
 الجملة

 
 تحوي  السلسلة Charتي ستعيد لك مصفوفة من نوع  وال()ToCharArrayأيضا، لديك الطريقة 

  .النصية
 
Dim Words As Char() = MyString.ToCharArray() 

 موجه بشكل خاص للحروف الأبجدية الإنجليزية، حيث تقلب         ()ToUpperاخيرا، الطريقة   

  : إلى الصغيرة()ToLower بينما Capital Lettersالحروف إلى الحروف الكبيرة 
 
Dim MyString As String = "I like Visual Basic .NET" 
 
ArabicConsole.WriteLine(MyString.ToUpper) ' I LIKE VISUAL BASIC .NET 
ArabicConsole.WriteLine(MyString.ToLower) ' I like visual basic .net 

  
   الحروفمقارنة

، حيث تعود ()Compareلمشتركة يمكنك إجراء المقارنات بين القيم الحرفية باستخدام الطريقة ا
- ان كانت الأولى اكبر من الثانية، والقيمة 1 عند تساوي الوسيطتين المرسلتين، والقيمة 0بالقيمة 

  : ان كانت الأولى اصغر من الثانية1
 
Dim s1 As String = "AAA" 
Dim s2 As String = "BBB" 
 
Select Case String.Compare(s1, s2) 
    Case 0 
        ArabicConsole.WriteLine("s1 = s2") 
    Case 1 
        ArabicConsole.WriteLine("s1 > s2") 
    Case -1 
        ArabicConsole.WriteLine("s1 < s2") 
End Select 

  
، Case-sensitiveأود ان انوه هنا بان المقارنة تشمل حالة الحروف الابجدية الانجليزيـة           

، وان أردت ()Compare عند التعامـل مـع الطريقـة    ”TURKI“ ي لا تساو  "turki"حيث ان   
  :Trueتجاهل حالة الحروف، يمكن إرسال الوسيطة 

 
ArabicConsole.WriteLine(String.Compare("TURKI", "turki", True)) ' 0 
 
وعند الحديث عن الحروف العربية، فيتوجب عليك معرفة ان علامات التشكيل لها اثر كبير        

" محمد"بينما علامة الكشيدة سيتم تجاهلها حيث أن ) لاحقا ستتعلم كيف يمكنك تجاهلها(ند المقارنة ع
  ":مــحــمــد"تساوي 

 
ArabicConsole.WriteLine (String.Compare ("ٌكلمة" ,"كَلِمة")) ' 1 
ArabicConsole.WriteLine(String.Compare("مــحــمــد", "محمد")) ' 0 

    ملاحظة

  
رغم ان الكشیدة تم تجاھلھا في الشیفرة الـسابقة، إلا أن ذلـك قـد لا             
يحدث دائماً، فعند تجربتي للشیفرة السابقة كانت الاعدادات الإقلیمیـة   

المملكة العربیة (العربیة في نظام التشغیل لجھازي الشخصي ھي 
، لذلك قد تظھر لك نتائج مخالفة عند تنفیذ الشیفرة على        )السعودية

  .مھیأة لمناطق أخرىأنظمة 
  

  
 السابقة وهي الطريقـة     ()Compareتوجد طريقة أخرى أسرع بخمس مرات من الطريقة         

CompareOrdinal()              حيث ان المقارنة هنا تتم استنادا إلى قيمة الحرف في جـدول الترميـز ،
UNICODE الخ...سواء العربية، الإنجليزية ( وتتجاهل مقاييس ومواصفات اللغة الحالية:(  

 
ArabicConsole.WriteLine(String.CompareOrdinal ("محمد" ,"محمد")    ' 0 
ArabicConsole.WriteLine(String.CompareOrdinal ("مـحـمـد" ,"محمد") ' -19 

  
  CultureInfoالفئة 

 NET. ذكرت لك ان إطار عمل Visual Basic .NETتعرف على في الفصل الأول 

Frameworkت العالم الطبيعية وبما أن اللغة العربية إحدى هذه اللغات، فان  موجه إلى جميع لغا
 في الاعتبار، ةوتوزيع محارف لوحات المفاتيح العربية مأخوذنظام الفرز والمواصفات والمقارنة 

  .الخ...كذلك الحال مع مواصفات البيئة كتنسيق العملة، الوقت والتاريخ، الارقام، 
م الطبيعية الأخرى في برامجك، فستجد كل ما تريده فـي   عندما تريد ان تخاطب لغات العال     

، حيث يوفر لك هذا المجال مجموعة من الفئات التـي           System.Globalizationمجال الاسماء   
 NET.يمكن أن استخدامها لإعطاء نكهة محلية لبرامجك ومشاريعك التي تنجزها في إطار عمل 

Framework  
، وبما انني لـن  NET Documentation.ندات ستجد شرحا مفصلا لهذه الفئات في مست

 في هذا الكتاب، فستجد في ثنايا هذا الكتاب بعض الشروح      Microsoftأتجرأ على ترجمة مراجع     
  .CultureInfo هذه الفئات، والبداية ستكون مع الفئة -وليس كل-التي تتناول بعض 
 فإذا اردت تطبيق انظمـة       من محاكاة البيئة المحلية في الدولة،      CultureInfoتمكنك الفئة   

عنـدما  ) الـخ ...كتنسيق نظام العملة، اللغة الرسمية، الوقت والتاريخ (الدولة المعينة في برامجك     
 للغة والدولة مـع     LCIDرقم المعرف   تنشئ كائن جديد من هذه الفئة، ارسل الاسم المختصر أو           

سـتدعاء الطريقـة   المشيد، وان كنت لا تعلـم الاسـم او رقـم المعـرف المناسـب، يمكنـك ا          

CurrentCulture()         الشيفرة التالية  .  التي تعود بكائن يمثل الاعدادات الإقليمية في النظام الحالي
ستزودنا بالاعدادات الإقليمية للجهاز الحالي، والمخرجات المعروضة أمامك تمثـل الاعـدادات            

  :الإقليمية لجهازي الشخصي
 
لتاليبافتراض انه تم استيراد مجال الاسماء ا '  
Imports System.Globalization 
… 
… 
 العودة بالاعدادات الاقلیمیة للجهاز الحالي '
Dim KSA As CultureInfo = CultureInfo.CurrentCulture() 
 
With KSA 
    ' ar-SA 
    ArabicConsole.WriteLine(.Name)       
    ' Arabic (Saudi Arabia) 
    ArabicConsole.WriteLine(.EnglishName) 
)المملكة العربیة السعودیة(العربیة  '      
    ArabicConsole.WriteLine(.NativeName)  
    ' 1025 
    ArabicConsole.WriteLine(.LCID) 
End With 

  
 او رقم ar-SA يرمز لها بالرمز المملكة العربية السعوديةمن الشيفرة السابقة، علمت ان 

  :طيع إنشاء كائن يحمل اعدادات المملكة العربية السعودية على الفور، لذلك أست1025المعرف 
 
Dim KSA As New CultureInfo("ar-SA") 
 او '
Dim KSA As New CultureInfo(1025) 

  
  :يمكنك استخدام الأسماء الأخرى لباقي الدول العربية والموضحة في الجدول بأعلى الصفحة التالية

  ةالدول  المعرف  الرمز
ar-SA  0x0401 المملكة العربية السعودية  
ar-DZ  0x1401 الجزائر.  
ar-BH  0x3C01 البحرين.  
ar-EG  0x0C01 مصر.  
ar-IQ  0x0801 العراق.  
ar-JO  0x2C01 الاردن.  

ar-KW  0x3401 الكويت.  
ar-LB  0x3001 لبنان.  
ar-LY  0x1001 ليبيا.  

ar-MA  0x1801 المغرب.  
ar-OM  0x2001 عمان.  
ar-QA  0x4001 قطر.  
ar-SY  0x2801 سورية.  
ar-TN  0x1C01 تونس.  
ar-AE  0x3801 الامارات.  
ar-YE  0x2401 اليمن.  

ar 0x0001 باقي الدول العربية.  
  

والان سأريك كيف يمكن الاستفادة من هذا الكائن لإجراء المقارنات، لنبدأ أولا بمعرفة مدى    
ذكرت ان تجاهل الكشيدة قد     ) 243صفحة  (لملاحظة السابقة   تأثيرها على الكشيدة، حيث أني في ا      

يختلف من نظام إلى آخر، ولكن الشيفرة التالية ستتجاهل الكشيدة عند المقارنـة مهمـا اختلفـت                 
  :الاعدادات الإقليمية لنظام التشغيل الحالي

 
' 0 
ArabicConsole.WriteLine(String.Compare("مــحــمــد", "محمد"), _  
    , False, New CultureInfo("ar-SA"))) 
 

 تحتوي على نسخة محسنة من      CompareInfo هي   CultureInfoتوجد خاصية في الفئة     
بحيث تمكنك من تجاهل علامات     ) Stringالموجودة في الفئات من النوع       (()Compareالطريقة  

  : لهاCompareOptions.IgnoreSymbolsالتشكيل وذلك بارسال القيمة 

Dim KSA As New CultureInfo("ar-SA") 
 
' 1 
ArabicConsole.WriteLine(String.Compare("ٌكلمة", "كَلِمة") 
 
' 0 
ArabicConsole.WriteLine(KSA.CompareInfo.Compare("ٌكلمة", "كَلِمة", _  
   CompareOptions.IgnoreSymbols)) 

  
 ان رغبـت    CompareOptions.IgnoreNonSpaceالمزيد أيضا، يمكنك إرسال القيمة      

  :في تجاهل الهمزة الموجودة على حرف الألف
' 0 
ArabicConsole.WriteLine(KSA.CompareInfo.Compare(" أحمد", "احمد ", _  
    CompareOptions.IgnoreNonSpace)) 

  
، لذلك Enumلا تنسى ان القيم التي ترسلها في الوسيطة الثالثة ما هي إلا قيم لتركيبات من النوع           

  :Or في مقارنة واحدة ان استخدمت المعامل الألفستطيع تجاهل علامات التشكيل وهمزة ت
 
ArabicConsole.WriteLine(KSA.CompareInfo.Compare(s1, s2, _  
    CompareOptions.IgnoreNonSpace Or CompareOptions.IgnoreSymbols)) 
 
 

    ملاحظة
  

 إلـى   CompareOptions.IgnoreSymbolsمن احد عیـوب إرسـال القیمـة         
، ھــو ان عملیــة التجاھــل ستــشمل ()CompareInfo.Compareالطريقـة  

تجاھــل علامــات ورمــوز أخــرى كالمــسافات، علامــات التنــصیص، والرمــوز  
  .الخ% ....رمز النسبة المئوية : الخاصة مثل

  

  
  البحث عن الحروف

ر الحرفي، حيث تعود برقم  تمكنك من البحث عن كلمة في المتغي()IndexOfاستخدامك للطريقة 
  : ان لم توجد القيمة التي تبحث عنها1-يمثل موقع الحرف الأول من القيمة المرسلة، او القيمة 

 
Dim x As String = "Can you find the word find?" 
 
ArabicConsole.WriteLine(x.IndexOf("find")) ' 8 
ArabicConsole.WriteLine(x.IndexOf("Find")) ' -1 

 مرنة جدا فهي تسمح لك بتحديد نقطة البداية للبحـث، لتـتمكن مـن        ()IndexOfالطريقة  
  :البحث عن مواقع جميع العبارات المتكررة

 
Dim x As String = "Can you find the word find?" 
Dim index As Integer = 0 
 
Do 
    index = x.IndexOf("find", index + 1) 
 
    If index < 0 Then 
        Exit Do 
    Else 
        ArabicConsole.WriteLine(index) 
    End If 
Loop 

  
وعند الحديث عن الحروف العربية، فالكشيدة سيتم تجاهلها بحسب الاعدادات الإقليمية للنظام 

  :الحالي، بينما سيتم أخذ علامات التشكيل بعين الاعتبار
 
Dim x As String = " ة؟ـث عن هذه الكَلِمة او تلك الكـلـمهل تستطیع البح " 
 
ArabicConsole.WriteLine(x.IndexOf("الكلمة")) ' 39 
ArabicConsole.WriteLine(x.IndexOf("الكَلِمة")) ' 23 

  
 مرة أخرى واستخـدم نـفـس اسـم CultureInfoولتجاهـل علامات التشكيل، عد إلى الفئة 

 مـع إرسال القيمـة CompareInfoللخـاصـية  ولكن الـتــابعـة ()IndexOfالطريقـة 
CompareOptions.IgnoreSymbols:  

 
Dim x As String = " ة؟ـهل تستطیع البحث عن هذه الكَلِمة او تلك الكـلـم " 
Dim KSA As New CultureInfo("ar-SA") 
 
' 23 
ArabicConsole.WriteLine(KSA.CompareInfo.IndexOf(x, "الكلمة", _  
    CompareOptions.IgnoreSymbols)) ' 23 

  
  Charالفئات من النوع 

، ولكن عليك معرفة ان القيمة التي تحملها المتغيرات Charلا يوجد الكثير لأخبرك به عن الفئة 
 مع الثابت عند إسناد هذا c لا تتعدى حرف واحد، كما أن عليك إرفاق الذيل Charمن النوع 

  :Stringن النوع الحرف لتمييزه عن الثابت الحرفي م
 
Dim X As Char = "A"c 
Dim Y As Char = "B" ' خطأ 

 Shared Methods على مجموعة من الطرق المـشتركة  Charتحتوي الفئات من النوع 

 Boolean، وتعود بقيمة منطقية من النوع )()Isxxxبالصيغة (التي تمكنك من تحديد نوع الحرف  
  :هاتمثل نتيجة الاختبار، اذكر لك بعضا من

 
 هل الحرف المرسل رقم '
ArabicConsole.WriteLine(Char.IsDigit("1"c))  ' True 
حرف ابجديهل الحرف المرسل  '  
ArabicConsole.WriteLine(Char.IsLetter("ت"c))  ' True 
رقم او حرفهل الحرف المرسل  '  
ArabicConsole.WriteLine(Char.IsLetterOrDigit("X"c)) ' True 
حرف ابجدي صغيررف المرسل هل الح '  
ArabicConsole.WriteLine(Char.IsLower("a"c))  ' True 
علامة تنصیص مفردة او مزدوجةهل الحرف المرسل  '  
ArabicConsole.WriteLine(Char.IsPunctuation(""""c)) ' True 

  
طة إضافية  إرسال وسيى إلى هذه الطرق ولكن لا تنسStringيمكنك أيضا إرسال قيمة من النوع 

  :تمثل رقم الحرف المطلوب اختباره في الوسيطة الأولى المرسلة
 
ArabicConsole.WriteLine(Char.IsDigit("A1", 0)) ' False 
ArabicConsole.WriteLine(Char.IsDigit("A1", 1)) ' True 

  
 Value Type هي بيانات من النوع ذو القيمة Charالبيانات من النوع : نقطة هامة أخيرة

لذلك، ضع الفروق بين البيانات . String كالبيانات من النوع Reference Typeوليست مرجعية 
  .المرجعية وذات القيمة في اعتبارك دائما عند التعامل مع هذه البيانات

  
  StringBuilderالفئات من النوع 

ير قيمتها أبدا، وإنما  لا تتغ-سواء كانت ثوابت او متغيرات–كما ذكرت لك سابقا، البيانات الحرفية 
كاستدعاء طريقة تغير عدد او محتـوى  (يتم إنشاء نسخ جديدة من القيمة الحرفية ان دعت الحاجة          

  :، فالشيفرة التالية)الحروف
 
Dim x As String = "برعي ابو" 
 
ArabicConsole.WriteLine(x)     ' برعي ابو 
ArabicConsole.WriteLine(x.Insert(8, " جبهة")) ' برعي ابو جبهة 
ArabicConsole.WriteLine(x.Remove(4, 4))   ' برعي 

و " جبهـة " و " برعـي ابـو  " هي Managed Heapقامت بإنشاء ثلاثة كائنات حرفية في ذاكرة 
  :، وكأنك قمت بإنشاء ثلاثة مؤشرات لها"برعي ابو جبهة"

 
Dim x As String = "برعي ابو" 
Dim y As String = x.Insert(8, " جبهة") 
Dim z As String = x.Remove(4, 4) 
 
ArabicConsole.WriteLine(x) ' برعي ابو 
ArabicConsole.WriteLine(y) ' برعي ابو جبهة 
ArabicConsole.WriteLine(z) ' برعي 

  
 فمن المؤكد أنه لن يتم Reference Typeو بما ان القيم الحرفية هي من النوع المرجعي 

وفي كل . Garbage Collection منها إلا باستخدام المجموعة Managed Heapتحرير ذاكرة 
مرة تسند قيمة او تعدل في محتوى قيمة متغير حرفي، سيتم إنشاء نسخة جديدة من الكـائن ممـا              

  .يؤدي إلى بطء في التعامل مع البيانات الحرفية بصفة عامة والحروف الطويلة بصفة خاصة
  

    انظر أيضا
  

 وكیفیــة  Garbage Collectionل عـن المجموعـة   كـان لـي حـديث مطـو    
 سابقا في الفصل Reference Typeتحرير الذاكرة من البیانات المرجعیة 

  .الفئات والكائناتالثالث 
  

  
، حيث تمكنك هـذه الفئـة مـن         System.Text.StringBuilderمن هنا يأتي دور الفئة      

لتتم عملية تعـديل قـيم ومحتـوى    ) String Bufferتسمى (احتجاز مساحة ثابتة للقيم الحرفية 
  .البيانات الحرفية في نفس المكان دون الحاجة إلى إنشاء نسخة جديدة من الكائن الحرفي

 او Capacity عن طريق الخاصـية  String Bufferيمكنك تحديد عدد الحروف في الـ 
  :لحظة إنشاء الكائن وإرسال القيمة إلى وسيطة المشيد

 
 حجم الـ String Buffer  بایت100 '
  حرف50فهو يحمل  '
Dim x As New System.Text.StringBuilder() 
x.Capacity = 100 
 
 او يمكنك إرسال عدد الحروف مع المشید '
Dim x As New System.Text.StringBuilder(50) 

 Stringتستطيع استخدام معظم الطرق والخصائص الموجودة في الفئات الحرفية من النوع          

مع الإشارة إلى ان قيمة المتغيـر       )  وغيرها ()Insert()  ،Removeكـ  (ل مع هذا المتغير     للتعام
  : ستتأثر باستخدام هذه الطرق-خلافاً للبيانات الحرفية الأخرى- StringBuilderمن النوع 

 
x.Insert(0, "برعي") 
x.Insert(4, " ابو جبهة") 
 
 تأثرت قیمة الكائن '
ArabicConsole.WriteLine(x) ' برعي ابو جبهة 

  
 تمكنك من إضافة نصوص     ()Appendتوجد طريقة إضافية في هذا النوع من الفئات هي          

  :في نهاية السلسلة الحرفية
 
Dim x As New System.Text.StringBuilder() 
Dim counter As Integer 
 
For counter = 1 To 9 
    x.Append(counter) 
Next 
 
ArabicConsole.WriteLine(x) ' 123456789 

  
 ليست كالمتغيرات الحرفية StringBuilderمن الضروري معرفة ان المتغيرات من النوع 

، فلا String، بل هي نوع خاص يختلف في بنيته التحتية عن الفئات من النوع             Stringمن النوع   
  :StringBuilderتحاول مثلا إسناد قيمة ثابت حرفي إلى متغير من النوع 

 
Dim Y As New System.Text.StringBuilder(50) 
 
Y = "رسالة خطأ ' "عباس السریع 

  
  الفئات العددية

 تختلف في NET Framework.الفئات العددية هي مجموعة من الأنواع المعرفة في إطار عمل 
، Byte ،Short ،Integer ،Long ،Single: نوع وحجم القيمة العددية التي تسندها اليها وهي

Double و ،Decimal ذات القيمة من الأنواع  هذا النوع من البياناتValue Type.  
، Int16 : هـي  مسميات أخرى تدل علـى حجمهـا       Long، و   Short  ،Integerللأنواع  

Int32   و ،Int64              على التوالي، قد تستخدمها لتضع حجم المتغيرات دائما في الاعتبـار لحظـة 
  .استخدامها

    انظر أيضا

  
فـي   إلـى المتغیـرات العدديـة     لتـي تـسندھا     تجد تفاصیل مجال القـیم ا     

 بالـصفحات وبالتحديد في الجدول الموجـود  لغة البرمجة  الثاني   الفصل
49-50.  
  

  
  الخصائص والطرق

 والقيم التي تحملها المتغيرات بالأعداد مرتبطةجميع الفئات العددية تتشارك في خصائص وطرق 
 اكبر و تمثللتان تعودان بقيمة ل واMinValue و MaxValue معك بالخاصيتين وسأبدأ. العددية
 الذي يمكن للمتغير العددي ان يحمله، مع العلم ان هذه الخصائص الأعدادضمن مجال قيمة اصغر 

  :Shared ومشتركة ReadOnlyللقراءة فقط 
 
Dim X As Byte 
Dim Y As Integer 
Dim Z As Long 
Dim W As Double 
 
ArabicConsole.WriteLine(X.MinValue) ' 0 
ArabicConsole.WriteLine(Y.MinValue) ' -2147483648 
ArabicConsole.WriteLine(Z.MaxValue) ' 9223372036854775807 
ArabicConsole.WriteLine(W.MaxValue) ' 1.79769313486232E+308 

  
علـى  ) Double و   Singleكـ   (Floating-point ذات الفاصلة العائمة     تحتوي الأنواع 

غيـر  – والتي تعود بقيمة اصغر عدد عشري موجـب          ()Epsilonإضافية كالخاصية    خصائص
  : يمكن للمتغير ان يحمله-الصفر

 
Dim X As Double 
Dim Y As Single 
 
 
ArabicConsole.WriteLine(X.MinValue) ' -1.79769313486232E+308 
ArabicConsole.WriteLine(X.Epsilon)  ' 4.94065645841247E-324 
 
ArabicConsole.WriteLine(Y.MinValue) ' -3.402823E+38 
ArabicConsole.WriteLine(Y.Epsilon)  ' 1.401298E-45 

  الأعدادتنسیق 

 يعود بالقيمة التي يحملها المتغير العددي ولكن من النوع الحرفي       ()ToStringاستدعائك للطريقة   
String:  

 
Dim x As Integer = 100 
 
ArabicConsole.WriteLine(x.ToString) ' 100 

  
، ()ToStringيمكنك إضافة مجموعة من الرموز لتنسيق الأعداد وإرسالها كوسيطة مع الطريقة            

  : الذي يمثل عدداً او الرقم صفر0 الذي يمثل عدداً او مسافة خالية، او رمز الصفر #كالرمز 
 
Dim x As Integer = 10 
 
ArabicConsole.WriteLine(x.ToString("####")) ' 10 
ArabicConsole.WriteLine(x.ToString("0000")) ' 0010 

  
توجد مجموعة إضافية من الرموز التي يمكنك استخدامها، تجد شرحا وافيا لها في مستندات   

.NET Documentation الذي يمثل فاصلة "," الذي يمثل الفاصلة العشرية، او الرمز "." كالرمز 
 الـذي يمكنـك مـن    "E"ذي يؤدي إلى عرض النسبة المئوية، او الرمز  ال"%"الآلاف، او الرمز  

  :Exponential Formعرض الرقم في الصورة الأسية 
 
Dim X As Single = 100.1234 
Dim Y As Integer = 9999999 
Dim Z As Double = 0.1 
Dim w As Long = 1500000000000 
 
ArabicConsole.WriteLine(X.ToString("0000.#")) ' 0100.1 
ArabicConsole.WriteLine(Y.ToString("###,###,###")) ' 9,999,999 
ArabicConsole.WriteLine(Z.ToString("##.0 %")) ' 10.0 % 
ArabicConsole.WriteLine(w.ToString("### E+0")) ' 150 E+10 

  
ات الإقليميـة   من ناحية أخرى، توجد مجموعة جاهزة من التنسيقات العددية تتأثر بالاعداد          

العربية الحالية في الجهاز، المخرجات التالية مرتبطة بالاعدادات الإقليمية لجهازي الشخصي وهي 
  ):المملكة العربية السعودية(

 
Dim MyDouble As Double = 123456789 
 
' Currency عملة 
ArabicConsole.WriteLine(MyDouble.ToString("C")) ' 123,456,789.00 س.ر.  
' Scientific علمي 
ArabicConsole.WriteLine(MyDouble.ToString("E")) ' 1.234568E+008 

' Percent النسبة المئویة 
ArabicConsole.WriteLine(MyDouble.ToString("P")) ' 12,345,678,900.00% 
' Number عدد 
ArabicConsole.WriteLine(MyDouble.ToString("N")) ' 123,456,789.00 
' Fixed-point فاصلة عائمة 
ArabicConsole.WriteLine(MyDouble.ToString("F")) ' 123456789.00 

  
 التابعة للفئة NumberFormatوحتى تستخدم اعدادات إقليمية أخرى، أرسل الخاصية 

CultureInfo كوسيطة ثانية للطريقة ToString() لاحظ (، المخرجات التالية من البيئة الالمانية
  ): استخدام الفواصل والنقطالفرق في

 
Imports System.Globalization 
 
… 
… 
 
Dim Germany As New CultureInfo("de-DE") 
Dim MyDouble As Double = 123456789 
 
With Germany 
    ' 123.456.789,00 € 
    ArabicConsole.WriteLine(MyDouble.ToString("C", .NumberFormat))   
    ' 1,234568E+008 
    ArabicConsole.WriteLine(MyDouble.ToString("E", .NumberFormat))   
    ' 12,345,678,900.00% 
    ArabicConsole.WriteLine(MyDouble.ToString("P", .NumberFormat))   
    ' 123.456.789,00 
    ArabicConsole.WriteLine(MyDouble.ToString("N", .NumberFormat))   
    ' 123456789,00 
    ArabicConsole.WriteLine(MyDouble.ToString("F", .NumberFormat))   
End With 

  
 مـا هـي إلا كـائن منـشأ مـن الفئـة              NumberFormatالمزيد أيـضا، الخاصـية      

NumberFormatInfo          خصائصها قابلة للقراءة والكتابة بحيث يمكنك تخصيص نظـام تنـسيق 
  :للاعداد خاص بك

 
Imports System.Globalization 
… 
Dim Custom As New NumberFormatInfo() 
Dim X As Double = -12345.6789 
 
 علامة الفاصلة العشریة ستكون فاصلة منقوطة '
Custom.NumberDecimalSeparator = ";" 
 علامة السالب ستكون نجمة '
Custom.NegativeSign = "*" 
ArabicConsole.WriteLine(X.ToString("", Custom)) ' *12345;6789 

  Mathالفئة 

رة من الطرق المشتركة  على مجموعة كبيMicrosoft.VisualBasic.Mathتحتوي الفئة 
لإجراء العمليات الحسابية والرياضية على البيانات العددية، لا تنسى استيراد مجال الأسماء 

Microsoft.VisualBasic في مشروعك حتى تتمكن من الوصول إلى هذه الفئة بكتابة اسمها 
  :فقط

 
Imports Microsoft.VisualBasic 

  
 Absolute التي تعود بالقيمـة المطلقـة   ()Absقة   الطريMathمن الطرق التابعة للفئة 

Value   والطريقة ،Sqrt()     للجذر التربيعي، والطريقة Pow()     للاس، وطريقة اللوغارثم Log() 
 ان كان العدد موجب او      1 التي تعود بالقيمة     ()Sign، والطريقة   ()Log10او اللوغارثم العشري    

  : ان كان صفر0 ان كان سالب او 1-
 
ArabicConsole.WriteLine(Math.Abs(-10))   ' 10 
ArabicConsole.WriteLine(Math.Sqrt(9))   ' 3 
ArabicConsole.WriteLine(Math.Pow(2, 3))   ' 8 
ArabicConsole.WriteLine(Math.Log(9, 3))   ' 2 
ArabicConsole.WriteLine(Math.Log10(100))   ' 2 
ArabicConsole.WriteLine(Math.Sign(Double.MinValue))  ' -1 
ArabicConsole.WriteLine(Math.Sign(Double.Epsilon))  ' 1 

  
 السابقة تعود بالجذر التربيعي للعدد، وان أردت الحصول على الجذر النوني فلا ()Sqrtالطريقة 

  : بنفسك()NthSqrمخرج لك إلا بتطوير الطريقة 
 
Function NthSqr(ByVal num As Double, ByVal root As Double) As Double 
    Return num ^ (1 / root) 
End Function 
 
… 
… 
 
ستدعائهامثال لا '  
ArabicConsole.WriteLine(NthSqr(8, 3)) ' 2 
 

 التي تعود بباقي القسمة،     ()IEEERemainderطرق أخرى تتعامل مع الاعداد كالطريقة       
قيمة صحيحة اكبر من او تساوي العدد بالتي تحذف الفاصلة العشرية وتعود    ()Ceilingوالطريقة  

  : اصغر من او تساوي العدد المرسلصحيحةقيمة ب تعود ()Floor ةالمرسل، بينما الطريق

ArabicConsole.WriteLine(Math.IEEERemainder(9, 2)) ' 1 
ArabicConsole.WriteLine(Math.Ceiling(1.2))  ' 2 
ArabicConsole.WriteLine(Math.Ceiling(-1.2))  ' -1 
ArabicConsole.WriteLine(Math.Floor(1.2))  ' 1 
ArabicConsole.WriteLine(Math.Floor(-1.2))  ' -2 

  
 حيـث  ()Roundيمكنك التحكم اكثر في خانة الفاصلة العشرية باستخدام طريقة التقريـب       

  :تحدد عدد خانات الفاصلة العشرية
 
ArabicConsole.WriteLine(Math.Round(1.1256, 1)) ' 1.1 
ArabicConsole.WriteLine(Math.Round(1.1256, 2)) ' 1.13 
ArabicConsole.WriteLine(Math.Round(1.1256, 3)) ' 1.126 

  
الخ تعود بالقيمة المناسبة استنادا      ... ()Sin()  ،Cos()  ،Tanاخيرا، طرق الدوال المثلثية     

 لمزيد من التفاصـيل  NET Documentation.راجع مستندات . إلى الزاوية المرسلة بالراديان
  .حول هذه الطرق والطرق الأخرى

  
  Random Numbers العشوائیة تولید الأعداد

 على مجموعة من الطرق التي تقوم بتوليد أعداد عشوائية، فمثلا System.Randomتحتوي الفئة 
  : موجبهInteger تعود بقيمة عشوائية صحيحة ()Nextالطريقة 

 
Dim Rnd As New Random() 
Dim counter As Integer 
 
For counter = 1 To 10 
    ArabicConsole.WriteLine(Rnd.Next) ' 2345684 
Next 

  
  :1 إلى 0 مجاله من Double فهي تعود بعدد عشوائي عشري ()NextDoubleاما الطريقة 

 
Dim Rnd As New Random() 
Dim counter As Integer 
 
For counter = 1 To 10 
    ArabicConsole.WriteLine(Rnd.NextDouble) ' 0.01234865 
Next 

  
 يمكنك تخصيص مجال معين للاعداد العشوائية بارسال اصغر قيمة واكبـر            المزيد أيضا، 

  :قيمة كوسيطات مع الطريقتين السابقتين

Dim Rnd As New Random() 
Dim counter As Integer 
 
 
For counter = 1 To 10 
10 إلى 10- تولید اعداد عشوائیة مجالها من  '      
    ArabicConsole.WriteLine(Rnd.Next(-10, 10)) 
Next 

  
 بأرقام عشوائية فـي خطـوة واحـدة         Byteاخيرا، يمكنك ملء مصفوفة عددية من النوع        

  :()NextBytesباستخدام الطريقة 
 
Dim Rnd As New Random() 
Dim counter As Integer 
Dim x(99) As Byte 
 
Rnd.NextBytes(x) 
 
For counter = 0 To UBound(x) 
    ArabicConsole.WriteLine(x(counter)) 
Next 

  
  فئات أخرى

  .لغة البرمجةسأتناول في هذا القسم بعض الفئات التي تطرقت لها سابقا في الفصل الثاني 
  

  الوقت والتاريخفئات 
، وهمـا   System.DateTime يعني تعريفك لمتغير من الفئـة        Dateفك لمتغير من النوع     تعري

  :يمثلان نفس الفئة
 
Dim today As Date 
Dim yesterday As DateTime 

  
، يمكنك إسناد هذه Date وتاريخ Time يحملان قيمة تشمل وقت yesterday و todayالمتغيران 

كارسال وسيطات السنة، الشهر، واليوم على التوالي مع : القيمة لحظة تعريف المتغير بعدة اساليب
  :المشيد

 
2002 سیبتمبر لعام 26 '  
  صباحا12:00الساعة  '
Dim today As New Date(2002, 9, 26) 

 صباحا، ويمكن تحديد الوقـت بإرسـال   12:00الوقت الذي يحمله المتغير السابق سيكون الساعة    
  :الساعة، الدقيقة، والثانية على التوالي

 
2002 سیبتمبر لعام 26 '  
مساء 1:30الساعة  '  
Dim today As New Date(2002, 9, 26, 13, 30, 0) 

  
  : التي تعود بالوقت والتاريخ الحالي()Nowلخاصية المشتركة او يمكنك إسناد ا

 
2002 سیبتمبر لعام 26 '  
صباحا 3:30:23الساعة  '  
Dim today As Date = Date.Now 

  
  :# و # او إرسال ثابت الوقت والتاريخ بين علامتي 

 
2003 لعام فبرایر 20 '  
صباحا 2:00:00الساعة  '  
Dim today As Date = #1/20/2003 2:00:00 AM# 

  
بعد إسنادك لقيمة ابتدائية لمتغير التاريخ، يمكنك الاستعلام عن أجزاء من القيمة عن طريق              

، Year  ،Month  ،Day  ،Hourهـي   ) ReadOnlyللقـراءة فقـط     (مجموعة من الخصائص    
Minute و ،Secondتمثل السنة، الشهر، اليوم، الساعة، الدقيقة، والثانية على التوالي :  

 
Dim today As New Date(2002, 9, 26, 13, 30, 0) 
 
ArabicConsole.WriteLine(today.Year)    ' 2002 
ArabicConsole.WriteLine(today.Month)   ' 9 
ArabicConsole.WriteLine(today.Day)     ' 26 
ArabicConsole.WriteLine(today.Hour)     ' 13 
ArabicConsole.WriteLine(today.Minute)  ' 30 
ArabicConsole.WriteLine(today.Second)  ' 0 

  
، فتوجـد مجموعـة مـن    Dateوعند الحديث عن العمليات الرياضية على متغير من نوع  

: الطرق لتضيف إلى قيمة المتغير الحالي سنة، شـهر، يـوم، سـاعة، دقيقـة، او ثانيـة وهـي          
AddYears()  ،AddMonths()  ،AddDays()  ،AddHours()  ،AddMinutes()  و ،

AddSeconds():  

Dim today As New Date(2002, 9, 26) 
Dim adate As Date 
 
  ایام لیكون التاریخ10اضافة  '
' 2002/10/06 
adate = today.AddDays(10) 

  
وان رغبت في إنقاص قيمة من التاريخ او الوقت، فاستخدم نفس الطرق السابقة ولكن مع إرسـال       

  :أرقام سالبة
 
Dim today As New Date(2002, 9, 26) 
Dim adate As Date 
 
  لیكون التاریخیوم 27 قاصان '
' 2002/8/30 
adate = today.AddDays(-27) 

  
 تعود بعـدد الأيـام      ()DaysInMonths، توجد طريقة مشتركة قد تفيدك كثير هي         أخيرا

  :الموجودة في الشهر الحالي
 
ArabicConsole.WriteLine(Date.DaysInMonth(2002, 1)) ' 31 

  
  :دعم التقويم الهجري

من بداية إسناد القيم للمتغير حتى الاستعلام (يات التي قمنا بها في السطور السابقة ان جميع العمل
 على التقويم الميلادي، ولكن ان حاولت -بشكل افتراضي–، كانت تعتمد )عن القيمة التي يحملها

  :طباعة قيمة المتغير، فان المخرجات ستتأثر بالاعدادات الإقليمية الحالية
 
المیلاديبالتقويم لت القیمة هنا خاد '  
Dim today As New Date(2002, 9, 26) 
 
  الحالیةالاعدادات الاقلیمیة إلىظهرت المخرجات استنادا  '
)المملكة العربیة السعودیة(العربیة  '  
 بالتقويم الهجري '
ArabicConsole.WriteLine(today) ' 1423/7/20 

  
ي، حيث ان القيمة التي يتعامل معها المتغير السبب الذي قد يفسر لك عملية التحويل السابقة تقن

today السابق تستخدم التقويم الميلادي، ولكن عند الإخراج تم استدعاء الطريقة ToString() 
  .والتي تتأثر بالاعدادات الاقليمية الحالية في النظام

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


  الفئات الأساسیة: السادسالفصل   259  
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  

 Calendarيمكنك تحويل المتغير التاريخي إلى متغير يعتمد التقويم الهجري بارسال الكائن   
  :CultureInfoالتابع للفئة 

 
Imports System.Globalization 
 
… 
… 
 
Dim KSA As New CultureInfo("ar-SA") 
 
 القیم المدخلة هنا تعتمد التقويم الهجري '
Dim today As New Date(1423, 7, 20, KSA.Calendar) 
 

المملكـة  (العربيـة  ان اعدادات رغم – السابق  todayاذا تم اعتماد التقويم الهجري في المتغير        لم
 الافتراضي تقاويم؟ والجواب هو لان التقويم الهجري هو التقويم          6 تحتوي على    )العربية السعودية 

، )مـصر (العربيـة  ، ولو حاولت تطبيق اعدادات    )المملكة العربية السعودية  (العربية  في اعدادات   
  :الميلادي وليس الهجريفعليك الحذر كل الحذر حيث ان التقويم الافتراضي لها هو 

 
Imports System.Globalization 
 
… 
… 
 
Dim Egypt As New CultureInfo("ar-EG") 
 
المیلاديالقیم المدخلة هنا تعتمد التقويم : تنبیه '  
 ولیس الهجري '
Dim today As New Date(1423, 7, 20, Egypt.Calendar) 

  
 بـدون   -ان كنت ترغب  –لتقويم الهجري   الغرض من التنبيه السابق هو نصيحتك لاعتماد ا       

الاعتماد على الاعدادات الاقليمية للجهاز كالطريقة السابقة، وانما تعريف متغير لكائن من الفئـة              
HijriCalendarبشكل مباشر :  

 
Imports System.Globalization 
… 
… 
Dim hijra As New HijriCalendar() 
 
ريالقیم المدخلة هنا تعتمد التقويم الهج '  
Dim today As New Date(1423, 7, 20, hijra) 

 تمثل التقويم الهجري بكل ما تحمل الكلمة من معنى وهي مشتقة من   HijriCalendarفئة  ال

مع ذلك، توجد مجموعة كبيرة من      . ، لذلك فهي تشتق جميع الطرق والخصائص      Calendarالفئة  
وبعضها تـم إعـادة تعريفهـا      - HijriCalendar في الفئة    Overridesالطرق تم إعادة قيادتها     

Overloadsكــــ - أيـــضا AddMonths() ،AddYears() ،GetDayOfMonth() ،
GetDayOfWeek() ...  راجع مستندات (الخ.NET Documentation  لمزيد من التفاصـيل 

  ).حول التعديلات التي طرأت عليها
ى شـهر، لـذلك عليـك    عدد الايام في الشهر الحالي يختلف من سنة إلى سنة ومن شهر إل   

  : لمعرفة عدد الايام في الشهر الحالي()GetDaysInMonthاستدعاء الطريقة 
 
Dim hijra As New HijriCalendar() 
 
ArabicConsole.WriteLine(hijra.GetDaysInMonth(1423, 7)) ' 30 
ArabicConsole.WriteLine(hijra.GetDaysInMonth(1423, 8)) ' 29 

  
 NET.شهر رمضان المبارك، فبكل تأكيد لن تعتمد على إطار عمل وعند الحديث عن 

Framework في تحديد عدد ايام الشهر الحالي، حيث ان رسول االله صلى االله عليه وسلم امرنا 
في حديثه الشريف بالاعتماد على رؤية الهلال وليس على الحساب الفلكي الذي تعتمده أجهزة 

لذلك، عليك إعلام المستخدم بانه يتوجب عليه إجراء ). يوماً 30 او 29فقد يكون الشهر (الحاسب 
، او أن تحصر ) بالصفحة المقابلة2-6شكل (التعديلات اللازمة للتقويم الهجري من لوحة التحكم 

 في إجراء التعديلات اللازمة بالتوغل داخل مسجل النظام -كمبرمج عربي–المسؤولية عليك 
Windows Registryفتاح وتعديل قيمة الم:  

 
HKEY_CURRENT_USER\Control Panel\International\AddHijriDate 

 Control في لوحة الـتحكم  Regional Options تعديل قيمة التاريخ الهجري من :2-6شكل 

Panel.  
  

    انظر أيضا
  

 يعــرض لـك مجموعــة مــن  مواضــیع متقدمــةالفـصل الــسادس عـشر   
  .Windows Registryالفئات تمكنك من تعديل مسجل النظام 

  
  

  :تنسيق الوقت والتاريخ
باختصار شديد، يمكنك استخدام مجموعة من الطرق التي تعود بقيمة حرفية للوقت والتاريخ الذي              

  :يحمله المتغير
 
Dim x As Date = Date.Now 
 
ArabicConsole.WriteLine(x.ToShortDateString) ' 26/09/2002 
ArabicConsole.WriteLine(x.ToLongDateString)  ' 26 September, 2002 
ArabicConsole.WriteLine(x.ToLongTimeString)  ' 10:41:32 م 
ArabicConsole.WriteLine(x.ToShortTimeString) ' 10:41 م 

مخرجات الطرق السابقة تتأثر باعدادات النظام الحالي، فقد تظهر لك نتائج مختلفة فـي جهـازك           

يقة عرض الوقـت التـاريخ بنفـسك يمكنـك اسـتخدام الطريقـة           الخاص، إن أردت تنسيق طر    
ToString():  

 
Dim X As New Date(2002, 12, 20, 23, 30, 0) 
 
' 2/12/20 
ArabicConsole.WriteLine(X.ToString("y/M/d")) 
' 02/12/20 
ArabicConsole.WriteLine(X.ToString("yy/MM/dd")) 
' 2002/Dec/Fri 
ArabicConsole.WriteLine(X.ToString("yyy/MMM/ddd"))     
' 2002/December/20 Friday 
ArabicConsole.WriteLine(X.ToString("yyyy/MMMM/d dddd")) 
' 11:30:00 P 
ArabicConsole.WriteLine(X.ToString("hh:mm:ss t")) 
' 11:30:00 PM 
ArabicConsole.WriteLine(X.ToString("hh:mm:ss tt")) 
' 23:30:00 
ArabicConsole.WriteLine(X.ToString("HH:mm:ss")) 
' 2002/30/20 11:30:00 PM 
ArabicConsole.WriteLine(X.ToString("yyyy/m/d hh:mm:ss tt"))  

  
    ملاحظة

  
 mm يمثل شھر، بینما رمـز الحـروف الـصغیرة          MMرمز الحروف الكبیرة    

  .فیمثل دقیقة
  

  
ونـوع التقـويم   ) الـخ ...في أسماء الأشهر او الايـام  (من ناحية أخرى، اللغة المستخدمة  

أما إن أردت الـتحكم  . تعتمد اعتماداً كلياً على اعدادات النظام الحالي    ) هجري، ميلادي، وغيرها  (
بالمناسـبة،  الخاصـية      (DateTimeFormatInfoيها، فلا بـد لـك مـن اسـتخدام الفئـة             ف

DateTimeFormat والتابعة للفئة CultureInfo كائن من النوع DateTimeFormatInfo:(  
 
Dim X As New Date(2002, 12, 20) 
Dim arabicDateFormat As DateTimeFormatInfo 
 
arabicDateFormat = New CultureInfo("ar-SA").DateTimeFormat 
arabicDateFormat.Calendar = New HijriCalendar() 
 
 الجمعة 16/شوال/1423 '
ArabicConsole.WriteLine(X.ToString("yyyy/MMMM/d dddd", _ 
      arabicDateFormat)) 

 للحصول على المزيد مـن  NET Documentation.انصحك بإلقاء نظرة إلى مستندات 
  .DateTimeFormatInfoاصيل حول استخدام الفئة القوية التف

  
  Enumالفئات من النوع 
 -بشكل تلقائي– التي تعرفها في شيفراتك المصدرية، هي مشتقة Enumجميع التركيبات من النوع 

 من تعريـف  Visual Basic .NET، والطريقة الوحيدة التي يمكنك System.Enumمن الفئة 
لغة  كما رأينا في الفصل الثاني       Enum باستخدام الكلمة المحجوزة      هو Enumتركيب من النوع    

  :البرمجة
 
Enum Programmer 
    VisualBasic 
    CSharp 
    CPlusPlus 
    Java 
    Delphi 
End Enum 

  
 بحيث Overrides تم إعادة قيادتها    Enum التابعة للفئات من النوع      ()ToStringالطريقة  

  :م العضو الذي تم تعريفه في التركيب عوضا عن القيمة التي يحملهاتعود بقيمة تمثل اس
 
Dim Turki As Programmer 
 
Turki = Programmer.VisualBasic 
 
ArabicConsole.WriteLine(Turki)  ' 0 
ArabicConsole.WriteLine(Turki.ToString) ' VisualBasic 

  
 التي تمكنـك    ()Parseيقة المشتركة    السابقة، توجد الطر   ()ToStringعلى عكس الطريقة    

  :من إسناد قيمة للمتغير بكتابة اسم العضو حرفيا
 
 في حالة '
' Option Strict Off 
Turki = [Enum].Parse(GetType(Programmer), "CSharp") 
 
ArabicConsole.WriteLine(Turki)  ' 1 
ArabicConsole.WriteLine(Turki.ToString) ' CSharp 

  
، اما ان Option Strictلتعليق السابق، لن تعمل الشيفرة السابقة إلا عند تعطيل العبارة كما في ا

  :()CTypeكانت مفعلة فعليك العود إلى استخدام المعامل 

 في حالة '
' Option Strict On 
Turki = CType([Enum].Parse(GetType(Programmer), "CSharp"), _  
        Programmer) 
 
ArabicConsole.WriteLine(Turki)  ' 1 
ArabicConsole.WriteLine(Turki.ToString) ' CSharp 

  
، Case-Sensitive تتأثر بحالة الأحرف الكبيرة والصغيرة ()Parseضع في اعتبارك ان الطريقة 

  :Trueمع ذلك تستطيع تجاهل حالة الاحرف بإرسال القيمة 
 
Turki = [Enum].Parse(GetType(Programmer), "CSharp", True) 

  
 قيم عدديـة، بحيـث   Enumالقيم التي تحملها المتغيرات المعرفة من التركيبات من النوع        

  :يمكنك إسنادها مباشرة إلى المتغير
 
Dim Turki As Programmer 
 
Turki = CType(4, Programmer) 
 
ArabicConsole.WriteLine(Turki)  ' 4 
ArabicConsole.WriteLine(Turki.ToString) ' Delphi 

  
  : للتحقق من دعم التركيب إلى العدد قبل إسناده للمتغير()IsDefinedيفضل استخدام الطريقة 

 
If [Enum].IsDefined(GetType(Programmer), 3) Then 
    Turki = CType(3, Programmer) 
End If 

  
ركيـب باسـتخدام    اخيرا، يمكنك الاستعلام عن جميع القيم واسماء الأعـضاء التابعـة للت           

 تحمـل  Stringالأولى تعود بمصفوفة من النـوع    . ()GetValues و   ()GetNamesالطريقتين  
  : تمثل قيم تلك الأعضاءObjectاسماء أعضاء التركيب والثانية بمصفوفة من النوع 

 
Dim names() As String = [Enum].GetNames(GetType(Programmer)) 
Dim values As Array = [Enum].GetValues(GetType(Programmer)) 
Dim counter As Integer 
 
For counter = 0 To UBound(names) 
    ArabicConsole.WriteLine(names(counter) & " = " & _  
         CInt(values.GetValue(counter))) 
Next 

  :مخرجات الشيفرة الموجودة في أسفل الصفحة السابقة ستكون كالتالي
 
VisualBasic = 0 
CSharp = 1 
CPlusPlus = 2 
Java = 3 
Delphi = 4 

  
  :Bit-Coded من النوع Enumتركيبات 

 بحيث Bit-Codedبتية  على انها تركيبات Enumفي البرامج الجدية، يتم تعريف تركيبات 
، Or ،Andيمكن للمتغير الواحد الاحتفاظ باكثر من قيمة يتم دمجها باستخدام المعاملات المنطقية 

XOr... لعمل ذلك، أضف المواصفة . الخFlags Attributeعند تعريف التركيب :  
 
<Flags()> _ 
Enum Programmer 
    None = 0 
    VisualBasic = 1 
    CSharp = 2 
    CPlusPlus = 4 
    Java = 8 
    Delphi = 16 
End Enum 

  
هـذه المواصـفة، فمـثلا الطريقـة        معظم الطرق السابق ذكرها ستتأثر إن تم اسـتخدام          

ToString()ستستخلص اسماء جميع الأعضاء التي يمثلها المتغير :  
 
Dim Ali As Programmer 
 
Ali = Programmer.VisualBasic Or Programmer.CSharp 
 
ArabicConsole.WriteLine(Ali)    ' 3 
ArabicConsole.WriteLine(Ali.ToString) ' VisualBasic, CSharp 

  
  Arrayالفئات من النوع 

.  عن المصفوفات وطريقة التعامل معهالغة البرمجةتحدثت بما فيه الكفاية في الفصل الثاني 
– المصفوفات التي تعرفها في برنامجك، تصبح مشتقة وراثيا جميعوسأضيف في هذه الفقرة أن 

فوفات من النوع ، ودعني أذكرك هنا بان جميع المصSystem.Array من الفئة -بشكل تلقائي
، الشيفرة Value Type حتى وان كان نوع البيانات من ذات القيمة Reference Typeالمرجعي 

  :التالية خير دليل

Dim X() As Integer = {1, 2, 3} 
Dim Y As Array = X 
 
Y.SetValue(100, 0) 
Y.SetValue(200, 1) 
Y.SetValue(300, 2) 
 
 
' X() وفةلاحظ ان قیمة عناصر المصف  
 قد تغيرت '
ArabicConsole.WriteLine(X(0)) ' 100 
ArabicConsole.WriteLine(X(1)) ' 200 
ArabicConsole.WriteLine(X(2)) ' 300 

  
 التي تم ()Cloneاما ان اردت نسخ قيم المصفوفات وليس مؤشراتها، فيمكنك استخدام الطريقة 

  :ذكرها سابقا
 
Dim X() As Integer = {1, 2, 3} 
Dim Y() As Integer = CType(X.Clone, Integer()) 
 
Y(0) = 100 
Y(0) = 200 
Y(0) = 300 
 
 لاحظ ان قیمة العناصر لم تتغير في '
' X() المصفوفة 
ArabicConsole.WriteLine(X(0)) ' 1 
ArabicConsole.WriteLine(X(1)) ' 2 
ArabicConsole.WriteLine(X(2)) ' 3 

  
 التي تمكنك من الاستعلام عـن عـدد أبعـاد           Rankاصية   الخ Arrayمن خصائص الفئة    

 تعود بقيمة تمثل عدد العناصر التي يمكن ان تحتويها المصفوفة،           Lengthالمصفوفة، والخاصية   
  : تعود بقيمة تمثل عدد العناصر في البعد المعين التابع للمصفوفةGetLenghtبينما الخاصية 

 
Dim X(9, 5) As Integer 
 
ArabicConsole.WriteLine(X.Rank)  ' 2 
ArabicConsole.WriteLine(X.Length)  ' 60 
ArabicConsole.WriteLine(X.GetLength(0)) ' 10 
ArabicConsole.WriteLine(X.GetLength(1)) ' 6 
 

  
 على المصفوفات، كن متأكـداً بـان   For Each … Nextالمزيد أيضا، عند تطبيق حلقة 

أي أن البداية ستكون من (فة ستتم بشكل أفقي وليـس عـمودي عملية المسح على عناصر المصفو

 .... (1 ,1) فالعنـصر  (0 ,1) فالعنـصر  (x ,0) فالعنصر (1 ,0) فالعنصـر (0 ,0)العنصر 
  :وهكذا

 
Dim records(,) As String = {{"خالد" ,"احمد" ,"تركي"}, _ 
 {{"الكویت" ,"الامارات" ,"السعودیة"}    
 
Dim field As String 
 
For Each field In records 
    ArabicConsole.WriteLine(field) 
Next 

  
  :مخرجات الشيفرة السابقة ستكون كالتالي

 
 تركي
 احمد
 خالد
 السعودیة
 الامارات
 الكویت

  
  :فرز عناصر المصفوفة

  : إن أردت فرز عناصر المصفوفة()Sort أسرع وافضل من الطريقة NET.لن تجد في عالم 
 
Dim X() As Integer = {5, 2, 9} 
Dim counter As Integer 
 
Array.Sort(X) 
 
For counter = 0 To UBound(X) 
    ArabicConsole.WriteLine(X(counter)) 
Next 

  
، اما ان )من الصغير إلى الكبير(الشيفرة السابقة ستقوم بترتيب عناصر المصفوفة بشكل تصاعدي 

  : مباشرة بعد عملية الفرز()Reverseالطريقة أردت جعل الفرز تنازلي فاستدعي 
 
Dim X() As Integer = {5, 2, 9} 
Dim counter As Integer 
 
Array.Sort(X) 
Array.Reverse(X) 
… 
… 

 هو إمكانية تحديد العناصر التـي تـود فرزهـا،           ()Sortمن الأشياء الظريفة في الطريقة      

–الوسطى فقط لعدم الحاجة لفرز العناصر الأخرى فالاستدعاء التالي سيقوم بفرز العناصر الثلاثة 
  :مما يزيد من سرعة التنفيذ يا عسل

 
Dim X() As Integer = {1, 2, 3, 6, 4, 5, 7, 8, 9} 
Dim counter As Integer 
 
 فرز العناصر الثلاث الوسطى فقط '
Array.Sort(X, 3, 5) 
 
For counter = 0 To UBound(X) 
    ArabicConsole.WriteLine(X(counter)) 
Next 

  
 في ترتيب العناصر العددية، فيسرني إخبارك بـان هـذه      ()Sortإلى جانب كفاءة الطريقة     

الطريقة  تعمل بنفس الكفاءة مع العناصر الحرفية أيضا، حيث انها تقوم بترتيب العناصر اسـتنادا     
  :UNICODEإلى المقابل العددي للحروف في جدول 

 
Dim X() As String = {"Turki", "Ali", "Basmah"} 
Dim counter As Integer 
 
Array.Sort(X) 
 
For counter = 0 To UBound(X) 
    ArabicConsole.WriteLine(X(counter)) 
Next 

  
 تتجاهل علامة   ()Sortوعند الحديث عن لغتنا الجميلة، فيسرني أيضا اخبارك بان الطريقة           

  .بعملية الفرز، فلا تحمل هم فرز الحروف العربيةالكشيدة وعلامات التشكيل بحيث لا تخل 
  

    ملاحظة
  

رغم ان عملیة الفرز تتم استنادا إلى المقابل العددي للحرف الأبجـدي،     
إلا ان الطابع اللغوي يغلب اكثر، حیث يتم تجاھل حالة الاحرف الـصغیرة      

Small   والكبیرة Capital لـذلك، لا تعتقـد ان حـرف    .  للحروف الإنجلیزيةa 
  . الكبیرZصغیر قد يتبع حرف ال

  :البحث في عناصر المصفوفة
 لتعود برقم فهرس العنصر في المصفوفة، ستفيدك هذه الطريقة كثيرا ()IndexOfاستخدم الطريقة 

ان رغبت في البحث عن عنصر من عناصر المصفوفة، وان لم تجد القيمة المرسلة في الوسيطة                
  :م ان الطريقة تتأثر بحالة الأحرف الكبيرة والصغيرة الإنجليزية مع العل1-الثانية، ستعود بالقيمة 

 
Dim X() As String = {"Turki", "Ali", "Ahmed"} 
 
ArabicConsole.WriteLine(Array.IndexOf(X, "Ahmed")) ' 2 
ArabicConsole.WriteLine(Array.IndexOf(X, "turki")) ' -1 

  
 بعلامـات الكـشيدة   -مع الأسـف – تتأثر ()IndexOfبالنسبة للحروف العربية، فالطريقة  

 ان لم تكن القيم متطابقة تماما من ناحيـة علامـات التـشكيل              1-والتشكيل، لذلك ستعود بالقيمة     
  :والكشيدة

 
Dim X() As String = { "خالد" ,"احمد " ,"تـركـي"} 
 
 
ArabicConsole.WriteLine(Array.IndexOf(X, "تركي")) ' -1 
ArabicConsole.WriteLine(Array.IndexOf(X, "احمد")) ' -1 
ArabicConsole.WriteLine(Array.IndexOf(X, "خـالـد")) ' -1 

  
 والذي يعيبه Linear Search تستخدم خوارزم البحث الخطي ()IndexOfاخيرا، الطريقة 

 Binaryالبطء الشديد خاصة مع المصفوفات الكبيرة، لذلك أنصحك بشدة استخدام البحث الثنائي 

Search           خصوصا مع المصفوفات كبيرة الحجم عن طريق استدعاء الطريقة BinarySearch() 
  : إن كانت المصفوفة مرتبةإلاوالتي لن تعود بقيمة صحيحة 

 
Dim X() As String = {"Turki", "Ali", "Ahmed"} 
 
 
ArabicConsole.WriteLine(Array.BinarySearch(X, "Ahmed"))  ' -1 
من ترتیب عناصر المصفوفة لتطبیقلابد  '  
 البحث الثنائي علیها '
Array.Sort(X) 
ArabicConsole.WriteLine(Array.BinarySearch(X, "Ahmed"))  ' 0 

  System.Collections أسماء مجال

 على عشرات الفئات الخاصة باحتواء البيانات System.Collectionsيحتوي مجال الأسماء 
 مرنة بما فيه Arrayصحيح ان الفئات من النوع . Collectionsلمجموعات االمختلفة تسمى 

الكفاية لاحتواء البيانات، إلا ان استخدام الفئات المشمولة في مجال الاسماء 
System.Collections يعتبر إجراء أكثر حرفية، لما توفره لك هذه الفئات من طرق وخصائص 

  .المعقدة دون الاعتماد عليهالن تستطيع تطبيق الخوارزميات البرمجية 
 Interfacesيمكنك تعلم كل فئة من هذه الفئات على حدة، ولكني اعتقد أن إتقان الواجهات               

  .التي تشملها هذه الفئات سيسهل عليك فهمها، كما أنه يختصر الوقت
  

  IList و  ICollectionالواجھات 
 وهي ICollectionعلى الواجهة  تحتوي System.Collectionsجميع الفئات في مجال الاسماء 

مما يعني ) التي لمحت اليها في الفصل السابق (IEnumerableواجهة مشتقة وراثياً من الواجهة 
  . على هذه الفئاتFor Eachإمكانية تطبيق الحلقة 

 التي تعـود بعـدد العناصـر فـي          Count على الخاصية    ICollectionتحتوي الواجهة   
 التي تمكنك من نسخ البيانات إلى مصفوفة ()CopyToلى الطريقة المجموعة، كما تحتوي أيضا ع

Array.  
. IDictionary و   IList همـا    ICollectionتوجد واجهتان مشتقتان وراثيا من الواجهة       

 لإضـافة   ()Add فقط والتي تحتوي على مجموعة من الطرق منهـا           IListسأذكر هنا الواجهة    
 لحذف جميع ()Clear لحذف عنصر، ()Removeر،  لتعديل قيمة عنص()Insertعناصر جديدة، 

  .العناصر وغيرها من الطرق
  

    ملاحظة
  

بعــض الفئــات تقــوم بتــضمین الواجھــات الــسابقة ولكــن لا تظھــر كافــة   
تحتـوي علـى    ) Arrayوالمنشئة من الفئـة     (أعضائھا، فمثلا المصفوفات    

و  ()Addكــالطرق ( كــل أعــضاء الواجھــة ر ولكنھــا لا تظھــIListالواجھــة 
Remove().(  

سأقوم الآن بعـرض سـريع ومختـصر لـثلاث فئـات موجـودة فـي مجـال اسـماء            
System.Collections       انصحك بشدة بالإبحار في مكتبة ،MSDN     للحصول على شرح لجميع 

الطرق والخصائص المدعومة بها والفئات الأخرى من نفس مجال الأسماء، وتذكر انه يمكن لهذه              
كما تستطيع أيضا تطوير فئـات      . ى جميع الأنواع المختلفة من البيانات     المجموعات ان تحتوي عل   

  .خاصة بك وذلك بتضمين الواجهات السابق ذكرها فيها
  

  Stackالفئة 
حيث يكون أول عنصر ) Last-In-First-Out) LIFO من تطبيق خوارزم Stackتمكنك الفئة 

 لإضافة العنصر ()Pushطريقة استخدم ال. يضاف إلى هذه المجموعة هو اخر عنصر يخرج منها
  : لقراءة العنصر()Popوالطريقة 

 
Dim st As New Stack(100) 
 
st.Push(10) 
st.Push(20) 
st.Push(30) 
 
ArabicConsole.WriteLine(st.Pop) ' 30 
ArabicConsole.WriteLine(st.Pop) ' 20 
ArabicConsole.WriteLine(st.Pop) ' 10 

  

  
  .stبيانات الكائن  شكل توضيحي ل:3-6شكل 

  Queueالفئة 

حيث يكون أول عنصر ) First-In-First-Out) FIFO من تطبيق خوارزم Queueتمكنك الفئة 
 لإضافة ()Enqueueاستخدم الطريقة . يضاف إلى هذه المجموعة هو أول عنصر يخرج منها

  : لقراءة العنصر() Dequeueالعنصر والطريقة 
 
Dim qu As New Queue(100) 
 
qu.Enqueue(10) 
qu.Enqueue(20) 
qu.Enqueue(30) 
 
ArabicConsole.WriteLine(qu.Dequeue) ' 10 
ArabicConsole.WriteLine(qu.Dequeue) ' 20 
ArabicConsole.WriteLine(qu.Dequeue) ' 30 

  

  
  .qu شكل توضيحي لبيانات الكائن :4-6شكل 

  
  ArrayListالفئة 

، يمكن اعتبار الفئات من النوع IList على الواجهة ArrayListتحتوي الفئات من النوع 
ArrayList الفئات من النوع ( كالمصفوفات التقليديةArray( ولكنك سرعان ما ستكتشف بعض ،

  .المزايا الإضافية في هذه الفئات بعد قراءتك للسطور التالية

جموعة او سـيكون    عند إنشاء كائن من هذه الفئة، يفضل تحديد العدد الابتدائي لعناصر الم           
 عنصر بشكل افتراضي مع العلم انك تستطيع التحكم في عدد العناصر في أي وقت بإسناد قيمة 16

  :Capacityإلى الخاصية 
 
Dim X As New ArrayList() ' 16عنصر بشكل افتراضي  
Dim Y As New ArrayList(100) ' 100 101 عنصر ولیس  
 

  :()Insertتخدام الطريقة يمكنك البدء في تعيين قيم العناصر باس
 
' X  للمجموعةالأولالعنصر  
X.Insert (0, "تركي") 
' Y  للمجموعةالاخيرالعنصر  
Y.Insert (99. "Visual Basic .NET") 

  
  :()Addاو إضافة المزيد من العناصر إلى نهاية المجموعة باستخدام الطريقة 

 
X.Add ("غادة") 
X.Add ("عباس") 
X.Add ("برعي") 

  
المحـددة  (عن عدد العناصر الإجمالية ) ()Addباستخدام الطريقة (زاد عدد العناصر المضافة اذا  

سيتم مضاعفة عدد عناصر المجموعة بشكل تلقـائي إلـى   ) لحظة تعريف وإنشاء كائن المجموعة    
  . عنصرY 200 عنصرا والمجموعة 32 ستصبح Xالضعف، أي ان المجموعة 

ــة   ــوي الفئ ــضا، تحت ــد أي ــريقتين ArrayListالمزي ــى الط  و ()AddRange عل
RemoveRange()     واللتان تمكنانك من إضافة مجموعة إلى نفس المجموعة، او حذف مجموعة ،

  :عناصر من نفس المجموعة، الإجراء التالي سيدمج كلا المصفوفتين في واحدة
 
Function JoinTwoArrays(ByVal arr1 As ArrayList, _  
    ByVal arr2 As ArrayList) As ArrayList 
 
    JoinTwoArrays = New ArrayList(arr1.Count + arr2.Count) 
 
    JoinTwoArrays.AddRange(arr1) 
    JoinTwoArrays.AddRange(arr2) 
End Function 

  
 او الطريقـة    ()Removeأخيرا، يمكنك حذف احد عناصر المجموعة باستخدام الطريقـة          

Clear()     فالمجموعة (لعناصر بخطوة واحدة   ان أردت حذف جميع اArrayList   تحتـوي علـى 
  ):IListالواجهة 

X.Remove ("عباس") 
Y.Clear () 

  
  
  
  
  
  

يستحيل علي ذكر كل التفاصيل وشرح الخصائص والطرق للفئات السابق ذكرها، ولكن كل مـا               
ر عمـل   حاولت تقديمه في هذا الفصل هو مدخل إلى مجموعة من الفئات الأساسية التابعة لإطـا              

.NET Frameworkحاول إلقاء نظرة إلى مراجع .  والتي لا يكاد يخلو أي برنامج منها.NET 

Documentation         والآن دعنا  .  للحصول على كل التفاصيل الصغيرة والكبيرة حول هذه الفئات
 وننتقل إلـى موضـوع يتعلـق    NET Framework.نستريح قليلا من عرض فئات إطار عمل 

  .باكتشاف الأخطاء

  اكتشاف الأخطاء

  
 هي الحل Debuggingإن كانت عملية التنقيح : "قرأت مرة في احد مجلات البرمجة الجملة التالية

  ."لزوال الأخطاء، فان البرمجة هي سبب حدوث تلك الأخطاء
 وطريقة اكتشافها وتـداركها     الأخطاء البرمجية   حول هذا الفصل  معك في حديثي   سيتمحور

 قـسم  كما سأخصص NET Framework. عمل إطارن الفئات المقدمة من باستخدام مجموعة م
  .Visual Studio .NET التنقيح التي توفرها لك بيئة بأدواتتعلق يكامل 

  
  فكرة عامة

أخطاء وقت التصميم، أخطاء وقت التنفيذ، : تصنف الأخطاء في لغات البرمجة إلى ثلاثة أقسام هي    
  :وفيما يلي ذكرها. والشوائب

  
  وقت التصمیم أخطاء
 ـ (Design Time errors وقت التصميم أخطاء  Syntax  النحويـة بالأخطـاء  أحيانـا سمى ت

Errors (  المصدرية،  الشفرات للصيغة الصحيحة في كتابة      إتباعك م تحدث نتيجة لعد   أخطاءهي 
  .Next بـ إغلاقها بدون For كتابة حلقة أو لم يتم تعريفه، أو غير موجود لإجراءاستدعائك ك

 الـشفرات  اكتشافا وتنقيحا، حيث تظهر لك نافذة محـرر  أسهلها هو الأخطاء النوع من   هذا
Code Editor   خطا متعرجا عند مكان حدوث الخطأ، وبمجرد تحريك مؤشر الفأرة عنـد ذلـك 

  .)1-7شكل (موضحة سبب الخطأ  Tool tip التلميح أداةالخطأ ستظهر لك 
  

  
  .المتعرج لتوضيح خطأ وقت التصميم ظهور أداة التلميح عند الخط :1-7شكل 

 Visual Studio .NETإن حاولت ترجمة أو تنفيذ البرنامج، ستظهر لك بيئـة التطـوير   
  .)2-7شكل (موضحة بوجود خطأ أو مجموعة اخطأ " There were Build Errors"رسالة 

  

  
  . اكتشاف خطأ أو مجموعة اخطأ نحوية لحظة الترجمة:2-7شكل 

  
من ) لا توجد بها أخطاء   (، سيتم تنفيذ آخر نسخة صحيحة       Yesلى الزر   إن قمت بالضغط ع   

 سيوقف عملية الترجمة، وستظهر قائمة بكافة الأخطاء النحوية         Noالبرنامج، أما نقرك على الزر      
 View->Other والتـي تـصل إليهــا باختيـار الأمـر      )3-7شكل ( Task Listفي نافذة 

Windows->Task List.  
  

  
  .مة بكافة الأخطاء النحوية للمشروع الحالي قائ:3-7شكل 

  
الحل الذي يتبعه المبرمجون لمواجهة هذه الأخطاء بسيط جدا ولا يحتاج إلى فلسفة، فكل ما               

  .هو مطلوب منك الانتقال إلى السطر الذي وقع فيها الخطأ وتصحيحه
  

   وقت التنفیذأخطاء
 Run عن أخطاء وقـت التنفيـذ   ، انت المسئول الأول والأخيرVisual Basic .NETكمبرمج 

Time Errors     وهي عبارة عن أخطاء تظهر للمستخدم أثناء تنفيـذ البرنـامج وتوقـف عمـل ،
  .Exceptions بالاستثناءات خطاء تسمى هذه الأNET.في عالم . البرنامج

 في اغلب الأحوال هو عدم توقع المبرمج بالتغيرات Exceptionsسبب حدوث الاستثناءات   
 سترى  الملفات والمجلدات بيئة المنصة التي يعمل بها البرنامج، فمثلا في الفصل القادم           الخارجية ل 

 التابعـة للفئـة     ()Moveكيف يمكنك نقل ملف في الجهاز من مسار إلى اخر باستخدام الطريقة             
System.IO.File:  

 
System.IO.File.Move("D:\File.EXE", "C:\File.EXE") 

  
 عنـدك،  D:\File.EXEح في جهازك الشخصي وذلك لوجود الملف     قد يعمل السطر السابق بنجا    

في –ولكن ان قمت بتنفيذه في جهاز اخر لا يحتوي على نسخة من هذا الملف او نفذته مرة اخرى        
 Visual Studio .NET بعد عملية نقل الملف، ستظهر لك بيئـة التطـوير   -جهازك الشخصي

  . غير موجودD:\File.EXEن الملف  تفيدك با)4-7شكل ( Exceptionرسالة استثناء 
  

  
  .Exception رسالة الاستثناء :4-7شكل 

  
 مـن اخـذ   Visual Basic .NET عليك كمبـرمج  -بصيغة اخرى–كما أخبرتك سابقا 

( الاستثناء دائما وأبدا في عين الاعتبار، ففي المثال السابق يفترض التحقق مـن وجـود الملـف          
  :القيام بعملية النقلقبل  ) ()Existsباستخدام الطريقة 

 
If System.IO.File.Exists("D:\File.EXE") Then 
    System.IO.File.Move("D:\File.EXE", "C:\File.EXE") 
End If 

  
لا تعتقد انك تستطيع الالتفاف حول الاستثناءات بهذه السهولة، حيث ان الأمر اعقد مـن ان         

 موجود وعلى قيد الحياة، D:\File.EXEالسابق يتم حله بجملة شرط واحدة، تخيل مثلا ان الملف       
، لذلك لن تتمكن من نقله وستظهر رسالة الاسـتثناء  Read Onlyولكن عليه خاصية القراءة فقط 

  :()GetAttribtuesمرة اخرى ان لم تتحقق أيضا من قابلية الكتابة باستخدام الطريقة 

Imports System.IO 
 
If Not CBool(File.GetAttributes("C:\Test\program.exe") And _ 
 FileAttributes.ReadOnly ) 
    If System.IO.File.Exists("D:\File.EXE") Then 
        System.IO.File.Move("D:\File.EXE", "C:\File.EXE") 
    End If 
End If 

 
 Read Onlyليس هذا فقط، بل افترض انه موجود ولا توجد عليه خاصية القـراءة فقـط   

ولكنه مفتوح من قبل برنامج اخر وعليه خاصية الإقفال، او ان يكون المسار الهدف غيـر قابـل             
او ان المسار الهدف لا توجد به مساحة كافية لنقل الملـف             ،) مثلا CD-ROMكأقراص  (للكتابة  
 وغيرها الكثير من الاحتمالات التي لا يمكنك تداركها بالاعتماد علـى الجمـل            ... او... اليه، او 

  .الشرطية
 تمكنك من تدارك الاستثناءات، الطريقة الاولى Visual Basic .NETتوجد طريقتين في 

، والتان سأفصل فيهمـا فـي       Err والطريقة الثانية باستخدام الكائن      Exceptionباستخدام الكائن   
  .الفقرات القادمة من هذا الفصل

  
  الشوائب

رمجة، فانه يستحيل عليك كتابـة برنـامج لا       حتى لو وصلت إلى مرحلة ما بعد الاحتراف في الب         
 هي أخطاء غير متوقعة ينـتج عنهـا   -في لغات البرمجة- الشوائب  .Bugsشوائب  يحتوي على   

تصرف غير متوقع او منطقي للبرنامج، فمثلا الاجراء التالي يقوم بإنشاء عداد يمكنك من وقـف                
  : ثواني الانتظارتنفيذ شيفرات البرنامج لفترة من الوقت تحددها بارسال عدد

 
Sub Sleep(ByVal numOfSeconds As Integer) 
    Dim startTime As Double 
 
    startTime = Timer 
    Do 
        System.Windows.Forms.Application.DoEvents() 
    Loop Until Timer - startTime >= numOfSeconds 
End Sub 

  
 صحيحة في جهازك وفي جهاز غيرك دون اي مـشاكل،           قد يعمل الاجراء السابق بطريقة    
 برمجي خطير، حيث ان الإجـراء الـسابق يعتمـد    Bugولكن يوجد في الخوارزم السابق شائب     

اعتماد كلي على ساعة الجهاز، وان قمت بارسال عدد معين من الثواني بحيث يزيد ويتعدى عـن      
 إلى مالا نهاية، مما يضطر المستخدم إلى        سيتم تنفيذ الحلقة  ) 23:59:59الساعة  (وقت نهاية اليوم    

التي تصل اليهـا بالمفـاتيح    (Windows Task Managerايقاف عمل البرنامج باستخدام نافذة 
[Alt] + [Ctrl] + [Del] .(يمكنك تصحيح هذا الشائب بتعديل شيفرة الإجراء:  

 
Sub Sleep(ByVal numOfSeconds As Integer) 
    Const NUMOFSEC As Double = 24 * 60 * 60 
    Dim startTime As Double 
 
    startTime = Timer 
    Do 
        System.Windows.Forms.Application.DoEvents() 
    Loop Until CBool((Timer + NUMOFSEC - startTime) Mod _  
    NUMOFSEC >= numOfSeconds) 
End Sub 

  
 الشائب في الشيفرة السابقة، ولكن في البرامج الجدية يصبح صحيح اننا قد تمكنا من اكتشاف

ويؤسفني إخبارك بانه لا توجد طريقة او اسلوب متبع         . الوضع اكثر صعوبة وتعقيدا مما هو عليه      
 التي تستخدم للتقليل Debuggingلقمع الشوائب من برامجك، ولكن توجد العديد من ادوات التنقيح 

ستجدها في القسم الاخير مـن هـذا      ) تقليل الشوائب وليس تفاديها   لاحظ اني ذكرت    (من الشوائب   
  .Visual Studio .NETادوات التنقيح من بيئة الفصل 

  

 Exceptionالكائن 
، وفي الحقيقة   )الأخطاء( لرمي وتفادي الاستثناءات     Exception تعتمد الكائن    NET.جميع لغات   

مل مع الاستثناءات في جميع لغات البرمجة   هذا الاسلوب لتوحيد طريقة التعا     Microsoftاعتمدت  
  NET Framework.الموجه إلى إطار عمل 

 الاسم  . هو كائن يتم انشائه لحظة وقوع خطأ في شيفرة البرنامج          Exceptionتقنيا، الكائن   
 وهو كائن يتم انشائه بمجرد وقوع استثنــاء فـي          System.Exceptionالكامل لهذا الكائن هو     

  .البرنامج
الاحوال لن تستخدم هذا الكائن مباشرة، حيث توجد عشرات الكائنـات الاخـرى        في معظم   

، DivideByZeroExceptionوالتي تكون خاصة لاستثناءات معينة، فنجد مـثلا الاسـتثناءات           
OverFlowException   و NotFiniteNumberException      والخاصة بالاعـداد والعمليـات 

، FileNotFoundExceptionالرياضــــية، ونجــــد ايــــضا الاســــتثناءات    
DirectoryNotFoundException  ،EndOfStreamException...     الخ والخاصة بعمليـات

 NET.وغيرها الكثير من الاستثناءات التي يمكنك البحث عنها في مستندات . التعامل مع الملفات

Documentation.  
 مـن الفئـة   ضع في عين الاعتبار ان جميع الاستثناءات السابق ذكرهـا مـشتقة وراثيـا           

System.Exception .   لذلك جميع طرق وخصائص الفئةException ستكون موجودة ايضا في 
  .الفئات الاخرى

 على مجموعة من الخصائص والطرق، وسأكتفي بـذكر اهـم          Exceptionيحتوي الكائن   
 وتعودان بقيم ReadOnly، كلا الخاصيتين للقراءة فقط StackTrace و Messageخاصتين هما 

 Attempted to": مـثلا (  الاولى تمثل وصف نصي للاسـتثناء  -Stringمن النوع –ة حرفي

divide by zero." (  والثانية تعود بقيمة تمثل الإجراءات التي رمت الاستثناء والإجراءات التـي
  .تداركت الاستثناء

، اما Throw an exceptionترمي استثناء الشيفرة التي سببت في حدوث خطأ يقال انها 
. Catch the exception الاسـتثناء تتفـادى  لشيفرة التي تتفادى او تتدارك الخطأ فيقال عنها ا

  .سأبدأ بعرض طريقة تفادى الاستثناءات اولا ومن ثم رميها
  

  Catching Exceptionsتفادي الاستثناءات 
ام بدلا من كتابة عشرات الجمل الشرطية في كل سطر تتوقع حدوث استثناء فيه، يمكنـك اسـتخد             

اضف السطور المتوقع حدوث الاستثناء بها اسفل الكلمة . Try … Catch … End Tryالتركيب 
، ويمكنك كتابة الشيفرات التي ترغب في تنفيذها عند وقوع الاستثناء اسـفل كـل          Tryالمحجوزة  

  :Catchعبارة 
 
Try 
    ' 
 الشیفرة المتوقع حدوث استثناء بها '    
    ' 
Catch 
    ' 
لشیفرة التي ستنفذ ان وقع الاستثناءا '      
    ' 
End Try 

  
 عند كتابة شيفرة ردة الفعل لحظة وقـوع الاسـتثناء    Exceptionتطبيقيا، ستستخدم الكائن    

  :Catchحتى تميز نوع الاستثناء، يمكنك عمل ذلك مع إضافة اسم للمتغير بعد الكلمة 

Dim X, Y As Integer 
… 
… 
Try 
ءاتاستثنااحداث  '      
    X = 10 \ Y 
    X = CInt(10 ^ Y) 
Catch ex As Exception 
 كتابة ردة الفعل عند '    
ستثناءاتوقوع الا '      
    If ex.Message = "Attempted to divide by zero." Then 
 
        ArabicConsole.WriteLine("خطأ القسمة على الصفر") 
 
    ElseIf ex.Message = "Arithmetic operation resulted " &  
        _ & "in an overflow." Then 
 
        ArabicConsole.WriteLine(" لعدد خارج مجال المتغيرخروج اخطأ في  ") 
 
    Else 
 
        ArabicConsole.WriteLine("خطأ غير معروف") 
 
    End If 
End Try 

  
 لتحديد نوع الاستثناء طريقـة غيـر        -ال السابق كما في المث  –الاعتماد على القيمة النصية     

 لمعرفة اسماء كافة الفئات MSDNراجع مكتبة (محبذة، اذ يفضل استخدام نوع الفئة بكتابة اسمها  
  ):الاخرى

 
Dim X, Y As Integer 
… 
… 
Try 
    X = 10 \ Y 
    X = CInt(10 ^ Y) 
Catch ex As DivideByZeroException 
    ArabicConsole.WriteLine(" لقسمة على الصفراستثناء ا ") 
Catch ex As OverflowException 
    ArabicConsole.WriteLine(" في قیمة العدد خارج مجال المتغيراستثناء  ") 
Catch ex As Exception 
    ArabicConsole.WriteLine(" غير معروفاستثناء  ") 
End Try 

 Catch، سيتم التحقق من نوع الاستثناء بدئا من اول عبارة عند وقوع الاستثناء في الشيفرة السابقة
 Exceptionحتى النهاية، وان لم تتوافق الفئة مع الاستثناء فسيتم الانتقال إلى الاستثناء الاخيـر               

  . يمثل اي استثناء تم رميه مهما اختلف نوعهExceptionبشكل تلقائي، حيث ان الكائن 
 فـي اي وقـت   Try … Catch … End Tryركيب اخيرا، يمكنك الخروج من داخل الت

  .Exit Tryبكتابة العبارة 
  

    ملاحظة
  

ــتثناءات   ــات اســــــــ ــد كائنــــــــ  و StackOverflowExceptionتوجــــــــ
OutOfMemoryException         لا يوجد وقت أستطیع تحديده لـك لوقوعھـا . 

ان قمت بتفادي الاستثناءات السابقة، فانـصحك القیـام بإنھـاء البرنـامج          
حیث ان الوضع لن يكـون مـستقر ومناسـب        ) Catchلعبارة  اسفل ا (فورا  

  .لمواصلة تنفیذ البرنامج عند وقوع ھذه الاستثناءات
  

  
  :Whenاستخدام 

 ان أردت إضافة شـرط اضـافي   Catch برفقة العبارة Whenيمكنك استخدام الكلمة المحجوزة  
  :لعملية قنص الاستثناءات، فبدلا من كتابة جمل الشرط التالية

 
Dim x As Integer 
… 
… 
Try 
    … 
    … 
Catch ex As Exception 
    If x = 0 Then 
        … 
    ElseIf x = 1 Then 
        … 
    Else 
        … 
    End If 
End Try 

  : لكتابة اجمل وتصميم ارقى في شيفراتك المصدريةWhenتستطيع استخدام 
 
Dim x As Integer 
… 
… 
Try 
    … 
    … 
Catch ex As Exception When x = 0 
    … 
Catch ex As Exception When x = 1 
    … 
Catch ex As Exception 
    … 
End Try 

  
على سبيل المثال لا – في حالات معينة، منها    Whenيمكنك الاستفادة من الكلمة المحجوزة      

  : تحديد الخطوة او المرحلة التي وقع فيها الاستثناء-الحصر
 
Dim StepNum As Integer 
… 
… 
Try 
    StepNum = 1 
    … 
    … 
    StepNum = 2 
    … 
    … 
    StepNum = 3 
    … 
    … 
    StepNum = 4 
    … 
    … 
 
Catch ex As Exception When StepNum = 1 
    … 
Catch ex As Exception When StepNum = 2 
    … 
Catch ex As Exception When StepNum = 3 
    … 
… 
… 
End Try 

  :Finallyاستخدام 
 ان اردت Try … Catch … End Try في التركيب Finallyيمكنك استخدام الكلمة المحجوزة 

 هو تنفيذ الشيفرة -في هذا السياق–تنفيذ مجموعة من الشيفرات المصدرية دائما، اعني بكلمة دائما 
  .في حالة وقوع الاستثناء او عدم وقوعه

 
Try 
    ' 
  حدوث استثناء بهاالشیفرة المتوقع '    
    ' 
Catch 
    ' 
 الشیفرة التي ستنفذ ان وقع الاستثناء '    
    ' 
Finally 
    ' 
 سیتم تنفیذ هذه الشیفرة دائما '    
    ' 
End Try 

 
 سيتم تنفيذها دائما حتى لـو  Finallyاعيد واكرر، الشيفرة التي تقع اسفل الكلمة المحجوزة  

 و اي عبارات خروج Try … Catch … End Try داخل التركيب Exit Tryاستخدمت العبارة 
  ).Exit Sub ،Exit Function ،Returnكـ (من الإجراء 

  
    ملاحظة

  
ــستخدم الكلمــة المحجــوزة   ــذ   Finallyت ــوي تنفی ــدما تن ــا ذكــرت عن  كم

الشیفرة سواء وقع استثناء او لـم يقـع، مـن أمثلـة ھـذه الحـالات قطـع            
  . او اغلاق الملفاتالاتصالات بقواعد البیانات

  

  
  Throwing Exceptionsرمي الاستثناءات 

تحدثت في الفقرة السابقة عن طريقة قنص وتدارك الاستثناءات، وفي هذه الفقرة سـنقوم بتطبيـق         
يمكنك رمي الاستثناء في اي سطر مـن سـطور          . العكس اي نرمي الاستثناءات لنسبب الأخطاء     

  :نشاء كائن الاستثناء مع اThrowالبرنامج باستخدام الامر 
 
 رمي استثناء '
Throw New System.IO.FileNotFoundException() 

 لكائن الاستثناء لحظة الرمي باسـتخدام  Messageيمكنك تحديد نص الرسالة في الخاصية      
  :Throwالامر 

Throw New Exception("خطأ في عملیة التحدیث") 
  

 بنفسك ما دامت معظم فئات إطـار عمـل         قد تستغرب مدى الجدوى من رمي الاستثناءات      
.NET Framework ترمي استثناءات بمجرد حدوث أخطاء، ولكنك في حالات كثيرة قد تحتاج 

 من ارسال قـيم غيـر       -مثلا–إلى رمي استثناءات الأخطاء يدويا بنفسك لتحمي إجراءات فئاتك          
  .يباكما سترى قر–صحيحة، او ان رغبتك في رمي استثناءات خاصة ببرنامجك 

رمـي  (اعود إلى موضوع قنص وتدارك الاستثناءات ولكـن فـي سـياق هـذه الفقـرة                 
، حيث اود ان اخبرك ان الاستثناء لا يختفي ولا يموت ان وقع حتى يقابل اول عملية                )الاستثناءات

، وحتى تفهم ماذا اقصد من هذه الكلمات، Try … Catch … End Tryتفاديه باستخدام التركيب 
  : التاليةراقب الشيفرة

 
Sub AAA() 
    BBB() 
End Sub 
 
Sub BBB() 
    CCC() 
End Sub 
 
Sub CCC() 
    Throw New Exception() 
End Sub 

  .سيعود الاستثناء إلى الإجراء المستدعي حتى يجد من يتداركه: 5-7شكل 

  
لخـروج مـن    قمت برمي استثناء جديد، وبمجرد وقوع هذا الاستثناء سيتم ا          ()CCCفي الإجراء   

 والبحث عن شيفرة تفادي الاستثناء، وان لم تكتـب          ()BBBالإجراء والعودة إلى الإجراء السابق      
-7شكل (...  وان لم توجد فالعودة إلى الإجراء السابق وهكذا ()AAAسيتم الانتقال إلى الإجراء    

  .وان لم تتم عملية تدارك الاستثناء، فسيتم ظهور رسالة الخطأ) 5
 ويمكنك معرفة مسار الإجراءات     ()AAAتدارك الاستثناء في الإجراء الاول      جرب القيام ب  

  :كما لمحت اليها سابقا- StackTraceالتي يشملها الاستثناء عن طريق الخاصية 
 
Sub AAA() 
    Try 
        BBB() 
    Catch ex As Exception 
        ArabicConsole.WriteLine(ex.StackTrace) 
    End Try 
End Sub 
 
Sub BBB() 
    CCC() 
End Sub 
 
Sub CCC() 
    Throw New Exception() 
End Sub 

  :مخرجات الشيفرة السابقة ستكون شبيهه بما يلي
 
at MyNameSpace.Module1.CCC() in D:\VB.NET\Module1.vb:line 31 
at MyNameSpace.Module1.BBB() in D:\VB.NET\Module1.vb:line 28 
at MyNameSpace.Module1.AAA() in D:\VB.NET\Module1.vb:line 21 

  
اخيرا، عملية رمي الاستثناءات تستهلك الكثير من مصادر النظام فلا تحاول اعتبارها كنوع       

 ان لم يـتم     Falseمن الحلويات وتكثر شيفراتك المصدرية منها، فمثلا اجعل الإجراء يعود بقيمة            
  . التكلفة الزائدة وجعله يرمي استثناء كاملتنفيذه بالشكل الصحيح بدلا من

  
  Custom Exceptionsإنشاء فئات استثناءات خاصة 

 دائما، الا انك قد NET Framework.ينصح دائما برمي فئات الاستثناء المعرفة في إطار عمل 
ـ تحتاج إلى انشاء وتعريف استثناءات خاصة بك يوما من الايام، يمكنك من تعريف فئة استثناء ب               

Visual Basic .NET بــسهولة شــديدة عــن طريــق تعريــف فئــة تــشتق مــن الفئــة 
System.ApplicationException:  

 
Class UnableToLoadUserFileException 
    Inherits System.ApplicationException 
 
    … 
    … 
End Class 

  
الفئـة القاعديـة     خـصائص وطـرق      Overridesليس هذا فقط، بل يمكنك ايضا إعادة قيـادة          

Exception)  وهي التي تم اشتقاقSystem.ApplicationExceptionوراثيا منها :(  
 
Class UnableToLoadUserFileException 
    Inherits System.ApplicationException 
 
 
    ' Message اعادة قیادة الخاصیة 
    Public Overrides ReadOnly Property Message() As String 
        Get 
            Return "لم اتمكن من تحمیل ملف المستخدم" 
        End Get 
    End Property 
End Class 

 او رميـه  Try … Catch … End Try باسـتخدام  ءوالان يمكنك تفادي هذا الاسـتثنا 
  :كما تفعل مع باقي الاستثناءات الاخرى– في اي وقت Throwباستخدام 

 
Try 
    … 
    Throw New UnableToLoadUserFileException() 
    … 
Catch ex As UnableToLoadUserFileException 
    … 
    ArabicConsole.WriteLine(ex.Message) 
    … 
End Try 
 

    ملاحظة
  

 Customان كنــت ترغــب فــي رمــي كائنــات اســتثناءاتك الخاصــة       
Exceptions   مـع الحــالي لدقـة اكثــر  او المج– خـارج المـشروع الحــالي- 

الفصل التاسع . Serializableعلیك جعل فئة الاستثناء قابلة للتسلسل       
  . يتحدث بشكل مفصل عن ھذا الموضوعSerializationالتسلسل 

  
 

  
 Errالكائن 

 Err الاخرى بوجود الكـائن  NET. عن سائر لغات Visual Basic .NETتتميز لغة البرمجة 
 هـذا الكـائن     Microsoft وفي حقيقة الامر، وفرت      . الاستثناءات الذي يمكنك من تفادي ورمي    

 نسخة جديدة مـن  Visual Basic .NET اعتقادا منهم ان Visual Basic 6خصيصا لمبرمجي 
Visual Basic 6->1)   لقد وضحت لك وجهة نظري الشخصية حول هذا الموضوع سابقا فـي

  ).مقدمة هذا الكتاب ولا يوجد داعي لتكرار ما ذكرته
، يمكنك تفادي الاستثناءات ورميها ايضا مع الكائن        Exceptionوكما هو الحال مع الكائن      

Errولكن بصيغة مختلفة .  
  

  Catching Exceptionsتفادي الاستثناءات 
 Try … Catch … End من تفادي الاستثناءات بصيغة مختلفة عن التركيب Errيمكنك الكائن 

Try وهي استخدام العبارة ،On Errorالاولى تحدد فيها موقـع يـتم   :  والتي يكون لها صورتين
  :الانتقال اليه بمجرد حدوث استثناء في الشيفرة المصدرية

Sub MySub() 
    On Error GoTo Errorhandler 
    … 
    … 
 الشیفرات المتوقع حدوث استثناء بها '    
    … 
    … 
 لا تنسى كتابة العبارة التالیة حتى '    
 لا یتم تنفیذ السطور التالیة دائما '    
    Exit Sub 
 
Errorhandler: 
  هذا القسم انإلىسیتم الانتقال  '    
 وقع استثناء '    
    … 
    … 
End Sub 

  
والثانية تطلب الاستمرار من المترجم حتى ان صادف عشرات الاستثناءات في الشيفرة المصدرية         

  :ليتجاهلها وكأن شيئا لم يحدث
 
Sub MySub() 
    On Error Resume Next 
    … 
    … 
 الشیفرات المتوقع حدوث استثناء بها '    
    … 
    … 
End Sub 
 

 والـذي يحتـوي علـى       Errاستخدامك لأحد العبارات السابقة سينقل الاستثناء إلى الكائن         
 والتي تماثـل  Descriptionمجموعة من الطرق والخصائص المتعلقة بهذا الاستثناء، كالخاصية        

  :Exception في الكائن Messageالخاصية 
 
Sub MySub() 
    On Error GoTo Errorhandler 
    … 
    … 
    … 
جراءمن الجید الخروج من الإ '      
 حتى لا یتم تنفیذ الشیفرة بالاسفل '    
    Exit Sub 
 
Errorhandler: 
    ArabicConsole.WriteLine(Err.Description) 
End Sub 

    ملاحظة
  

 … Try … Catch و … On Errorلا يمكنك تفادي الاستثناءات بالصیغتین 
End Tryفي نفس الإجراء .  

  
  

  Throwing Exceptionsرمي الاستثناءات 
 يتم بسهولة شديدة، حيث كل ما هو مطلوب منك اسـتدعاء  Errرمي الاستثناءات باستخدام الكائن     

  :ل رقم لاستثناء وإرسا()Raiseالطريقة 
 
Err.Raise(11) 

  
 MSDN، راجع مكتبة    DivideByZeroException السابق يؤدي إلى رمي الاستثناء       11الرقم  

  : لرمي استثناءات خاصة بك1001لمعرفة ارقام الاستثناءات الاخرى، او يمكنك استخدام الرقم 
 
Err.Raise(1001, , "لم اتمكن من تحمیل ملف المستخدم") 

  
  Err و Exceptionر بین الاختیا

تدارك ورمي الاستثناء بالصيغ السابقة يعود اولا واخيرا على نوع الإجراء الذي امتلأ بالشيفرات              
 يعطيك مرونة كبيـرة فـي   Errالتعامل مع الكائن . المصدرية، ولا استطيع نصحك بأيهما افضل   

 On Errorستخدام الصيغة تدارك الاستثناءات ولكنه يسبب بطء في الشيفرة المصدرية، حيث ان ا

وليس جزء معين كمـا     – في اعلى الإجراء يؤدي إلى تدارك جميع الشيفرات التابعة للإجراء            …
  .Try … End Tryتفعل مع الصيغة الاخرى 

 Onمن المهم جدا ان تضع في عين اعتبارك دائما ان تفادي الاستثناءات باستخدام الصيغة 

Error … خاص بلغة البرمجة Visual Basic .NET   وذلك يعني ان لغـات برمجـة ،.NET 
قد لا يعنيك الامر حاليا وذلك لانـك لا  . الاخرى لا تتبع هذا الاسلوب وقد يؤثر على سير البرامج     

تنوي تعلم لغة برمجة اخرى، ولكن لا تنسى ان فئاتك قد تستخدم مـن قبـل مبـرمجين اخـرين            
  .يستخدمون لغات اخرى

 … Try والصيغة السابقة … On Errorستثناءات بالصيغة الفرق الجوهري بين تفادي الا

End Tryهو ان في الاولى يتم قتل الاستثناء بمجرد الانتهاء من الإجراء، راقب هذا المثال ،:  

Sub AAA() 
    Try 
        BBB() 
    Catch ex As DivideByZeroException 
 لن یتم تنفیذ هذه الشیفرة '        
        ArabicConsole.WriteLine(ex.Message) 
    End Try 
End Sub 
 
Sub BBB() 
    On Error GoTo ErrorHandler 
    Dim X As Integer = 0 
 
 احداث استثناء '    
    X = 10 \ X 
 
    Exit Sub 
 
ErrorHandler: 
    … 
    … 
End Sub 

  
ه وطباعة مخرجاته على الشاشة      واردت تدارك  ()AAAقمت بتفادي الاستثناء في الإجراء      

 الا ان عملية قنص الاستثناء لم       ()BBBان وقع، ورغم ان الاستثناء وقع فعلا في الإجراء الاخر           
 لم تنفذ، ()AAA في الإجراء Catchتتم بالشكل المتوقع، حيث ان الشيفرة الموجودة اسفل العبارة 

  . قد مات بعد نهاية الإجراء()BBB وذلك لسبب بسيط وهو ان الاستثناء الذي وقع في الإجراء
ان – ()BBBالذي يتوجب علينا فعله في هذه الحالة، هو إعادة احياء الاستثناء في الإجراء              

يـتم ذلـك    .  وعمل الازم  ()AAA قبل نهاية الإجراء، وذلك ليتم إخبار الإجراء المستدعى          -وقع
  :()Err.Raise  الطريقة اوThrow مع الامر Err.GetExceptionبرمي الاستثناء 

 
Sub BBB() 
    On Error GoTo ErrorHandler 
    Dim X As Integer = 0 
 
    X = 10 \ X 
 
    Exit Sub 
 
ErrorHandler: 
    Throw Err.GetException 
End Sub 

 وذلك  ()Err.Raise عوضا عن الطريقة     Throwيفضل رمي الاستثناء في الحاله السابقة بالامر        
  .تعلم ما هو الاستثناء الذي قد يقع في الإجراءلانك قد لا 

  
    ملاحظة

  
 On Error Goto Xتلاحظ في المثال السابق أننـي اسـتخدمت الـصیغة    

، وذلك عنـدما اسـتخدمت   On Error Resume Nextولم استخدم أختھا 
الــصیغة الثانیــة لــم تــنجح معــي طريقــة رمــي الاســتثناء إلــى الإجــراء   

  .Callerالمستدعي 
  

  
 Visual Studio .NETت التنقیح من بیئة أدوا

لنبتعد قليلا عن كتابة الشيفرات المصدرية، ودعنا نبحر في جولة سريعة حـول بيئـة التطـوير                 
Visual Studio .NET على مجموعة من الأدوات التي تفيـدك فـي   -بشكل سريع– لنتعرف 
  .Visual Basic .NETعمليات تنقيح برامجك بـ 

  
  أسالیب التنفیذ

 تقليدي، يمكنك تنفيذ البرنامج كما فعلنا منذ الصفحة الاولى لهذا الكتاب وذلك بالضغط على               بشكل
مع العلم ان البرنامج سيعمل داخل بيئة . Debug من القائمة  Start او اختيار الامر     [F5]المفتاح  
امج كبرن–اما ان اردت تنفيذ البرنامج .  وما زال تحت سيطرتهاVisual Studio .NETالتطوير 

 السابقة Debug من القائمة Start without Debugging فاختر الامر -مستقل عن بيئة التطوير
، مع العلم ان خدمات التنقيح لن تكون مدعومـة لحظـة   [F5] + [Ctrl]او اضغط على المفاتيح 

  .التنفيذ
  

    ملاحظة
  

 سـیفقد بیئـة   Start without Debuggingتنفیـذ البرنـامج باختیـار الامـر     
تطوير السیطرة على البرنامج ولن تتمكن من إيقافـه مـن داخـل بیئـة           ال

–، اي )Debug مــن القائمــة Stop Debuggingباختیــار الامــر (التطــوير 
 سـیتم تنفیـذ البرنـامج بـشكل مـستقل كمـا تنفـذه مـن              -بعبارة اخـرى  

  .Windows Explorerمستكشف النظام 

مج في اي سطر تريده من نافذه محرر  في البرنا  Breakpointsنقاط الوقف   يمكنك تحديد   
الشيفرات المصدرية، وذلك بالضغط بزر الفأرة الايمن على السطر المطلـوب واختيـار الامـر         

Insert Breakpoint شكل ( من القائمة المنبثقة، لترى ان السطر قد تغير لونه إلى اللون الاحمر
7-6(.  

  

  
  .مصدرية في الشيفرة الBreakpoints نقاط الوقف :6-7شكل 

  
يمكنك مشاهدة جميع نقاط الوقف التي وضعتها في شيفراتك المصدرية عن طريق النافـذة              

Breakpoints )  والتـي تـصل اليهــا باختيـار الامــر    )7-7شـكل Debug->Windows-
>Breakpoints  

  

  
  .Breakpoints نافذة نقاط الوقف :7-7شكل 

  
نفيذ البرنامج خطوة خطوة، وفي كـل   حيث يتم تStep Intoلديك اسلوب اخر للتنفيذ يسمى 

 من القائمة السابقة، سيتم تنفيـذ  Step Into او تختار الامر [F11]مرة تضغط فيها على المفتاح 
يفيدك هذا الاسلوب كثيرا ان اردت تتـابع  . )8-7شكل (سطر واحد فقط من الشيفرات المصدرية    

 الموجود Step Overبالنسبة للامر و. الأخطاء ومعرفة مصادرها بين ثنايا الشيفرات المصدرية

 السابق، ولكـن التنفيـذ سيـشمل كامـل     Step Into، فهو شبيه بالامر Debugفي نفس القائمة 
  .الإجراء
  

  
  .Step Into التنفيذ سطرا سطرا :8-7شكل 

  
  

    ملاحظة
  

ــسابقة    ــر ال ــل الاوام ــن تعم ــت ) Step Over و Step Into(ل الا اذا كان
 13-1شـكل   ( Release ولـیس    Debug ھي   Configurationالاعدادات  

  .سأعود إلى ھذه الاعدادات قريبا جدا. )32صفحة 
  

  
  نوافذ أخرى

 Pauseفي هذه الفقرة اعرض لك ثلاث نوافذ لن تتمكن من رؤيتها الا لحظة الايقـاف المؤقـت                  
 ـ ). Step Into أو التنفيذ Breakpointsوالذي قد يكون بسبب نقاط وقف (للبرنامج  ذه تفيـدك ه

 تعـرض لـك قـيم       Localفالنافـذة   . النوافذ في معرفة قيم المتغيرات وطابور تنفيذ الإجراءات       
 تعرض لك قيم المتغيرات Autos للإجراء الحالي، والنافذة Local Variablesالمتغيرات المحلية 

للاحقـة،  التابعة للكائن الذي يتم تنفيذه في السطر الحالي والكائنات في السطور الثلاث السابقة وا             
يمكنك عرض هذه النوافذ .  تعرض لك طابور الإجراءات التي يتم تنفيذهاCall Stackبينما النافذة 

  .)9-7شكل ( Debug->Windowsلحظة الإيقاف المؤقت باختيار الاوامر المناسبة من القائمة 

  .Call Stack، و Locals ،Autos النوافذ :9-7شكل 

  
مكنك ادارة جميع الاستثناءات التي تقع في البرنـامج او جميـع         بعيدا عن النوافذ السابقة، ي    

 Exceptions الاخرى عن طريق صندوق الحـوار  NET Framework.استثناءات إطار عمل 
  .)10-7شكل ( Debug من القائمة Exceptionsوالذي تصل اليه باختيار الامر 

  .Exceptions صندوق حوار الاستثناءات :10-7شكل 

  
  Configurationsت الاعدادا

 وذكـرت انهـا     Configurationsفي الفقرة الاخيرة من الفصل الاول لمحت إلى الاعـدادات           
في كل مـشروع جديـد     .  للبرنامج Compilingمجموعة من الخيارات الخاصة بعملية الترجمة       

، تـستطيع  Release و Debugتنشئه، تقوم بيئة التطوير بارفاق اسمين مـن الاعـدادات همـا             
  .)32 صفحة 13-1شكل (تيار بينهما من القائمة الموجودة في شريط الادوات الرئيسي الاخ

، يمكنك اضافة حذف تعديل اعدادات خاصة       Release و   Debugبالاضافة إلى الاعدادات    
 Build من القائمة Configuration Managerبك بالاسماء التي تريدها، يتم ذلك باختيار الامر 

  .)11-7شكل (

  .Configuration Manager صندوق حوار مدير الاعدادات :11-7شكل 

  
بعد انشاء اعداد جديد او تعديل الحالي، يمكنك تغيير مواصفات الاعداد وتكوينات الترجمة             

 والانتقال إلى خانـة  Project Property Pagesعن طريق صندوق حوار خصائص المشروع 
ع تحديد الاعداد الحالي في اعلـى صـندوق   ، مConfiguration Propertiesالتبويب اليسرى 

  .) بالصفحة المقابلة12-7شكل (الحوار 

  . تعديل الاعدادات يدويا من صندوق حوار خصائص المشروع:12-7شكل 

  
  
  
  
  

 من الآن وصاعدا عليك التأقلم NET.كان هذا كل ما وددت ذكره حول الأخطاء البرمجية في عالم 
 حتى لا تسبب لك إرباكا عند قراءتك لمستندات         Exceptionsات  الاستثناءعلى المصطلح الجديد    

.NET Documentations . لنغلق ونترك موضوع الاستثناءات جانبا ونتحدث عن موضوع آخر
  .عنوان الفصل الظاهر في بأعلى الصفحة المقابلة الملفات والمجلداتوهو 

  الملفات والمجلدات

  
 علـى  NET Framework.كتبة فئات إطـار عمـل    من مSystem.IOيحتوي مجال الاسماء 

مجموعة من الفئات التي تمكنك من ادارة وتحرير ملفات ومجلدات الجهاز، بدءاً مـن العمليـات              
  .البسيطة كالنسخ، النقل، او الحذف، وانتهاءا بتحرير محتويات الملفات برمجيا

جلدات واستخدام الفئـة    القسم الأول يتعلق بالم   : قسمت هذا الفصل القصير الى اربعة اقسام      
Directory       والثاني بالملفات والتعامل مع الفئة ،File        والثالث يشرح طريقة تحريـر الملفـات ،

، Path، اما القسم الاخير فيتطرق الى ثلاث فئـات هـي      Streamsومقدمة الى وحدات التخزين     
DirectoryInfo و ،FileInfoبشكل سريع .  

  
    ملاحظة

  
مجیــة، ســأفترض فــي ھــذا الفــصل انــك قمــت لاختــصار الــشیفرات البر

  :باستیراد مجال الأسماء التالي
Imports System.IO 

  

  
 Directoryالفئة 

 تمكنك من تنفيذ Shared Methods على مجموعة من الطرق المشتركة Directoryتحتوي الفئة 
 -واضحمن ال – والتي   ()CreateDirectory، كالطريقة   Foldersوظائف متعددة على المجلدات     

  :انها تنشئ مجلد جديد
 
Directory.CreateDirectory("C:\Test\My Folder") 
Directory.CreateDirectory("C:\لد الرئیسيا") 

    ملاحظة
  

قبل إجـراء اي عملیـة علـى الملفـات والمجلـدات، ينـصح بـشدة تـدارك          
  :استثناء دائما

 
Try 
    … 
    Directory.CreateDirectory ("C:\Test") 
    … 
Catch 
    … 
    … 
End Try 

  
  

  : تحذف مجلد موجود()Deleteعلى العكس من إنشاء المجلدات، فالطريقة 
 
Directory.Delete("C:\لد الرئیسيا") 

  
حيث سيظهر اسـتثناء ان     (ضع في اعتبارك ان السطر السابق لن يعمل الا مع المجلدات الخالية             

، وان اردت حذف المجلد وكافة المجلدات والملفات        )مجلدات فرعية احتوى المجلد على ملفات او      
  : مع الوسيطة الثانية للطريقة السابقةTrueالتابعة له، ارسل القيمة 

 
 حذف الد بما يحتویه '
Directory.Delete("C:\Test", True) 

  
    ملاحظة

  
ع  بشكل نھائي، فلا تتوق ـ-بما يحتويه– تحذف المجلد   ()Deleteالطريقة  

  . والخاصة بنظام التشغیلRecycle Binارساله الى سلة المھملات 
  

  
 الى مسار اخر سواء علـى       -بما يحتويه – ان اردت نقل المجلد      ()Moveاستخدم الطريقة   

  :نفس القرص او إلى قرص اخر
 
Directory.Move("C:\Test", "C:\لد الرئیسيا\Test") 

غيير اسم المجلد دون نقله، وذلك شريطة استخدام نفـس     ايضا لت  ()Moveيمكنك استخدام الطريقة    
  :المسار

Directory.Move("C:\Test", "C:\TestEx") 
  

منعا للشوائب او (يفضل دائما التحقق من وجود المجلدات قبل اجراء العمليات السابقة عليها   
 ان وجد المجلد True والتي تعود بالقيمة ()Exists، يتم ذلك باستدعاء الطريقة )وقوع الاستثناءات

  :المرسل
 
If Directory.Exists("C:\Test") = True Then 
    … 
    … 
End If 

  
 تعود بوقت وتاريخ عملية انشاء المجلد، بينمـا الطريقـة        ()GetCreationTimeالطريقة  

GetLastAccessTime()              تعود بوقت وتاريخ اخر عملية وصول الى المجلـد، امـا الطريقـة 
GetLastWriteTime()مع العلم ان جميع الطرق السابقة .  فتعود بوقت وتاريخ اخر عملية كتابة

  :Dateتعود بقيمة من النوع 
 
ArabicConsole.WriteLine(Directory.GetCreationTime("C:\Test")) 
ArabicConsole.WriteLine(Directory.GetLastAccessTime("C:\Test")) 
ArabicConsole.WriteLine(Directory.GetLastWriteTime("C:\Test")) 

  
ــرق   ــتخدام الط ــد باس ــسابقة للمجل ــيم ال ــديل الق ، ()SetCreationTimeيمكنــك تع

SetLastAccessTime() و ،SetLastWriteTime():  
 
ArabicConsole.WriteLine(Directory.GetCreationTime("C:\Test")) 
Directory.SetCreationTime("C:\Test", Date.Now) 
ArabicConsole.WriteLine(Directory.GetCreationTime("C:\Test")) 

  
  طرق تعود بمسارات

مـن  ( تعودان بقيمة حرفيـة      ()GetDirectoryRoot و   ()GetCurrentDirectoryالطريقتان  
  :، الاولى تمثل الدليل الحالي، والثانية الدليل الجذري للمسار المرسل)Stringالنوع 

 
' C:\Test 
ArabicConsole.WriteLine(Directory.GetCurrentDirectory)   
' C:\ 
ArabicConsole.WriteLine(Directory.GetDirectoryRoot("C:\Test")) 

، تستطيع تغيير المسار الحالي باستخدام الطريقة       ()GetCurrentDirectoryوعلى ذكر الطريقة    
SetCurrentDirectory():  

 
Directory.SetCurrentDirectory("C:\Test") 

  
ــرق   ــا الطــ ، و ()GetLogicalDrives() ،GetDirectories() ،GetFilesامــ

GetFileSystemEntries()       فهي تعود بمصفوفة من النوع String   اسماء -على التوالي – تمثل 
محركات الاقراص الموجودة في الجهاز، المجلدات الموجودة في المـسار المرسـل، الملفـات               

  :والمجلدات والملفات الموجودة في المسار المرسلالموجودة في المسار المرسل، 
 
Dim x As String 
 
For Each x In Directory.GetLogicalDrives 
    ArabicConsole.WriteLine(x) 
Next 
 
… 
… 
 
Dim x() As String = Directory.GetFileSystemEntries("C:\Test", _  
    "*.exe") 
 
Dim counter As Integer 
 
For counter = 0 To UBound(x) 
    ArabicConsole.WriteLine(x(counter)) 
Next 

  
 تعود بالدليل الأبوي للمسار المرسل، نوع القيمة التي تعود          ()GetParentاخيرا، الطريقة   
 من هذا فئات اخرى، سأتطرق لهذه الفئة في القسم الاخير   DirectoryInfoبها هي كائن من الفئة      

  .الفصل
  

  الملفات والمجلداتالبحث عن 
باستخدام الطرق السابق ذكرها، يمكنك الاستعلام والبحث عن كافة اسماء وخـصائص الملفـات              

 التالي ليطبع كافـة اسـماء     ()PrintFilesAndFoldersطورت الإجراء   . الموجودة في القرص  
يـة  الاسـتدعاءات التراجع  الملفات والمجلدات الموجودة في المسار المرسل، استخدمت اسـلوب          

Recursion      يمكنك .  في هذا الاجراء ليشمل السرد الملفات الموجودة في المجلدات الفرعية ايضا
  :تطوير هذا الاجراء ايضا ليعرض لك خصائص ومواصفات الملفات

Sub PrintFilesAndFolders(ByVal path As String, _ 
    Optional ByVal subDirectories As Boolean = False) 
 
    Dim folderName As String 
    Dim fileName As String 
 
 
    For Each folderName In Directory.GetDirectories(path) 
        ArabicConsole.WriteLine(folderName) 
        If subDirectories Then 
            PrintFilesAndFolders(folderName, True) 
        End If 
    Next 
 
    For Each fileName In Directory.GetFiles(path) 
        ArabicConsole.WriteLine(fileName) 
    Next 
End Sub 
يمكنك استدعاء الاجراء السابق للبحث عن الملفات والمجلدات في المسار المرسل، كما تـستطيع              

  : ان اردت البحث عن كافة محتويات المجلدات الفرعيةTrueارسال القيمة 
 
' C:\Test لدعرض محتویات ا 
PrintFilesAndFolders("C:\Test") 
 
' C:\Test لدعرض محتویات ا 
لفرعیةوالدات ا '  
PrintFilesAndFolders("C:\Test", True) 

  
  Fileالفئة 

 على العديد من الطرق التي تتعامل مع الملفات بشكل مباشر، وقبل ان اعرض              Fileتحتوي الفئة   
 Directoryلك هذه الطرق، اود ان انوه هنا الى أن بعض الطرق الموجودة في الفئـة الـسابقة                  

  :ن استخدامها يتم بنفس الطريقة كما أFileمدعومة ايضا في الفئة 
 
 حذف ملف '
File.Delete ("C:\Test\File.EXE") 
 
  ملفنقل '
File.Move("C:\Pic.BMP", "D:\Pic.BMP") 
 
 الحصول على وقت تاریخ انشاء الملف، واخر وصول، واخر تعدیل '
x = File.GetCreationTime("C:\MyFile.DAT") 
x = File.GetLastAccessTime("C:\MyFile.DAT") 
x = File.GetLastWriteTime("C:\MyFile.DAT") 

 تعدیل القیم السابقة '
File.SetCreationTime("C:\MyFile.DAT", Now) 
File.SetLastAccessTime("C:\MyFile.DAT", Now) 
File.SetLastWriteTime("C:\MyFile.DAT", Now) 
 
 التحقق من وجود الملف '
If File.Exists("C:\letter.doc") Then 
    … 
    … 
End If 

  
  : ان اردت نسخ ملف()Copyاستخدم الطريقة 

 
File.Copy("C:\program.EXE", "C:\program2.EXE") 

  
ستظهر رسالة خطأ وقت التنفيذ ان وجد اسم الملف الهدف، لذلك قد تحتاج الى الكتابة فوق الملف                 

  : مع الطريقةTrue وذلك بارسال القيمة -ان وجد–
 
 سیتم الكتابة فوق الملف الهدف '
File.Copy("C:\program.EXE", "C:\program2.EXE", True) 

  
ــة   ــات، فيمكنــك اســتدعاء الطريق ــى مواصــفات الملف امــا ان اردت الحــصول عل

GetAttributes() والتي تعود بقيمة تركيب من نوع Enum هي FileAttributes:  
 
Dim fileAttr As FileAttributes 
 
fileAttr = File.GetAttributes("C:\Test\program.exe") 
 
 هل الملف ارشیف '
If CBool(fileAttr And FileAttributes.Archive) Then 
    ArabicConsole.WriteLine("ارشیف") 
End If 
 
للقراءة فقطهل الملف  '  
If CBool(fileAttr And FileAttributes.ReadOnly) Then 
    ArabicConsole.WriteLine("قراءة فقط") 
End If 
 
مخفيهل الملف  '  
If CBool(fileAttr And FileAttributes.Hidden) Then 
    ArabicConsole.WriteLine("مخفي") 
End If 
… 
… 
… 

 -السابقة– تمكنك من تعديل خصائص الملف ()SetAttributesوعلى العكس، فان الطريقة 
  :بنفسك

 
 اجعل الملف للقراءة فقط '
File.SetAttributes("C:\program.exe", FileAttributes.ReadOnly)  
 
 اجعل الملف للقراءة فقط ومخفي '
File.SetAttributes("C:\program2.exe", FileAttributes.ReadOnly _ 
    Or FileAttributes.Hidden) 
 
 اضف خاصیة الاخفاء لخصائص الملف '
File.SetAttributes("C:\program3.exe", _  
    File.GetAttributes("C:\program3.exe") Or FileAttributes.Hidden) 
 
 الغ خاصیة الاخفاء من خصائص الملف '
File.SetAttributes("C:\program4.exe", _ 
    File.GetAttributes("C:\program4.exe") Xor FileAttributes.Hidden) 

  
    ملاحظة

  
ــان  ــدات  ()SetAttributes و ()GetAttributesالطريقت  تعمــلان مــع المجل

Folders              ايضا، ولكنك لن تستطیع الوصـول الیھمـا الا عـن طريـق الفئـة 
File.  

  

  
  Streamالفئة 

قد . Storage Mediumوحدة تخزين  مجموعة من البايتات تكتب وتقرأ من Streamتمثل الفئة 
، Memory Stream من الذاكرة ، مقطعFileتكون وحدة التخزين هذه ملف في القرص الصلب 

، او شـبكة اتـصال    )Parallel Portكمنفـذ متـوازي    (Virtual Deviceجهاز افتراضـي  
Network Stream.  

، لـذلك لـن   Abstract Classفئة مجردة  هي Streamضع في اعتبارك دائماً ان الفئة 
فئات اخرى مـشتقة منهـا   تقوم في العادة بانشاء الكائنات منها مباشرة، وانما ستنشئ كائنات من            

 مع العلم اننـا     NetworkStream، و   FileStream  ،MemoryStream كالفئات   )1-8شكل  (
  . فقط في هذا الفصلFileStreamسنتعامل مع الفئة 

  . العلاقة الوراثية بين فئات وحدات التخزين:1-8شكل 

  
    انظر ايضا

  
 فـي الفـصل   Abstract Classesقد لمحت سـابقا الـى الفئـات المجـردة     

  .الوراثةالرابع 
  

  
  الخصائص والطرق المشتركة

 Streamتوجد مجموعة من الخصائص والطرق المشتركة بين الفئات المشتقة وراثيا من الفئـة              
وسأبدأ معك بعرض الاعضاء المشتركة، ومن ثم اخصص        . باختلاف وحدات التخزين التي تمثلها    

  .FileStreamفقرة لاعضاء الفئة 
 التخزين تمكنك من الكتابة، القراءة، وتحريك مؤشر الكتابة والقراءة، مع ذلك      معظم وحدات 

 تسمح لك بالقراءة والكتابة دون امكانية تحريك -NetworkStreamكالفئة –توجد وحدات تخزين 
على اجراء هذه العمليات بالاستعلام عنها      ) وحدة التخزين (يمكنك معرفة مدى قدرة الفئة      . المؤشر

 ان كانـت    True والتي تعود بالقيمـة      CanSeek، و   CanRead  ،CanWrite في الخصائص 
  .العملية مدعومة

 عند كل عملية قراءة او كتابـة، فعنـد          Bufferingبعض الفئات تتبع اسلوب يسمى بالـ       
مثلا، فان عملية التحرير لا تتم علـى       ) FileStreamالفئة  (اجراء عمليات التحرير على الملفات      

 خاص بها ولـن   Bufferعد تنفيذ الامر، وانما سيتم الاحتفاظ ببيانات الملف في          المـلف مباشرة ب  

 لاجــراء   ()Flushتتم عملية التحديث الا بعد قيامك باغلاق الملـف، او باسـتدعاء الطريقـة               
  .)2-8شكل (التحديـثات على الملف 

  

  
  لملف او أغلق الملف لتحديث البيانات الى ا()Flush استدعي الطريقة :2-8شكل 

  
، لذلك فمعظـم    Stream مشتقة وراثيا من الفئة      -كما ذكرت –جميع فئات وحدات التخزين     

 التي تعود  Length ستجدها ايضا في الفئات المشتقة، كالخاصية        Streamخصائص وطرق الفئة    
 التي تحدد موقع مؤشر الكتابة والقراءة في Positionبحجم البيانات في وحدة التخزين، والخاصية 

  . التخزينوحدة
 التي تعود بمصفوفة ()Readاما الحديث عن الطرق، فتوجد مجموعة من الطرق كالطريقة       

 التي تقرأ بايـت ()ReadByte تمثل البايتات التي تم قراءتها، والطريقة Byte Arrayمن النوع 
 التي تمكنك من كتابة مجموعة مـن البايتـات فـي           ()Writeوعلى العكس، الطريقة    . واحد فقط 

 التي تكتب بايـت  ()WriteByte الى وحدة التخزين، والطريقة Byte Arrayفوفة من النوع مص
  .واحد فقط
  

    ملاحظة
  

عملیة القراءة والكتابة من والى وحدة التخزين تـؤدي الـى تغییـر موقـع          
 بـشكل تلقـائي، بحیـث       Positionمؤشر الكتابة والقـراءة فـي الخاصـیة         

  .ات التي تم قراءتھا او كتابتھاعدد البايت+ الموقع الحالي : يكون

 تغلق وحدة التخزين وتزيل كافة البيانات المتعلقة بها مـن           ()Closeالمزيد ايضا، الطريقة    
بـشكل  – سيستدعي ايضا ()Closeالذاكرة، وضع في الاعتبار ان اغلاق وحدة التخزين بالطريقة     

كالفئـة  -Buffering ب الــ   ان كانت وحدة التخزين تعتمـد اسـلو  ()Flush الطريقة -تلقائي
FileStream.  

تتم علـى   ) الخ ... ()Read()  ،Writeباستخدام الطرق السابقة    (عمليات القراءة والكتابة    
، ولكنك قد لا تتبع هذا الاسلوب فـي  Byte Arrayشكل بايتات موزعة في مصفوفات من النوع 

 ـ         رى التي تسهل عليـك عمليـة       اغلب تطبيقاتك، وقد تلجأ الى استخدام مجموعة من الفئات الاخ
كقراءة او كتابة سطر من نص، قراءة او كتابة مجموعة من البايتات تمثل نـوع       (القراءة والكتابة   

، يمكنـك عمـل ذلـك باسـتخدام الفئـات           )الـخ  ...Integer  ،Doubleمعين من البيانات كـ     
StreamReader     ،StreamWriter     ،BinaryReader     ،BinaryWriter  ،

XmlTextReader و ،XmlTextWriter . وهذه الفئات ليست سوى واجهة للمبرمج تستخدم في
  ).3-8شكل ( للوصول الى الملفات FileStreamبنيتها التحتية كائن 

  

  
  .FileStream أساليب متعددة لاستخدام كائن الملفات :3-8شكل 

 في كل مرة بانشاء بعد قيامك!): ما أكثر النقاط الهامة والأخيرة في كتابي(نقطة هامة اخيرة 
 عليك القيام باغلاقه باسـتدعاء الطريقـة        -مهما كانت وحدة التخزين والفئة التي تمثله      –اي كائن   
Close()دائما قبل حدوث الموت المنطقي له .  
  

    انظر ايضا
  

، StreamReader ،StreamWriterسنتعامل في ھذا الفصل مـع الفئـات      
BinaryReader و ،BinaryWriter .تخدم الفئتـــــــــین وقـــــــــد نـــــــــس

XmlTextReader و XMLTextWriter لاحقــا فــي الفــصل التاســع عــشر 
 المزيد ايضا، توجـد فئـات اخـرى لـم        .XMLربط البیانات والتكامل مع     

استخدمھا في ھـذا الكتـاب يمكنـك البحـث عـن تفاصـیلھا فـي مكتبـة                  
MSDN.  

  

  
  التعامل مع الملفات النصیة

 وتنـشئ  StreamWriter، فانك ستستخدم الفئة Textual Filesعند التعامل مع الملفات النصية 
كائن منها لانشاء ملفات نصية والكتابة فيها، لديك العديد من الطرق لانشاء الكائن وفـتح الملـف        

  :النصي والكتابة عليه، وهذه اسهلها
 
Dim textFile As New StreamWriter("C:\Test.TXT") 

  
كون موقع مؤشر الكتابه في بداية الملف، اما ان اردت فتح           عند وجود الملف سيتم الكتابة فوقه وي      

  : الى مشيد الفئة السابقةTrue، فارسل القيمة Appendالملف للاضافة 
 
Dim textFile As New StreamWriter("C:\Test.TXT", True) 

  
  :Unicode او ASCIIالمزيد ايضا، تستطيع تحديد نوع صفحة المحارف اما 

 
' ASCII 
Dim textFile As New StreamWriter("C:\Test.TXT", True, _  
 System.Text.Encoding.ASCII) 
 
' UNICODE 
Dim textFile2 As New StreamWriter("C:\Test2.TXT", True, _  
 System.Text.Encoding.Unicode) 

غبة  لوضع اعدادات اضافية عند عملية الفتح، كتحديد ر        FileStreamيمكنك استخدام الفئة    
انشاء ملف جديد او فتح نفس الملف، تحديد خاصية القراءة والكتابة، وتحديد خاصـية المـشاركة          

Sharing)  راجع مكتبةMSDNلمزيد من التفاصيل :(  
 
Dim textStream As FileStream = File.Open("C:\Test.TXT",_  
    FileMode.CreateNew, _ 
    FileAccess.ReadWrite, _  
    FileShare.Read) 
 
Dim textFile As New StreamWriter(textStream) 

  
 بالكتابة وارسال البيانات المطلوبـة اليـه، اسـتخدم    -على الارجح–بعد فتح الملف، ستبدأ  

 ستحول جميع ()Write للكتابة الى الملف، الطريقة  ()WriteLine او الطريقة    ()Writeالطريقة  
 لا تعمل الا مع القيمة الحرفية من        ()WriteLine، والطريقة   Stringالبيانات الى النوع الحرفي     

  : في نهاية السطرNewLineسطر جديد  حيث تضيف حرف Stringالنوع 
 
Dim textFile As New StreamWriter("C:\test.TXT") 
… 
… 
textFile.WriteLine("البیانات المحفوظة") 
textFile.Write(99.9) 
… 
… 
textFile.Close() 

  
    ظةملاح
  

 ولـیس  StreamWriter الـسابقة تابعـة للفئـة       ()Writeتذكر ان الطريقـة     
 تتعامـل   FileStreamحیث ان الطريقة التابعـة للفئـة        . FileStreamالفئة  

  . فقطByte Arrayمع المصفوفات من النوع 
  

  
 كما ذكرت في الفقـرات الـسابقة،   Bufferingعملية الكتابة الى الملفات تتبع اسلوب الـ   

 وستتم عملية   AutoFlush الى الخاصية    Trueكنك الغاء هذا الاسلوب ان اردت باسناد القيمة         يم
 فتمنع عملية التحديث المباشـر  Falseالتحديث مباشرة بعد كل عملية كتابة الى الملف، اما القيمة       

و ا) ()Closeباستدعاء الطريقـة    (وسيتم حفظ البيانات بشكل مؤقت في الذاكرة حتى تغلق الملف           
  .()Flushاجراء التحديث يدويا باستدعاء الطريقة 

على العكس من عملية الكتابة، يمكنك قراءة البيانات من الملفات النصية باسـتخدام الفئـة               
StreamReader . وكما هو الحال مع الفئةStreamWriter السابقة، لديك العديد من الطرق التي 

  :تمكنك من فتح ملف للقراءة، وهذه اسهلها
 
Dim textFile As New StreamWriter("C:\Test.TXT") 

 
الطريقة . بعد فتح الملف، فسيكون لديك عدة طرق تمكنك من قراءة البيانات بصور مختلفة            

ReadLine() تقرأ سطر كامل وتعود بقيمة حرفية من النوع String:  
 
Dim textFile As New StreamReader("C:\test.TXT") 
… 
ArabicConsole.WriteLine( textFile.ReadLine() ) 
… 
textFile.Close() 

  
 والتي تقرأ كامـل الملـف بخطـوة    ()ReadToEndان كان حجم الملف صغير، فلديك الطريقة    

  :واحدة
 
Dim textFile As New StreamReader("C:\test.TXT") 
… 
ArabicConsole.WriteLine(textFile.ReadLineToEnd) 
… 
textFile.Close() 

  
    حظةملا
  

ــسبة ل ــد  ()Readلطريقــة بالن ــم فق ــادةت ــا إع ــصور Overloads تعريفھ  ب
  . لمزيد من التفصیلاتMSDN بالرجوع الى مكتبة أنصحكمختلفة، لذلك 

  
  

 للتعامـل مـع     StreamReader و   StreamWriterفي الامثلة السابقة استخدمت الفئتين      
ول الى الطرق الاضافية لوحدة التخـزين       أود ان انوه هنا إلى أنك تستطيع الوص       . الملفات النصية 

 Length، هنا اسـتخدمت الخاصـية       BaseStreamعن طريق الخاصية    ) FileStreamالفئة  (
  :لمعرفة حجم الملف

Dim textFile As New StreamReader("C:\test.TXT") 
… 
ArabicConsole.WriteLine( textFile.BaseStream.Length ) 
… 
textFile.Close() 

  
  التعامل مع الملفات الثنائیة

 BinaryWriter، فيستحسن استخدام الفئات Binary Filesان اردت التعامل مع الملفات الثنائية 
مع ذلك، لن تستطيع فتح الملفات منها مباشرة بكتابة اسم الملف مع المشيد،             . BinaryReaderو  

  :الى مشيدات هذه الفئات وارساله FileStreamوانما عليك انشاء نسخة كائن من النوع 
 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.Create) 
Dim binaryFile As New BinaryWriter(st) 

  
 والتي تقبل معظم انـواع البيانـات        ()Writeتستطيع الكتابة الى الملف باستخدام الطريقة       

  ): الخ ...String ،Char ،Integer ،Doubleكـ (القياسية 
 
Dim st As FileStream = File.Open("C:\test.dat", _  
    FileMode.OpenOrCreate) 
 
Dim binaryFile As New BinaryWriter(st) 
 
' Integer قیمة من النوع 
binaryFile.Write(10) 
' Double قیمة من النوع 
binaryFile.Write(200.5) 
' Boolean قیمة من النوع 
binaryFile.Write(True) 
… 
… 
… 
binaryFile.Close() 
 

، والتي تحتوي على مجموعة مـن       BinaryReaderوعند القراءة، فستنشئ كائن من الفئة       
  : حيث تحدد نوع البيانات التي تود قراءتها()ReadXXXالطرق بالاسم 

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim binaryFile As New BinaryReader(st) 
 
' Integer قیمة من النوع 
ArabicConsole.WriteLine(binaryFile.ReadInt32()) 
' Double قیمة من النوع 
ArabicConsole.WriteLine(binaryFile.ReadDouble()) 
' Boolean قیمة من النوع 
ArabicConsole.WriteLine(binaryFile.ReadBoolean()) 

    ملاحظة
  

 والذي يتوافق مـع     ()ReadXXXان لم تحدد الاستدعاء الصحیح للطريقة       
  .النوع المناسب، فستعود الطريقة بقیمة غیر صحیحة

  
  

 فإن ذلك يتطلب بعض التفصيل، حيث ان ارسـال          Stringsوللتعامل مع البيانات الحرفية     
روف تمثل عدد الحروف     يؤدي إلى كتابة قيمة اضافية قبل الح       ()Writeقيمة حرفية الى الطريقة     

  :التي تضمنتها القيمة الحرفية المرسلة، فالعمليات التالية
 
Dim st As FileStream = File.Open("C:\test.dat", _ 
    FileMode.OpenOrCreate) 
Dim binaryFile As New BinaryWriter(st) 
 
binaryFile.Write("عباس السریع") 
binaryFile.Write(" ةبرعي ابو جبه ") 
binaryFile.Write("زكریا زعتر") 

 
تؤدي الى اضافة حجم الثابت الحرفي قبل الثابت نفسه في الملف، وذلك اشارة الى عدد الحروف                 

 بعين الاعتبار فـي كـل مـرة يـتم           ()ReadStringالتي يتضمنها الثابت حتى تأخذها الطريقة       
  :استدعائها

 
Dim st As FileStream = File.Open("C:\test.dat", _  
    FileMode.OpenOrCreate) 
Dim binaryFile As New BinaryReader(st) 
 
ArabicConsole.WriteLine(binaryFile.ReadString()) ' عباس السریع 
ArabicConsole.WriteLine(binaryFile.ReadString()) ' برعي ابو جبهة 
ArabicConsole.WriteLine(binaryFile.ReadString()) ' ریا زعترزك  

  
مع ذلك، انصحك بشدة اخذ الحيطة والحذر عند اتباع الاسلوب السابق عند التعامـل مـع                
البيانات الحرفية وذلك بسبب كثرة الاخطاء والشوائب البرمجية الناتجة عن نسيان تحريك موقـع              

 او ()Seekمؤشر القراءة والكتابة خاصة وان كنت تحـركه كثيرا يدويا بنفسك باستخدام الطريقة          
  .Positionالخاصية 

 الـى الطريقـة   Char Arrayلديك اسلوب اخر مفضل وهو ارسال مصفوفة من النـوع  
Write()    عند الكتابة الى الملف، واستدعاء الطريقة ReadChars() مع تحديد عدد الحروف التي 

  .تود قراءتها من الملف

    ملاحظة
  

حروف مـسبقة  بـال كإضافة لثقافتك البرمجیة، يسمى الاسـلوب الاول     
ــرف   Prefixed-length Stringsالحجــم  ــاني فیع ــا الاســلوب الث  ام

اتمنى منك التمییـز  . Fixed-length Stringsبالحروف ثابتة الحجم 
ــستندات      ــي مـ ــا فـ ــا يومـ ــصطلحات ان رأيتھـ ــذه المـ ــین ھـ  NET.بـ

Documentation.  
  

  
   خاصةCustom Streamsتكوين 

) الخ ...StreamReader  ،StreamWriter  ،BinaryReader(الفئات السابقة التي استخدمناها     
)  وغيرها Integer  ،String  ،Boolean  ،Double(تعمل بشكل رائع مع انواع البيانات القياسية        

  : التاليةPersonولكن ان حاولت استخدام انواع بيانات خاصة بك كالفئة 
 
Class Person 
    Public Name As String 
    Public Age As Integer 
End Class 

  
  :فعليك التعامل مع كل حقل على حدة عند القراءة والكتابة

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim binaryFile As New BinaryWriter(st) 
Dim Turki As New Person() 
 
Turki.Name = "تركي العسيري" 
Turki.Age = 99 
 
binaryFile.Write(Turki.Name) 
binaryFile.Write(Turki.Age) 
 

  
ولو كانت الفئة تحتوي على عشرات الحقول والخصائص، فمن غير العملي القيام بكتابة او قراءة               

  .كل حقل على حدة مع الملف
في مثل هذه الحالات، قد تطور فئات خاصة لتمكنها من التعامل مع انواع بيانـات اخـرى    

، حيث أن كل المطلوب منك هنـا     Inheritance تستطيع عمل ذلك بفضل الوراثة       .تعرفها بنفسك 
هو اشتقاق فئة من فئات الكتابة السابقة وتطبيقها مع الملفات، هذا مثـال اشـتققت فيـه الفئتـين                   

BinaryWriter و BinaryReader:  

 فئة الكتابة '
Class PersonWriter 
    Inherits BinaryWriter 
 
    Sub New(ByVal st As System.IO.Stream) 
        MyBase.New(st) 
    End Sub 
 
    Public Overloads Sub Write(ByVal personObject As Person) 
        MyBase.Write(personObject.Name) 
        MyBase.Write(personObject.Age) 
    End Sub 
End Class 
 
القراءةفئة  '  
Class PersonReader 
    Inherits BinaryReader 
 
    Sub New(ByVal st As System.IO.Stream) 
        MyBase.New(st) 
    End Sub 
 
    Function ReadPerson() As Person 
        Dim tmpPerson As New Person() 
 
        tmpPerson.Name = MyBase.ReadString 
        tmpPerson.Age = MyBase.ReadInt32 
 
        Return tmpPerson 
    End Function 
End Class 

  
  :وهنا تطبيق عملي يستخدم فئاتنا الجديدة، هذه شيفرة الكتابة الى الملف

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim MyFile As New PersonWriter(st) 
Dim Turki As New Person() 
 
Turki.Name = "تركي العسيري" 
Turki.Age = 99 
 
MyFile.Write(Turki) 
 
MyFile.Close() 

  
  :وهذه للقراءة من الملف

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim MyFile As New PersonReader(st) 
Dim Turki As Person 

Turki = MyFile.ReadPerson 
 
ArabicConsole.WriteLine(Turki.Name) ' تركي العسيري 
ArabicConsole.WriteLine(Turki.Age) ' 99 
 
MyFile.Close() 
 
 
 

    ملاحظة
  

 السابقة، علیك اضـافة     Personعند اضافة المزيد من الحقول في الفئة        
 والقـراءة فـي الفئـة    PersonWriterشیفرات الكتابة الاضافیة في الفئـة       

PersonReader . تسلسل الكائنات مع ذلك، سأريك في الفصل القادم
Object Serialization كیف يمكن للفئة ان تعلم بجمیع حقولھا دون 

الحاجــة لتحــديثھا واضــافة المزيــد مــن الــشیفرات المــصدرية والخاصــة   
  .بالكتابة او القراءة

  

  
  فئات أخرى

 ذكر ثلاث فئات ستسهل عليك الكثير من المهام المتعلقة بالملفـات     ، بودي قبل ان اختم هذا الفصل    
  .سنت استخدامها بذكاءوالمجلدات ان اح

  
  Pathالفئة 

 Paths على مجموعة من الطرق والخصائص التي تتعلق بحروف المسارات           Pathتحتوي الفئة   
خذ مثلا هذه الخصائص التي تعود بالحروف المستخدمة فـي          . المستخدمة في الملفات والمجلدات   

ية على انظمة تشغيل اخـرى      قد تختلف النتائج عند تطبيق الشيفرة التال      (تحديد فواصل المسارات    
  ):Windowsغير 

 
' / 
ArabicConsole.WriteLine(Path.AltDirectorySeparatorChar) 
' \ 
ArabicConsole.WriteLine(Path.DirectorySeparatorChar) 
' "<>| 
ArabicConsole.WriteLine(Path.InvalidPathChars) 
' ; 
ArabicConsole.WriteLine(Path.PathSeparator)    
' : 
ArabicConsole.WriteLine(Path.VolumeSeparatorChar) 

المزيد ايضا، هذه مجموعة اضافية من الطرق تمكنك من استخلاص جـزءاً معينـاً مـن                
  :المسار، كاسم الملف، المجلد، الامتداد وغيرها

 
Dim X As String = "C:\Test\File.EXE" 
 
' C:\Test 
ArabicConsole.WriteLine(Path.GetDirectoryName(X)) 
' File.EXE 
ArabicConsole.WriteLine(Path.GetFileName(X)) 
' .EXE 
ArabicConsole.WriteLine(Path.GetExtension(X)) 
' C:\ 
ArabicConsole.WriteLine(Path.GetPathRoot(X)) 
' True 
ArabicConsole.WriteLine(Path.HasExtension(X)) 
' File 
ArabicConsole.WriteLine(Path.GetFileNameWithoutExtension(X)) 

  
  FileInfo و DirectoryInfoالفئات 

 ملف معين، لانشاء كائنات مـن هـذه      FileInfo مجلد معين، والفئة     DirectoryInfoتمثل الفئة   
  :الفئات، ارسل قيمة حرفية تمثل مسار المجلد او الملف مع مشيد الفئة

 
 مجلد '
Dim folder As New DirectoryInfo("C:\Windows") 
 ملف '
Dim file As New FileInfo("C:\Autoexec.bat") 

  
 وبالتالي فإنهما تحويان مجموعة من FileSystemInfoكلا الفئتان مشتقتان وراثيا من الفئة 

، Name ،FullName ،Extension ،Exists ،Attributes: الطرق والخصائص المشتركة مثل
CreationTime ،LastWriteTime ،LastAccessTime ،Refresh() و ،Delete():  

 
' Windows 
ArabicConsole.WriteLine(folder.Name)   
' Autoexec.bat 
ArabicConsole.WriteLine(file.Name) 
 
' C:\Windows 
ArabicConsole.WriteLine(folder.FullName)  
' C:\Autoexec.bat 
ArabicConsole.WriteLine(file.FullName)  
… 
… 

 من الطـرق والخـصائص الخاصـة بهـا،           على مجموعة  DirectoryInfoتحتوي الفئة   
 التي تنشئ مجلد ()Create التي تعود بالمجلد الأبوي للمجلد الحالي، والطريقة Parentكالخاصية 

 التي ()GetDirectoriesهذا مثال يشرح استخدام الطريقة     . جديد، ومجموعة من الطرق الاخرى    
لفرعية التابعة للمجلد الـذي يمثلـه       تمثل المجلدات ا  ) DirectoryInfoمن النوع   (تعود بكائنات   

  :الكائن
 
Dim folder As New DirectoryInfo("C:\Windows") 
Dim subfolder As DirectoryInfo 
 
For Each subfolder In folder.GetDirectories 
    ArabicConsole.WriteLine(subfolder.Name) 
Next 

  
عة من الطـرق والخـصائص الخاصـة     فهي ايضا تحتوي على مجموFileInfoاما الفئة  

 التي تعود بحجم الملف، ومجموعة من الطرق الاضافية التـي         Lengthبطبيعة عملها كالخاصية    
هذا مثال اخر شبيه بالمثـال الـسابق، ولكنـه يـستدعي الطريقـة       .تستخدم لفتح وانشاء الملفات   

GetFiles())  التابعة للفئةDirectoryInfo (ع للعودة بكائنات من النوFileInfo:  
 
Dim folder As New DirectoryInfo("C:\Windows") 
Dim subfile As FileInfo 
 
For Each subfile In folder.GetFiles("*.EXE") 
    ArabicConsole.WriteLine(subfile.Name) 
Next 

  
  

    ملاحظة
  

يمكنك تطوير الشیفرتین السابقتین لتعرض أحجام ومواصـفات الملفـات    
ــا  ــات الحلقــة       وغیرھ ــتخدام خــصائص كائن ــات، وذلــك باس مــن معلوم

subfolder و subfile.  

في هذا الفصل تعرفنا على مجموعة من الفئات والخاصة بالتعامل مع الملفات والمجلدات، وقـد               
 وطريقة عملها، وفي الحقيقة كان لدي هدف اخر من          Streamsتطرقت فيه الى وحدات التخزين      

بتسلـسل   لتطبيق مـا يعـرف       Streamsيدك الى استخدام وحدات التخزين      هذا الفصل وهو تمه   
  . عنوان الفصل التاليObject Serializationالكائنات 

  التاسعالفصل  9

 

  
  

  
 Objectتسلسل الكائنات 

Serialization  
  

 مـن الكـائن، وكانـت    Instance طريقة انشاء نسخة الفئات والكائناتتعلمت في الفصل الثالث   
هذا الكائن محصورة فقط في مؤشر ذلك الكائن والمتمثلـة في متغير داخـل  العلاقة التي تربطك ب   

 اكثر والتحكم فيه، وكذلك Object Instanceيمكنك السيطرة على نسخة الكائن . شيفرة البرنامج
  .Serializationالتسلسل نقله من مكان الى اخر باستخدام 

، ودعني انوه Object Serializationهذا الفصل هو مدخلك المبدئي الى تسلسل الكائنات 
 NET.هنا إلى أن موضـوع التسلسل من المواضيع المتقدمـة جدا فـي برمجة اطـار عمـل   

Framework            وبالتالي سأفترض انك مبرمج متمكن جدا، حيث ان هذا الموضوع موجـه الـى ،
  .المبرمجين الجادين

  
    ملاحظة

  
  :تالیةمعظم فئات ھذا الفصل تجدھا في مجالات الاسماء ال

 
Imports System.Runtime.Serialization 
Imports System.Runtime.Serialization.Formatters.Binary 

  
لذلك لا تغفل عن استیرادھا في ملفات مـشروعك، كــما سأستخـــدم     
ايضـــاً مجموعة من فئات الفصل السابق والموجودة في مجـال الاسماء 

  :التالي
 
Imports System.IO 

  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

 لك إلى تسلسل الكائناتمدخ
سأبدأ معك هذا الفصل بهذا القسم الذي يعتبر مدخلا الى التسلسل، حيث سـنبدأ بتعريـف عمليـة        

 ثم نقوم بتطبيقه بشكل عملي وذلك بعرض مثال بسيط يـستخدم مفهـوم             NET.التسلسل في عالم    
 عرض لكيفية  التسلسل لحفظ الكائن في ملف على القرص الصلب واستعادته مرة أخرى، يلي ذلك          

 Attributesتطبيق التسلسل على الفئات التي تعرفها بنفسك، وذلك باستخدام بعض المواصـفات             
  .التي تجعل الفئة قابلة او غير قابلة للتسلسل

  
  ما ھو التسلسل؟

نـسخة  نظراً للارتباط الوثيق بين التسلسل والكائنات فاسمح لي أن اذكرك بالمقصود من عبـارة           
نسخة الكائن هي تلك المنطقة من الذاكرة التي تحمل بيانات كائن تم . Object Instanceالكائن 

انشائه من فئة، وهي لا تمثل متغيرات الكائن، اذ ان نسخة الكائن قد يشير لها اكثر من متغير، ففي 
  :الشيفرة التالية

 
Dim Turki As New Person 
Dim Ali As New Person 
Dim X As Person 
… 
… 
X = Turki 

  
 رغم وجود ثـلاث مؤشـرات، حيـث ان    Two object instancesأنشأت نسختين من الكائن 

 Both Turki and X pointers point to يشيران الى نفس نسخة الكائن X و Turkiالمؤشرين 

the same object instance)   معذرة على كتابتها باللغة الإنجليزية ولكن يهمني جدا اسـتيعابها
  .)1-9شكل ()  الامر تعلم اللغة الصينيةحتى لو كلفني

  

  
  .Person ثلاث مؤشرات تشير الى نسختين كائن من الفئة :1-9شكل 

 فـي  Object Instance هي عملية حفظ بيانات نسخة الكـائن  Serializationالتسلسل 
قد تكون وحدة التخزين هذه ملف في القرص الصلب، مقطع من الذاكرة،             (Streamوحدة تخزين   

، ويمكنك لاحقا اعادة احياء الكائن وارفاقه بمؤشر فـي      )الخ...ل في جدول داخل قاعدة بيانات       حق
  .Deserializingبعكس عملية التسلسل شيفرتك وذلك 

 لن تخشى على كائناتك من الموت أو الضياع، اذ ان           -وبفضل التسلسل –من الان وصاعدا    
 سواء كان Scopeى خروج مؤشر الكائن من مداه بقاء الكائن على قيد الحياة لن يعتمد بعد الآن عل
  .ذلك بنهاية الإجراء، أو بالإنهاء الكلي للبرنامج

، ولا يقتـصر  NET Framework.يلعب التسلسل دورا حيويا في تطبيقات إطار عمـل  
استخدامه على حفظ الكائنات في وحدات تخزين كملفات في الاقراص، بـل أنـه يتجـاوز ذلـك         

 التقنية تمكن المبرمجين من تبادل كائناتهم بين البرامج المختلفة، او حتى بـين   بمراحل، اذ أن هذه   
 او تطبيقـات  Web Servicesدول العالم المتباعدة عن طريق شبكة الانترنت باستخدام خـدمات  

ASP.NET .      إلـى ان الاسـتثناءات الخاصـة       اكتشاف الأخطـاء  وقد اشرت في الفصل السابع 
Custom Exceptionsتكون قابلة للتسلسل  يجب ان Serializable    لتتمكن مـن رميهـا بـين 

  .المجمعات المختلفة
إن جعل الفئة قابلة للتسلسل سيوفر عليك الكثير من الجهد والوقت، حيث أنها ستمكنك مـن          

كمـا  (حفظ بيانات الفئات دون الحاجة لكتابة سطور اضافية عند اضافة حقول او خصائص للفئة               
  ). خاصةCustom Streams عندما طورنا الملفات والمجلداتق فعلنا في الفصل الساب

  
  Binary Serializationالتسلسل بالصیغة الثنائیة 
 يعرف جيـدا كيـف يجـري    NET Framework.في البداية دعني أطمئنك إلى ان إطار عمل 

، String ،Integer ،Doubleكــ   (Primitive Typesالتسلسل على انواع البيانات الأوليـة  
Boolean... فكل ما تحتاجه لتطبيق التسلسل على هذه الانواع هو كائن )الخ ،Formatter.  

وهي واجهة موجودة في     (IFormatter هو كائن يحتوي على الواجهة       Formatterالكائن  
، يمكنك تطوير فئات خاصة بك تشتمل على      )System.Runtime.Serializationمجال الاسماء   

 يحتوي على مجموعة جيدة من الفئـات  NET Framework.ر عمل هذه الواجهة، رغم ان إطا
تجدها فـي مجـال      (BinaryFormatterالتي تستطيع استخدامها مباشرة، من هذه الفئات الفئة         

وهي فئة تستخدم لحفـظ     ) System.Runtime.Serialization.Formatters.Binaryالاسماء  
  .Binary Formatبيانات الكائن بالصيغة الثنائية 

سأعرض لك الان مثالا يطبق التسلسل على البيانات القياسية، وفي الفقرات التالية سـنطبقه       
كل المطلوب منك هو انشاء كائن من الفئة     . على بيانات خاصة تعرفها على شكل فئات خاصة بك        

BinaryFormatter    واستدعاء الطريقة Serialize()        لحفظ بيانات الكائن، تتطلب هذه الطريقـة 
 فـي هـذا     FileStreamاستخدمت كائن من النـوع       (Stream الاولى وحدة التخزين     وسيطتين

  :، والثانية هو الكائن المطلوب تطبيق التسلسل عليه)المثال
 
 

Dim data As String() = {"عباس" ,"برعي" ,"حسونة"} 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
 
 بدء عملیة التسلسل '
SerialObj.Serialize(st, data) 
 
 
' Stream اغلق وحدة التخزین 
st.Close() 

  
والان تستطيع في اي وقت وباستخدام اي برنامج ومن اي مكان على وجه الارض إعـادة              

 ولكن باستدعاء الطريقة    BinaryFormatterإحياء الكائن من جديد، يتم ذلك باستخدام نفس الفئة          
Deserialize()              هذه المرة لعكس عملية التسلسل، تتطلب هذه الطريقة وسيطة واحدة فقط وهـي 
عليك اسـتخدام    (Object، وستعود الطريقة بكائن حي يرزق من النوع         Streamوحدة التخزين   

  ):Option Strict On لإجراء عملية التحويل المناسبة وذلك بسبب العبارة CTypeالمعامل 
 
 

Dim data As String() 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
 
' Deserializing عكس عملیة التسلسل 
data = CType(SerialObj.Deserialize(st), String()) 
 
ArabicConsole.WriteLine(data(0)) ' حسونة 
ArabicConsole.WriteLine(data(1)) ' برعي 
ArabicConsole.WriteLine(data(2)) ' عباس 
st.Close() 

  
 Object Instanceان ما فعلناه في الشيفرتين السابقتين ليس سوى نقل نسخة بيانات الكائن 

، ومن ثـم قمنـا     ()Serialize الى وحدة تخزين باستدعاء الطريقة       dataوالتي يشير لها المؤشر     

 باسـتدعاء الطريقـة     dataبعكس العملية ونقلنا نسخة الكائن من وحدة التخزين الـى المؤشـر             
Deserialize() 2-9شكل ( والخاصة بكائن التسلسل(.  

  

  
  .عكس تسلسل نسخة الكائن/ تسلسل:2-9شكل 

  
لتي تعرفهـا بنفـسك   والان دعنا نرى كيف يمكننا تطبيق التسلسل على البيانات الاخرى وا          

  . كما في الفقرة التاليةClass … End Classباستخدام التركيب 
  

  )غیر قیاسیة(تسلسل أنواع بیانات مخصصة 
 Serializableتطبيقيا، كل ما تحتاجه لجعل فئاتك قابلـة للتسلـسل هـو اسـتخدام المواصـفة      

Attributeوجعلها في اعلى شيفرة تعريف الفئة فقط ،:  
 
<Serializable()> _ 
Class Person 
    … 
    … 
End Class 

) Privateحتـى وان كانـت خاصـة        (ما يجب عليك معرفته هو ان جميع الحقول التابعة للفئة           
 داخل إجراءات الفئة سيطبق عليها التسلسل، ولكـي  Static Variablesوالمتغيرات الاستاتيكية 

  :Class … End Classتتحقق من ذلك اضف هذه الشيفرة بين فكي التركيب 
 
 

<Serializable()> _ 
Class Person 
    Public Name As String ' Public متغير عام 
 
    Private m_Age As Integer ' Private متغير خاص 
    Property Age() As Integer 
        Get 
            Return m_Age 
        End Get 
        Set(ByVal Value As Integer) 
            m_Age = Value 
        End Set 
    End Property 
 
    Function GetValue() As Integer 
        Static x As Integer ' Static متغير ستاتیكي 
 
        x += 1 
        Return x 
    End Function 
End Class 

  
أي قيم لحقولها وخصائصها ولا تنـسى اسـتدعاء       والآن ابدأ باستخدام الفئة السابقة، واسند       

 عدة مرات حتى تزيد من قيمة المتغير الستاتيكي المحفوظ بها، ومن ثم قم              ()GetValueالطريقة  
  :بتطبيق التسلسل على الكائن

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
Dim Turki As New Person() 
 
 
Turki.Name = "تركي العسيري" 
Turki.Age = 99 
Turki.GetValue() 
Turki.GetValue() 
Turki.GetValue() 
 
SerialObj.Serialize(st, Turki) 
 
st.Close() 

 ان اردت عكس عملية التسلسل، فلا داعي لانشاء نسخة جديدة مـن الكـائن باسـتخدام الكلمـة                 
وسنقوم بإسنادها الى   ) الملف(، اذ ان النسخة موجودة وجاهزة في وحدة التخزين          Newالمحجوزة  

  ): كما في الشيفرة التاليةTurkiوهو (مؤشر 
 

Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
Dim Turki As Person 
 
Turki = CType(SerialObj.Deserialize(st), Person) 
 
ArabicConsole.WriteLine(Turki.Name)  ' تركي العسيري 
ArabicConsole.WriteLine(Turki.Age)  ' 99 
ArabicConsole.WriteLine(Turki.GetValue) ' 4 
 
st.Close() 
 
 

    ملاحظة
  

د مـن ان تكـون الفئـة    عند عكس عملیة التسلسل لفئاتك الخاصة، فلاب  
المراد عكس تسلسلھا معرفة في البرنامج الذي يعید كائنھا، وذلك امـا   
بتعريــف الفئــة فــي شــیفرة البرنــامج، او اضــافتھا مــن قائمــة المراجــع   

Reference.  
 

  
من كان يصدق ان بيانات الكائنات يتم حفظها واسترجاعها بهذه السهولة؟ لدي اضافة بسيطة 

هذه المواصفة ستجعل جميع : Serializable Attributesاستخدام المواصفة بودي توضحيها عند 
 قابلة للتسلسل، ان لم ترغب في أن تكون جميع عناصـر الفئـة قابلـة          -كما قلت –بيانات الكائن   

مثل الوقـت الحـالي او   (للتسلسل، كأن تكون بعض الخصائص القابلة للعودة بقيمة غير محفوظة    
، او أن بعض الحقول تابعة لفئات غير قابلة للتسلسل اصلاً،          )خرىاستخلاص قيمة من خصائص ا    

ومهما يكن هدفك من تطبيق مفهوم التسلسل، يمكنك منع المتغير من التسلسل ضمن عناصر الفئة               
  :NonSerialized Attributesباستخدام المواصفة 

 
<Serializable()> _ 
Class Person 
    Public Name As String 
یتم تسلسل هذا المتغيرلن  '      
    <NonSerialized()> Private MotherName As String 
    Private m_Age As Integer 
    … 
    … 
End Class 

  Object Graphخريطة الكائنات 
 Objectالاهـرام الكائنيـة   في لغات البرمجة السابقة، لم تخلـو برامجنـا ومـشاريعنا مـن     

Hierarchies   الاحتواء  ل علاقة    والتي تنجز بفضContainment      بين الفئات، والتي تسمى -
، فما بالك بمشاريعك التي تنـجز تحـت اطار عمل Has aيحتوي على  بعلاقة -OOPفي عالم 

.NET Framework والتي يعد كل شيء فيها عبارة عن كائن Object؟  
ــات  ــة الكائن ــات Object Graphخريط ــي العلاق ــع Relationships ه  والمراج

References الموجودة بين الفئات في الهرم الكائني Object Hierarchy  فلو كانت لدينا هـذه ،
  :الفئات الثلاث

 
<Serializable()> _ 
Class CarEngine 
    Public Cylinder As String 
    Public Volume As String 
End Class 
 
<Serializable()> _ 
Class Car 
    Public Name As String 
    Public Model As String 
    Public Engine As New CarEngine() 
End Class 
 
<Serializable()> _ 
Class Person 
    Public Name As String 
    Public Car As New Car() 
End Class 

  
 بالـصفحة  3-9بالشكل ( لهذه الفئات كما Object Graphفيمكن رسم خريطة الـكائنات 

 بـين الفئـات ولـيس    Containment معرفة ان الشكل يبين علاقة الاحتـواء    عليكو. )لةالمقاب
  .Inheritanceالاشتقاق الوراثي 

  . علاقة الاحتواء بين الفئات:3-9شكل 

  
    ملاحظة

  
 تـشمل الكائنـات مـن النـوع المرجعـي      Object Graphخريطة الكائنـات  
Reference Typeفقط، اما البیانات ذات القیمة  Value Type  فھي تتبـع 

  .Has aيحتوي على الكائن نفسه، ولا يقال علیھا العلاقة 
 

  
الذي اريد ان اصل إليه معك من الكلام السابق، هو ان تطبيق التسلسل على كائن ما سيؤدي 

 لهذا الكائن، وسيطبق Object Graphإلى تضمين جميع الكائنات الموجودة في خريطة الكائنات 
  : جميعاً، وهذا هو الدليلالتسلسل عليها

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
Dim Turki As New Person() 
 
 اسناد قیم للحقول '
Turki.Name = "تركي العسيري" 
Turki.Car.Name = "BMW" 
Turki.Car.Model = "Class 7" 
Turki.Car.Engine.Cylinder = "12 V" 
Turki.Car.Engine.Volume = "999 xm" 
 
 تطبیق التسلسل '
SerialObj.Serialize(st, Turki) 

 مؤشر للاختبار '
Dim Test As Person 
 
 لابد من تحریك مؤشر الملف الى البدایة '
st.Seek(0, SeekOrigin.Begin) 
 
 عكس التسلسل '
Test = CType(SerialObj.Deserialize(st), Person) 
 
  وباقي خریطة الكائناتطباعة محتویات الكائن '
ArabicConsole.WriteLine(Test.Name)   ' تركي العسيري 
ArabicConsole.WriteLine(Test.Car.Name)  ' BMW 
ArabicConsole.WriteLine(Test.Car.Engine.Cylinder) ' 12 V 
… 
… 
… 
 
st.Close() 

  
  :Circular Reference المرجعية الدائرية حالة

 بتركيز وتمعن، فـبلا شـك       الفئات والكائنات ان كنت يا عزيزي القارئ قد قرأت الفصل الثالث          
ستتذكر ان المرجعية الدائرية ما هي الى علاقة بين كائنين يشيران الى بعضهما البعض، فلو كان                

  : حقل من نفس نوع الفئةPersonللفئة 
 
<Serializable()> _ 
Class Person 
    Public Name As String 
    Public Brother As Person 
End Class 

  
  :وقمت بانشاء كائنين وأحدثت مرجعية دائرية بينهما

 
Dim Abbas As New Person() 
Dim Burey As New Person() 
 
Abbas.Name = "عباس" 
Burey.Name = "برعي" 
Abbas.Brother = Burey 
Burey.Brother = Abbas 

  
 Object، فاستنادا الى قاعدة خريطة الكائنات Abbasثم اردت تطبيق التسلسل على الكائن الاول 

Graph     التي تحدثت عنها قبل قليل، سيتم تطبيق التسلسل على الكائن الاول والبحث عـن جميـع 
 Bureyيشمل الكائن الثـاني     ، ل Brotherالعلاقات التي يحتويها، وهي العلاقة المتمثلة في الحقل         

، ويعود مرة اخرى الى الكـائن       Brotherويبحث عن جميع العلاقات التي يحتويها وهي العلاقة         
  . وهكذا الى مالا نهايةBurey، ومن ثم الى Abbasالاول 

السيناريو السابق صحيح نظرياً، ولكن يسرني اخبارك بان مطوري فئات التسلسل لديهم من         
ما يكفي لتجاوز هذه المشكلة، حيث ان فئات التسلسل المقدمة من اطـار عمـل               الخبرة البرمجية   

.NET               لا تقوم بتسلسل الكائن اكثر من مرة، مما يعني ان الكائنين الـسابقين Abbas   و Burey 
  :سيتم تسلسلهما مرة واحدة فقط، والشيفرة التالية خير دليل

 
Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
Dim Abbas As New Person() 
Dim Burey As New Person() 
 
Abbas.Name = "عباس" 
Burey.Name = "برعي" 
Abbas.Brother = Burey 
Burey.Brother = Abbas 
 
 لاتوجد اي مشاكل في عملیة التسلسل '
SerialObj.Serialize(st, Abbas) 
 
 
 
Dim Test As Person 
 
st.Seek(0, SeekOrigin.Begin) 
Test = CType(SerialObj.Deserialize(st), Person) 
ArabicConsole.WriteLine(Test.Name)  ' عباس 
ArabicConsole.WriteLine(Test.Brother.Name) ' برعي 
st.Close() 

  
  نسخ الكائنات
 ICloneable عرضت عليـك الواجهـة       لتفويض، والمواصفات الواجهات، ا في الفصل الخامس    

وذكرت لك أن بامكانك تضمينها في فئاتك ان اردت اعطاء امكانية نسخ كائناتها، وذلك باستدعاء               
مع ذلك، توجد مشكلة لم اخبرك . Object التابعة للفئة القاعدية ()MemberwiseCloneالطريقة  

وقد حان وقت مناقشتها الآن، وتتمثل في ان الكائنات الاخرى ) لان حلها يتم باستخدام التسلسل(بها  
  :التي تحتويها الفئة لا يمكن نسخها، راقب هذه الفئات

 
Class Car 
    Public Name As String 
    Public Model As String 
End Class 

Class Person 
    Implements ICloneable 
 
 
    Public Name As String 
    Public Car As New Car() 
 
    Private Function PrivateClone() As Object Implements 
ICloneable.Clone 
        Return Me.Clone() 
    End Function 
 
    Public Function Clone() As Person 
        Return CType(Me.MemberwiseClone(), Person) 
    End Function 
End Class 

  
  :حسنا، دعنا نجرب نسخ الكائن ونرى ما اذا كانت الكائنات المحضونة فيه سيتم نسخها ام لا

 
Dim Turki As New Person() 
Dim Turki2 As Person 
 
Turki.Name = "تركي العسيري" 
Turki.Car.Name = "BMW" 
Turki.Car.Model = "Class 7" 
 
Turki2 = Turki.Clone 
 
ArabicConsole.WriteLine(Turki2.Name)  ' تركي العسيري 
ArabicConsole.WriteLine(Turki2.Car.Name) ' BMW 

  
 قد تم نسخه بالفعل، وبـذلك يكـون         Carمن المخرجات السابقة، يتضح لنا ان الكائن المحضون         

  ".الكائنات الاخرى التي تحتويها الفئة لايمكن نسخها ان"مؤلف الكتاب مخطئا في عبارته 
 نسخ مؤشره الى    تم لم يتم نسخه بالفعل وانما       Carت مصراً على ان الكائن      مع ذلك، مازل  

 الـذي   Is، والدليل سينفجر امام عينيك باسـتخدام المعامـل          Turki2 والتابع للكائن    الحقل الاخر 
 يشيران الى نفس نسخة الكائن، مما يعني Turki2.Car و  Turki.Carسيكشف لنا ان المؤشرين     

  :)4-9شكل (ه أن الكائن لم يتم نسخ
 
ArabicConsole.WriteLine(Turki.Car Is Turki2.Car) ' True 

  . فقطCar تم نسخ مؤشر الكائن المحضون :4-9شكل 

  
، حيـث أن  Shallow Copying الـسطحي بالنسخ  يسمى النسخ السابق NET.في عالم 

امـا ان  . ؤشراتها فقط، وانما مObject Graphالنسخ لا يشمل الكائنات التابعة لخريطة الكائنات 
بالنـسخ  اردت نسخ جميع الكائنات الموجودة في خريطة الكائنات، فانك تريد تطبيق مـا يـسمى     

لعمل ذلك، لن تجد اسهل من استخدام التسلسل، اعد كتابـة الطريقـة   . Deep Copyingالعميق 
Clone() في الفئة Person   ولا تنـسى اسـتخدام المواصـفة    ( بهـذا الـشكلSerializable 

Attributes في الفئات Person و Car:(  
 
 

<Serializable()> _ 
Class Car 
    … 
    … 
End Class 
 
<Serializable()> _ 
Class Person 
    Implements ICloneable 
    … 
    … 

    Public Function Clone() As Person 
        Dim st As FileStream = File.Open("C:\temp.tmp", _ 
            FileMode.OpenOrCreate) 
         
        Dim SerialObj As New BinaryFormatter() 
 
        SerialObj.Serialize(st, Me) 
        st.Seek(0, SeekOrigin.Begin) 
        Return CType(SerialObj.Deserialize(st), Person) 
        st.Close() 
        File.Delete("C:\temp.tmp") 
    End Function 
End Class 

  
  :وهذا هو الاختبار الذي يثبت انه تم بالفعل نسخ جميع الكائنات المحضونة في الكائن

 
 

Dim Turki As New Person() 
Dim Turki2 As Person 
 
Turki.Name = "تركي العسيري" 
Turki.Car.Name = "BMW" 
Turki.Car.Model = "Class 7" 
 
Turki2 = Turki.Clone 
 
ArabicConsole.WriteLine(Turki2.Name)  ' تركي العسيري 
ArabicConsole.WriteLine(Turki2.Car.Name) ' BMW 
 
ArabicConsole.WriteLine(Turki.Car Is Turki2.Car) ' False ç اهم اثبات 

  .Deep Copying النسخ العميق :5-9شكل 

  
    ملاحظة

  
 BinaryFormatter التابعـة للفئــة  ()Serializeكمـا تعلـم فــان الطريقـة    

ـــوع    ـــن مـــن الن  لتطبیــق Streamتتطلــب فــي وســیطتھا الأولــى كائ
 فـي الفئـة     ()Cloneعنـدما اعـدت كتابـة الطريقـة         . التسلسل علیھــا 

Person      الاخیرة، استخدمت وحدة التخزين FileStream    لانـشاء ملـف 
  . لزيادة السرعةMemoryStreamمؤقت، يمكنك استخدام 

 

  
  اصة خCustom Serializationانشاء 

قد تحتاج الى تطوير اسلوب خاص لتطبيق التسلسل على فئاتك الخاصة، اسـتخدامك للمواصـفة               
Serializable Attributes   محدود جدا حيث أن كل ما ستفعله بهذا الاسلوب هو جعـل جميـع 

 لمنع NonSerialized Attributesالحقول قابلة للتسلسل، صحيح انك تستطيع استخدام المواصفة 
تخيل مثلا اننا نريد منع     . عن أي حقل، الا ان عملية المنع تتم وقت التصميم وليس التنفيذ           التسلسل  

عملية التسلسل للحقول التي تكون قيمتها سالبه وقت التنفيذ، في هذه الحالة عليك تطوير تسلسلات               
  .  كما سترى في الفقرات التاليةCustom Serializationsخاصة 

  ISerializableالواجھة 
 في الفئة، تحتـوي  ISerializableان اردت تطوير تسلسلات خاصة بك، فعليك تضمين الواجهة        

  : وصيغتها()GetObjectDataهذه الواجهة على طريقة واحدة فقط هي 
 
Sub GetObjectData(ByVal info As SerializationInfo, _  
    ByVal context As StreamingContext) 
    … 
    … 
End Sub 

  
يتم استدعاء هذه الطريقة تلقائيا بمجرد تطبيق التسلسل على الفئة، استخدم الوسيطة المرسلة الاولى    

info                  ان اردت اضافـة بيانات تود تضمينها في تسلسل الكـائن الحـالي باسـتخدام طريقتهـا 
AddValue():  

 
info.AddValue("Age", Me.Age) 
 

  : لابد من انشاء مشيد مخفي بهذه الصيغة في الفئة،ISerializableعند تضمين الواجهة 
 
Private Sub New(ByVal info As SerializationInfo, _  
    ByVal context As StreamingContext) 
    … 
    … 
End Sub 

  
 infoسيتم استدعاء المشيد المخفي السابق لحظة عكس عملية التسلسل، يمكنك استخدام الوسـيطة             

  :()GetXXX باستخدام الطرق ايضا لقراءة البيانات
 
Me.Age = info.GetInt32("Age") 

  
 في الفئة، وكمـا ذكـرت فـي    Private Constructerبما انك صرحت عن مشيد مخفي 

 من انك لن تتمكن من انشاء نسخة كائن من هذه الفئة، لذلك عليك              الفئات والكائنات الفصل الثالث   
 قابل للوصول حتى تتمكن من انشاء كـائن مـن    المشيد واضافة مشيدOverloadsاعادة تعريف   

  :الفئة
 
Public Sub New() 
 
End Sub 

  مثال تطبیقي
 والتي تحتوي على اربعـة حقـول مـن النـوع     TestClassفي المثال التالي قمت بتطوير الفئة   

Integer)        قد تفضل استخدام مصفوفة من النوعInteger        ولكن هدفي هو توضيح الفكرة بـشكل 
 لتطوير تسلسل خاص بها، حيث ان       ISerializable هذه الفئة قمت بتضمين الواجهة       ، في )اسهل

ركز فـي شـيفرات   . الحقول موجبة القيمة هي التي سيتم تسلسلها فقط، اما السالبة فسيتم تجاهلها           
 Private Sub حيث يتم استدعائه لحظة التسلسل، والمشيد المخفي ()GetObjectDataالاجراء 

New()التسلسل لحظة عكس :  
 

<Serializable()> _ 
Class TestClass 
    Implements ISerializable 
 
    Public X As Integer 
    Public Y As Integer 
    Public Z As Integer 
    Public W As Integer 
 
    Public Sub New() 
 لا احتاج هذا المشید حالیا في هذه الفئة '        
كن من الضروري تعریفه حتى تتمكن منول '          
 انشاء كائن من هذه الفئة '        
    End Sub 
 
عكس التسلسلسیتم استدعاء هذا الاجراء لحظة  '     
   ' Deserialization 
    Private Sub New(ByVal info As SerializationInfo, _  
  ByVal context As StreamingContext) 
 
        On Error Resume Next 
        Me.X = info.GetInt32("X") 
        Me.Y = info.GetInt32("Y") 
        Me.Z = info.GetInt32("Z") 
        Me.W = info.GetInt32("W") 
 
    End Sub 
  
 سیتم استدعاء هذا الاجراء لحظة التسلسل '   
   Public Sub GetObjectData(ByVal info As SerializationInfo, _  
  ByVal context As StreamingContext) Implements _  
    ISerializable.GetObjectData 
 
        If Me.X >= 0 Then 
            info.AddValue("X", Me.X) 
        End If 
        If Me.Y >= 0 Then 
            info.AddValue("Y", Me.Y) 
        End If 

        If Me.Z >= 0 Then 
            info.AddValue("Z", Me.Z) 
        End If 
        If Me.W >= 0 Then 
            info.AddValue("W", Me.W) 
        End If 
    End Sub 
End Class 

  
لاحـظ ان  (عكس التسلسل على كائن من الفئـة الـسابقة   /وهنا مثال تطبيقي يقوم بتطبيق التسلسل     

  ):الحقول السالبة لن يتم تسلسلها
 

Dim st As FileStream = File.Open("C:\test.dat", FileMode.OpenOrCreate) 
Dim SerialObj As New BinaryFormatter() 
Dim TestObject As New TestClass() 
 
TestObject.X = 1 
TestObject.Y = -1 
TestObject.Z = -5 
TestObject.W = 5 
 
SerialObj.Serialize(st, TestObject) 
 
Dim tmp As TestClass 
st.Seek(0, SeekOrigin.Begin) 
tmp = CType(SerialObj.Deserialize(st), TestClass) 
 
ArabicConsole.WriteLine(tmp.X) ' 1 
ArabicConsole.WriteLine(tmp.Y) ' 0 
ArabicConsole.WriteLine(tmp.Z) ' 0 
ArabicConsole.WriteLine(tmp.W) ' 5 

  
  XMLالتسلسل بصیغة 

، فقد تطبق التسلـسل علـى الكائنـات    Web Servicesان كنت مقبلا على تطوير خدمات ويب 
 بسهولة شـديدة   XML القياسية، حيث يمكن تطبيق التسلسل على الكائنات بصيغة          XMLبصيغة  

  .NET Framework. المقدمة من اطار عمل XmlSerializerوذلك بفضل الفئة 
الفئة من الضروري تنبيهك إلى ثلاث نقاط ضعف يمكن اعتبارها قصوراً           قبل استخدام هذه    

  : كبيراً في الفئة
  :Publicانه لا يمكنك تطبيق التسلسل الا على الفئات المعرفة بمحدد الوصول : الاولى

 يمكن تطبیق التسلسل بصیغة XML علیها '
Public Class Person 
    … 
End Class 
 
 غير ممكن '
Class Car 
    … 
End Class 

  
  : هي التي سيتم تطبيق التسلسل عليها فقطPublicان الحقول العامة : الثانية

 
Public Class Test 
    Public X As Integer ' ممكن 
    Friend Y As Integer ' غير ممكن 
    Private Z As Integer ' غير ممكن 
    … 
End Class 

  
 فيما لو Object Graphي فشل تطبيق التسلسل على خريطة الكائنات اما النقطة الثالثة فتتلخص ف

 بعكس الحال مع التسلـسل  Circular Referenceاحتوت الخريطة على حالة المرجعية الدائرية 
  .Binaryبالصيغة الثنائية 

، قد تـستغرب الاصـرار علـى        XMLمع كل هذا القصور في تطبيق التسلسل بالصيغة         
 لتبادل البيانـات مـع   XMLلاحوال ستطبق التسلسل على كائناتك بصيغة     استخدامها؟ في اغلب ا   

، وهي صيغة مفهومـة لانهـا   Web Servicesالبرامج الاخرى خاصة عند تطوير خدمات ويب 
  .XMLالفقرات التالية ستكشف لك مزايا وامكانيات اضافية للتسلسل بصيغة . قياسية

  
    ملاحظة

  
ضـافیة علیـك اسـتیرادھا مـن     ھذا القسم من الفـصل يـستخدم فئـات ا      

  :مجال الاسماء التالي لاختصار الشیفرات البرمجیة
 
Imports System.Xml.Serialization 

  XmlSerializerالفئة 
، فلست مـضطرا لاسـتخدام المواصـفة        XMLان كنت تنوي جعل فئاتك قابلة للتسلسل بصيغة         

Serializable Attributeاً لاستخدام محدد الوصول ، ولكنك بالتأكيد ستكون مضطرPublic:  
 
Public Class Person 
    Public Name As String 
    Public Age As Integer 
End Class 

  
الموجـودة   (XmlSerializer على الفئة السابقة، فاستخدم الفئة       XMLولتطبيق التسلسل بصيغة    

 بنفس اسلوب اسـتخدام     واستخدامها يتم ) System.XML.Serializationضمن مجال الاسماء    
  : ولكنها تتطلب تحديد نوع الكائن المراد تسلسله في مشيدهاBinaryFormatterالفئة 

 
Dim st As FileStream = File.Open("C:\test.xml", FileMode.OpenOrCreate) 
Dim SerialObj As New XmlSerializer(GetType(Person)) 
Dim Turki As New Person() 
 
Turki.Name = "تركي العسيري" 
Turki.Age = 99 
 
SerialObj.Serialize(st, Turki) 
 
st.Close() 

  
 والذي سيحتوي بيانات الكـائن بالـصيغة        C:\Test.XMLالشيفرة السابقة ستقوم بانشاء الملف      

XML:  
 
<?xml version="1.0" ?>  
    <Person xmlns:xsd="http://www.w3.org/2001/XMLSchema"  
        xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"> 
        <FullName>تركي العسيري</FullName>  
        <Age>99</Age>  
    </Person> 
 
 

  انظر ايضا
  

سأفترض في ھذه الفقرة والفقرات التالیة بانك تجیـد التعامـل مـع لغـة             
لغــة : ك الملحـق أ ، وان لـم تكـن كــذلك فقـد يفیـد    XMLوصـف البیانـات   
  . والملقى في نھاية ھذا الكتابXMLوصف البیانات 

ان كـان  -، يمكنك عكس عمليـة التسلـسل      BinaryFormatterكما فعلنا سابقا مع الفئة      
، كما ترى   XmlSerializer والتابعة للكائن    ()Deserialize باستدعاء الطريقة    -XMLبصيغة  

  :في الشيفرة التالية
 
Dim st As FileStream = File.Open("C:\test.xml", FileMode.OpenOrCreate) 
Dim SerialObj As New XmlSerializer(GetType(Person)) 
 
st.Seek(0, SeekOrigin.Begin) 
Dim Test As Person = CType(SerialObj.Deserialize(st), Person) 
 
ArabicConsole.WriteLine(Test.Name) ' تركي العسيري 
ArabicConsole.WriteLine(Test.Age) ' 99 
 
st.Close() 

  
  مواصفات اضافیة

 التي تعطيـك تحكمـا      Attributes مجموعة كبيرة من المواصفات      NET.يوفر لك اطار عمل     
  :من هذه المواصفات. XMLوسيطرة اكبر على عملية التسلسل بصيغة 

  
  :XmlRoot Attributesالمواصفة 

 Root XML من تحديد الاسـم للعنـصر الجـذري    XmlRoot Attributesتمكنك المواصفة 

Element في مستند XML Document بالاضافة الى مجال الاسماء ،Namespaceللعنصر  .
  :يتم استخدام هذه المواصفة في الفئة

 
<XmlRoot("PersonRecord", namespace:="http://www.dev4arabs.com")> _ 
Public Class Person 
    Public Name As String 
    Public Age As Integer 
End Class 

  
عند تطبيق التسلسل على كائن منشأ من الفئة السابقة، ستلاحظ ان اسم العنـصر الجـذري هـو                  

PersonRecord وليس اسم الفئة Person:  
 
<?xml version="1.0" ?>  
    <PersonRecord xmlns:xsd="http://www.w3.org/2001/XMLSchema"  
    xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"  
    xmlns="http://www.dev4arabs.com"> 
        <Name>تركي العسيري</Name>  
        <Age>99</Age>  
    </PersonRecord> 

  :XmlElement Attributesالمواصفة 
 XML Tagsء حقوله في وسوم ، فسيتم حفظ اسماXMLعند تطبيق التسلسل على كائن بصيغة 
 عند تسلـسله بالـصيغة    اما ان رغبت في تغيير اسم الحقل      . كما هي موجودة بالشيفرة المصدرية    

XML فاستخدم المواصفة ،XmlElement Attributes:  
 
Public Class Person 
    <XmlElement("FullName")> Public Name As String 
    Public Age As Integer 
End Class 

  
 قد تمت تـسميته  Nameعند تطبيق التسلسل على كائن منشأ من الفئة السابقة، ستلاحظ ان الحقل           

  : في الوسم الخاص بهFullNameبـ 
 
<?xml version="1.0" ?>  
    <Person xmlns:xsd="http://www.w3.org/2001/XMLSchema"  
        xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"> 
        <FullName>تركي العسيري</FullName>  
        <Age>99</Age>  
    </Person> 

  :XmlAttributeAttribute Attributesالمواصفة 
 من تسلسل الكائن وحفظه كواصـف  XmlAttributeAttribute Attributesتمكنك المواصفة 

  : التاليIDلحقل ، فاElement بدلا من عنصر XML Documentsلمستند 
 
Public Class Person 
    … 
    … 
    <XmlAttributeAttribute("PersonID")> Public ID As String 
    … 
    … 
End Class 

  
  :ستكون مخرجاته مشابهة للتالي

 
<?xml version="1.0" ?>  
    <Person xmlns:xsd="http://www.w3.org/2001/XMLSchema" 
        xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance" 
        PersonID="1"> 
        <Name>تركي العسيري</Name>  
        <Age>99</Age>  
    </Person> 

  :XmlText Attributesالمواصفة 
 XML على الحقل ليتم تسلسله دون استخدام وسـوم  XmlText Attributesتستخدم المواصفة 

Tags:  
 
Public Class Person 
    … 
   <XmlText()> Public Note As String 
    … 
End Class 

  
، بـافتراض ان  XML Tags تسلسل الحقل السابق، سيتم تسلسل قيمته فقط ويتم تجاهل وسوم عند

  :، ستكون مخرجات الملف شبيهه بـ"ملاحظات اضافية حول الشخص= "قيمة الحقل 
 
<?xml version="1.0" ?>  
    <Person xmlns:xsd="http://www.w3.org/2001/XMLSchema"  
        xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"> 
        <Name>تركي العسيري</Name>  
        <Age>99</Age>  
 ملاحظات اضافیة حول الشخص        
    </Person> 

  
  :XmlArray and XmlArrayItem Attributes تانالمواصف

 عند التعامل  جنبا الى جنبXmlArray and XmlArrayItem Attributesتعمل المواصفتان 
الاولـى  تستخدم المواصفة ، حيث )وبالتحديد مصفوفات الكائنات(مع الحقول من نوع المصفوفات      

XmlArray Attributes لتعريف اسم عنصر XML Elemetent ة  الخارجي بينما المواصـف
  : التاليةCustomerراقب الفئة . الثانية للعنصر الداخلي

 
Public Class Order 
    <XmlAttributeAttribute("OrderId")> Public OrderID As String 
    Public [Date] As Date 
    Public Total As Decimal 
End Class 
 
Public Class Customer 
    Public Name As String 
    Public Address As String 
    <XmlArray("CustomerOrders"), XmlArrayItem("CustOrder")> _ 
    Public Orders(3) As Order 
End Class 

 السابقة، واسند قيم لجميع خصائصها وحاول تطبيق التسلسل         Customerقم بانشاء كائن من الفئة      
لاحظ الوسوم بالخط الـسميك     (ك شيئا مثل     في ملف خارجي، ليظهر ل     XMLعلى الكائن بصيغة    

Bold المواصـفتان   لتعرف كيف تؤثرXmlArray and XmlArrayItem Attributes  فـي 
  ):المخرجات

 
<?xml version="1.0" ?>  
    <Customer xmlns:xsd="http://www.w3.org/2001/XMLSchema" 
    xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"> 
            <Name>تركي العسيري</Name>  
            <Address>المملكة العربیة السعودیة</Address>  
        <CustomerOrders> 
            <CustOrder OrderId="1"> 
                <Date>2002-12-12T02:09:48.9743296+03:00</Date>  
                <Total>100</Total>  
            </CustOrder> 
            <CustOrder OrderId="2"> 
                <Date>2002-12-10T00:00:00.0000000+03:00</Date>  
                <Total>200</Total>  
            </CustOrder> 
            <CustOrder OrderId="3"> 
                <Date>2002-12-09T00:00:00.0000000+03:00</Date>  
                <Total>300</Total>  
            </CustOrder> 
            <CustOrder xsi:nil="true" />  
        </CustomerOrders> 
    </Customer> 

  
  :XmlIgnore Attributesالمواصفة 

 قبل اسم الحقل سيتم تجاهله ولن تطبق عليه XmlIgnore Attributesلمواصفة لإرفاقك اخيرا، 
  : التالي لن يتم شمله في التسلسلMotherNameفالحقل . عملية التسلسل

 
Public Class Person 
    … 
    … 
    <XmlIgnore()> Public MotherName As String 
    … 
    … 
End Class 

  أحداث تقع عند عكس التسلسل
 على اربعة احداث يتم اطلاقها عند عكس التسلسل على الكائنـات           XmlSerializerتحتوي الفئة   

وبمـا ان وحـدة     . XMLوبالتحديد لحظة قراءة عنصر غير معرف في وحدة التخزين بالصيغة           
، فان امكانية تعديل هذه الملفـات مـن         XML هي ملفات    -في اغلب الاحوال  -التخزين ستكون   

لذلك، قد تحتاج الى هذه الاحداث التابعة . برامج التحرير المختلفة امر قد يحدث ولو بطريق الخطأ    
  .للكائنات التي تنشئها من هذه الفئة

 والذي يتم إطلاقه    UnknownElementاعرض لك في هذه الفقرة طريقة استخدام الحدث         
، يمكنك الاستفادة مـن هـذا       XML معرف في وحدة التخزين بصيغة       بمجرد وجود عنصر غير   

الحدث لمعرفة العناصر الغير معرفة والتي لا تتوافق مع عناصر فئة الكائن الذي يجري عكـس                
ان لم تكن قد كتبت شيئا في هذا الحدث وتم العثور على عنصر غيـر معـرف، فـسيتم            . تسلسله

  .حقله الافتراضيتجاهل هذا العنصر ولن يتم استرجاعه في 
 يدويا ان XMLمع ذلك، قد تستفيد من هذا الحدث بتعبئة الحقول التي تم تسلسلها في ملفات   

  : التاليةPersonمثلا، افترض انه تم تعريف الفئة . كانت العناصر غير معرفة
 
Public Class Person 
    Public Name As String 
    Public Age As Integer 
End Class 

  
 على XMLد تطبيق التسلسل على كائن منشأ من الفئة السابقة، ستحتوي وحدة التخزين بصيغة         عن

  :شيئا مثل
 
<?xml version="1.0" ?>  
    <Person xmlns:xsd="http://www.w3.org/2001/XMLSchema"  
    xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"> 
        <Name> سيريتركي الع </Name>  
        <Age>99</Age>  
    </Person> 

  
والان تخيل ان احد المستخدمين قد احدث تغييراً في محتويات الملف السابق، وقام بتقسيم العنصر               

Name الى عنصرين هما FirstName و LastName:  
 
<?xml version="1.0" ?>  
    <Person xmlns:xsd="http://www.w3.org/2001/XMLSchema"  
    xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"> 
        <FirstName>تركي</FirstName>  
        <LastName>العسيري</LastName>  

        <Age>99</Age>  
    </Person> 

 
قل  بشكل صحيح، اما الح    Ageعند عكس التسلسل على الملف السابق، سيتم استعادة الحقل          

Name      لذلك، عليك قراءة العنصرين فـي ملـف         .  فلن يحتوي على شئ والسبب واضحXML 
 مـثلا   UnknowXMLElemntEventووضعهما في المكان المناسب، اكتب اجراء جديد باسم         

  :واملأه بهذه الشيفرة
 
Sub UnknowXMLElemntEvent(ByVal sender As Object, _  
    ByVal e As XmlElementEventArgs) 
 
    Dim tmpObj As Person = CType(e.ObjectBeingDeserialized, Person) 
 
    If e.Element.Name = "FirstName" Then 
        tmpObj.Name = e.Element.InnerXml 
    ElseIf e.Element.Name = "LastName" Then 
        tmpObj.Name = tmpObj.Name & " " & e.Element.InnerXml 
    End If 
End Sub 

  
الاجراء السابق سيتم استدعائه في كل مرة يتم اكتشاف عنصر غير معرف لحظة عكس التسلسل،               

 تمثل الكائن الذي يتم عكـس تسلـسله، والخاصـية           e.ObjectBeingDeserializedالخاصية  
e.Element.Name  ــزين ــي وحــدة التخ ــم العنــصر ف ــية XML هــي اس ، امــا الخاص

e.Element.InnerXml   الاجراء السابق سيدمج العنصرين     . ا فتمثل القيمة نفسهFirstName  و 
LastName      في حقل واحد وهو Name .    يمكنك تجربة الملف والاجراء السابق بعكـس عمليـة

 او WithEvents بالكلمـة المحجـوزة   UnknownElementالتسلسل، ولا تنسى قنص الحدث     
  : كما فعلت في الشيفرة التاليةAddHandlerباستخدام الامر 

 
Dim st As FileStream = File.Open("C:\test.xml", _  
    FileMode.OpenOrCreate) 
Dim SerialObj As New XmlSerializer(GetType(Person)) 
 
 
st.Seek(0, SeekOrigin.Begin) 
  اولاقنص الحدث '
AddHandler SerialObj.UnknownElement, AddressOf UnknowXMLElemntEvent 
 عكس التسلسل '
Dim Test As Person = CType(SerialObj.Deserialize(st), Person) 
 
ArabicConsole.WriteLine(Test.Name) ' تركي العسيري 
ArabicConsole.WriteLine(Test.Age) ' 99 
 
st.Close() 

 Objectتسلـسل الكائنـات   هذا ليس كل شيء، ولا أريد منك أن تعتقد انك عرفت الكثير عـن  

Serialization  بعد قراءتك للصفحات السابقة، حيث ان هذا الفصل لا يعتبر الا مقدمة بسيطة الى
ان كنت من المبرمجين الجادين و مهتما بهذا . Visual Basic .NETالتسلسل وطرق تطبيقه بـ 

الموضوع، فأنصحك بقضاء الأسابيع القادمة في دراسة المزيد والمزيد من التفاصيل المتقدمة حول 
اما الصفحات التالية .  او البحث عن مقالات اخرى في شبكة الانترنتMSDNمكتبة التسلسل من   

مـسارات التنفيـذ    وهـو  -موجـه ايـضا للمبـرمجين الجـادين     –فستأخذك الى موضوع اخر     
Threading.  

 Threadingمسارات التنفیذ 

  
ارات التنفيذ  تتطلب مبرمجين شجعان، فان برمجة مسWindows APIان كانت برمجة إجراءات 

 تتطلب مبرمجين لا يخشون لومة لائم ولديهم استعداد للتضحية بعقولهم Multithreadingالمتعددة 
  .ومنطقهم البرمجي

 NET Framework.هذا الفصل مخصص بأعقد موضوع من مواضيع برمجة إطار عمل 
نفيذ في حياته   ، وان كنت مبرمج لا ينوي التلاعب بمسارات الت        Threadingوهو مسارات التنفيذ    

  !البرمجية، فأنصحك بتجاهل هذا الفصل حتى تنعم بحياة برمجية اجمل
  

    ملاحظة
  

استرد مجال الأسـماء التـالي، حیـث ان جمیـع فئـات مـسارات التنفیـذ               
  :المذكورة في ھذا الفصل مشمولة فیه

 
Imports System.Threading 

  

  
 مقدمة إلى مسارات التنفیذ

 Threadمسار التنفيذ يكون . الشيفرةارة عن مسار يتبعه المعالج لتنفيذ عب Threadمسار التنفيذ 
البرنامج عندما يتم تنفيذه ويكون بالذاكرة يطلـق عليـه          ( واحد فقط    Process لبرنامج عامل تابع  

نطلـق  ، وفي هذه الحالة     مسار تنفيذ  على أكثر من     البرنامجقد يحتوي   . )Processبرنامج عامل   
جميع البرامج التي طورناها . Multi-Threaded Processمسارات التنفيذ متعدد عليه برنامج 

 Singleبـرامج أحاديـة مـسار التنفيـذ     من الفصل الأول لهذا الكتاب حتى هذا الفصل تسمى 

Threaded Process  بغـض النظـر عـن    ( والسبب انها تحتوي على مسار تنفيذ واحد فقـط
 NET Framework.ام بعض كائنـات إطـار عمـل    مسارات التنفيذ المخفية الناتجة من استخد

  ). مثلاGarbage Collectionكالمجموعة 

، وقد يقرب   لمسارات التنفيذ ك  باع من فهمك واستي   أتأكد حتى   الأخرىلن أنتقل إلى التفاصيل     
بصورة افضل، نفترض ان لديك برنامج يقوم بثلاث عمليات         مسارات التنفيذ   لك هذا المثال مفهوم     

. اظهار البيانات على الشاشة للبيانات، حفظ وتخزين البيانات، الإملائيهي التدقيق في وقت واحد و
 وبما ان هـذه العمليـات    إظهار،) 3تخزين  ) 2 إملائيتدقيق  ) 1 ذكرت ثلاثة عمليات     أننيلاحظ  

الثلاثة تتم في وقت واحد، نستطيع ان نقول ان هذا البرنامج يحتوي على ثلاثة مـسارات تنفيذيـة        
Three Threads.  مسار تنفيـذ  ، والإملائي عملية التدقيق لإجراء خاص مسار تنفيذفهنا يوجد 

مسار اما لو كان البرنامج يحتوي على . لإظهار البيانات خاص مسار تنفيذ وأخيرا ،خاص للتخزين
.  في وقـت واحـد  إجراءهاأي لا تستطيع -، فلابد ان تتم كل عملية بعد نهاية الأخرى         تنفيذ واحد 

إظهار البيانـات   وأخيرا ثم عملية التخزين أو الحفظ الإملائيسيقوم البرنامج بعملية التدقيق    لذلك،  
  .على الشاشة

برنامج الن تجعل   أقد تفيدك مسارات التنفيذ المتعددة في تنفيذ عدة عمليات في وقت واحد، ك            
رنامج بعرض بيانـات     بينما يقوم نفس الب    ،يقوم بعملية الطباعة مثلا أو الاستعلام في قاعدة بيانات        

وحتى .  العملية أو حتى عمل أي شيء آخر       بإلغاء فرصة للمستخدم    إعطاءأخرى على الشاشة أو     
مثلا، فأنت بالفعل تحتاج إلى     لشبكة محلية    Serverخادم  في جهاز   اذا قمت بتطوير برنامج يعمل      

 في  الانتظارامه دون   مسارات التنفيذ المتعددة حتى تمكن جميع المستخدمين لهذا الخادم من استخد          
 واحد فقط، فلن يستطيع إلا ان يخـدم عميـل   مسار تنفيذيوجد به إلا  طابور، فلو كان برنامجك لا    

Client العملية الحالية التي يقوم بها البرنامج انتهاء واحد في وقت واحد، مما يضطر أي مستخدم 
 العـصر مـن تطـوير    هذاوهو بلا شك، شيء مرفوض في . لخدمة مستخدم آخر ومن ثم خدمته 

  .التطبيقات
مع ذلك، التلاعب بمسارات التنفيذ في شيفراتك المصدرية فيه شيء كبيـر مـن الخطـر                

ليس هذا فقط، بل ان إدارة مسارات التنفيذ في برامجـك  . والحصول على نتائج سلبية غير متوقعة 
ة من المسارات التي    عملية معقدة وذلك بسبب ان التنفيذ لا يتبع مسار منطقي واحد، وانما مجموع            

حيث ان نجاح تنفيذ البرنامج في المـرة  . تجعلك تعيد النظر عشرات المرات قبل اعتماد برنامجك      
  !الأولى، لا يشترط تنفيذه بنجاح في التجربة الثانية أو الثالثة أو حتى العاشرة

  
  أنواع مسارات التنفیذ

 Free Threadingيـذ الحـرة   مـسارات التنف : مسارات التنفيذبصفة عامة، يوجد نوعان من 
في مسارات التنفيذ الحرة، يكون كـل  . Apartment Threadingمسارات التنفيذ المتباعدة و

 التي Global Variable جميع بيانات البرنامج والمتغيرات العامة إلى قابل للوصول مسار تنفيذ

 المـسارين هـذين  ، فكـلا  Two Threadsمسارين تنفيذين فلو كان البرنامج يحتوي على . فيه
  . جميع المتغيرات العامة للبرنامجإلىيستطيعان الوصول 

 ان التعامل معـه يتطلـب       إلا يعطيك مرونة كبيرة،     مسارات التنفيذ رغم ان هذا النوع من      
 وذلك بسبب التعارضات التي     ، التي به صعبة الاستكشاف    والأخطاءمهارة كبيرة، وكثرة الشوائب     

 فكرة عن مثـل  ولإعطائك. لمختلفة على المتغيرات العامة في البرنامج امسارات التنفيذقد تحدثها   
، ولهذا البرنامج X ان لدينا برنامج يحتوي على متغير عام باسم -مثلا–افترض  هذه التعارضات،   

  :ه الشيفرة، ويحتوي البرنامج في احد سطوره على هذB و A هما مسارين تنفيذين
 
10 If X < 0 Then 
20     ArabicConsole.WriteLine ("اقل من الصفر X قیمة المتغير") 
30 End If 

  
 اقـل مـن صـفر،    X جعل قيمة المتغير A مسار التنفيذوالان راقب هذا السيناريو، نفترض ان       

 اصغر من صفر، وقبل ان Xلان قيمة - سيحقق الشرط 10 السطر إلى مسار التنفيذوعندما وصل 
 قيمة إلى X بزيادة قيمة المتغير )Bوهو ( الثاني مسار التنفيذ  قام20 السطر إلى مسار التنفيذينتقل 

 وسـيظهر الرسـالة     الشيفرة، سينفذ   20 السطر   إلى Aمسار التنفيذ   عندما يصل   ف ،اكبر من صفر  
  .وهي بلا شك رسالة خاطئة المنطق والتوضيح X والخاصة بقيمة المتغير

 تخـصيص المتغيـرات   ا يتمهفي فApartment Threadingاما مسارات التنفيذ المتباعدة 
 ستكون له نـسختان،  -في المثال السابق - Xفالمتغير العام   . مسار التنفيذ العامة في البرنامج لكل     

 مما يضمن عدم حدوث التعارضـات فـي   ،B ونسخة أخرى لمسار التنفيذ Aلمسار التنفيذ نسخة  
 يمكنك من إنشاء مسارات Visual Basic .NET في .لمسارات التنفيذ المتعددةالمتغيرات العامة 

 حتى لا تـسبب التعـارض بـين    Synchronizationالتـزامن  تنفيذ حرة، ولكنك ستحتاج إلى  
  .كما سترى لاحقا في هذا الفصل–المتغيرات العامة 

  
  متى تستخدم مسارات التنفیذ المتعددة؟

معالجة، وذلك بسبب  تزيد من عملية سرعة ال المتعددةون ان مسارات التنفيذدمعظم المبرمجين يعتق
قد يكون هذا الكلام صحيح في حالات معينة وخاصة        .  في وقت واحد    المتعددة قابليتها لتنفيذ المهام  

، لأن كل معالج سيقوم بتنفيذ تعليمـات  Processorفي الاجهزة التي تحتوي على أكثر من معالج         
لجهاز على معالج واحـد  لكن في معظم الاحوال، سيحتوي ا.  بشكل مستقل عن الآخر مسار التنفيذ 

لن تكون هنالك     يقوم بها معالج واحد فقط، وبالتالي      مسارات التنفيذ فقط، أي ان عملية تنفيذ جميع       
  .سرعة اضافية

 في ا سرعة المعالجة، لانه   إبطاء  في سببت قد   مسارات التنفيذ المتعددة   ان لا تعتقد ان      أرجو
 الحالتينن تيوسنأخذ ها.  على نوعية وكثرة المهامنقصها، لكن القضية تعتمدتزيدها ولا تالحقيقة لا 

  :بينان لك عيوب ومزايا مسارات التنفيذ المتعددةتان تال
 ثواني،  10برنامج سيقوم بتنفيذ مهمتين، الواحدة منها تستغرق        النفترض ان   : ى الأول الحالة

 ثواني 10 تستغرق  سيقوم بتنفيذ المهمة الأولى التيSingle Threadفلو كان مسار التنفيذ احادي 
 ثانيـة وسـيكون   20 ثواني ويصبح المجموع للمهمتـين       10ومن ثم المهمة الثانية التي تستغرق       

 مسارات التنفيذ متعـددة، فـان       كانتاما لو   .  ثانية لكل مهمة   15 = 2÷  ) 10 + 20( المتوسط  
 لكـل مهمـة    ثانية20 = 10 + 10المهمتان سيتم انجازهما بشكل متوازي، ويكون المعدل دائما         

 الأحاديـة ، مما يبين لنا ان مسارات التنفيذ )10طبعا لو وجد معالجين في الجهاز سيكون المعدل      (
  .افضل من مسارات التنفيذ المتعددة

 ثواني والثانية تستغرق ثانيـة      10ن، الأولى تستغرق    ينفترض ان لدينا مهمت   : ة الثاني الحالة
 1في وقت يتراوح ما بـين       ة الواحدة   عملية تنفيذ المهم  يتم  س،  الأحاديةفي مسارات التنفيذ    . واحدة

اما مع .  ثانية لكل مهمة6 = 2÷  ) 1 + 11( ثانية، مما ينتج عنه معدل مقارب إلى     11ثانية إلى   
 ثانية مما 2 ثانية إلى 1غرق وقت يتراوح ما بين تمسارات التنفيذ المتعددة، فان المهمة الثانية ستس     

وبالتالي يتضح إلـى ان مـسارات التنفيـذ    . للمهمة الثانية  ثانية   1.5 يؤدي إلى معدل مقارب إلى    
  .الأحاديةالمتعددة افضل من مسارات التنفيذ 

 مسارات التنفيـذ   افضل من    ت ليس مسارات التنفيذ المتعددة  ان  : ، ننهي هذه الفقرة ونقول    اًذإ
 أستطيع.  ونوع الحاجةعليك ان تحدد النوع المناسب في الحالة المناسبة      . والعكس صحيح الاحادية  

 المهـام   لإنجـاز  افضل في حال كون الوقت المستغرق        ي ه مسارات التنفيذ المتعددة  ان اقول ان    
 اذا كنت تود من القيام بعملية مسارات التنفيذ المتعددةوقد تحتاج إلى . مختلف من مهمة إلى أخرى 

لو تلاحظ طريقة عمـل     ف.  دون ان توقف عمل المستخدم     Backgroundتنفيذ مهمات في الخلفية     
، ستكتشف انه Microsoft Word التابع لبرنامج Auto Spell-Checker الفوري الإملاءمدقق 

 خاص ومستقل مما يجعلك قادرا على الكتابة في أي وقت تريده بينما يقـوم               مسار تنفيذ يعمل في   
  . بنفس الوقت على المستند في الخلفيةالإملائيالمدقق بالتدقيق 

    ملاحظة

  
كلمـا زاد عـدد مـسارات       : حكمة برمجیة ضعھا دائما فـي عـین اعتبـارك         

  .التنفیذ في شیفراتك، كلما زاد تعقیدھا وصعوبة اكتشاف أخطائھا
 

  
  إنشاء مسار تنفیذ

 يتطلب Thread، مسار التنفيذ ما هو إلا كائن منشأ من الفئة NET Framework.في إطار عمل 
لابـد ان لا يحتـوي       (AddressOfام الكلمة المحجوزة     باستخد Delegateمشيده إجراء مفوض    

  :، راقب هذا المثال)Parametersالإجراء على أية وسيطات 
 

Sub Main() 
    Dim counter As Integer 
    Dim newThread As New Thread(AddressOf DoSomething) 
 
    newThread.Start() 
 
    For counter = 1 To 1000 
        Console.Write("$"c) 
    Next 
 
End Sub 
 
Sub DoSomething() 
    Dim counter As Integer 
 
    For counter = 1 To 1000 
        Console.Write("*"c) 
    Next 
End Sub 

  
  :مخرجات الشيفرة السابقة، ستكون شيئا شبيها بـ

 
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$********************************* 
******************************************************$$$$$$$$$$$$$$ 
$$$$$$$$$$$$$$$$$$$$$$$$$$$$************************$$$$$$$$$$$$ … 
 
 

    ملاحظة
  

ــائن     ــسابق اســتخدام الك ــال ال ــي المث  عوضــا عــن  Consoleتعمــدت ف
ArabicConsoleحقا في ھذا الفصل، وذلك لمشاكل ستفھمھا لا.  

، وبعد ذلـك  newThreadفي الشيفرة السابقة، قمت بإنشاء مسار تنفيذ جديد وأسندته إلى المؤشر    
وهـو  ( ليتم تشغيل مسار التنفيذ وتنفيذ الإجـراء المفـوض لـه           ()Startقمت باستدعاء الطريقة    

DoSomething()(                اء ، من الضروري ان أذكرك هنـا بـان الـسطور التـي تلـي الاسـتدع
newThread.Start()سيتم تنفيذها أيضا في نفس الوقت .  

  
  الطرق والخصائص

يمكنك استدعاء .  توقف عمله()Abort تشغل مسار التنفيذ، فان الطريقة ()Startإن كانت الطريقة 
 Thread والتي تعود بكائن من النـوع        Thread.CurrentThreadالطريقة من خلال الخاصية     

  :الييمثل مسار التنفيذ الح
… 
… 
Sub DoSomething() 
    Dim counter As Integer 
 
    For counter = 1 To 1000 
        If counter = 100 Then 
            Thread.CurrentThread.Abort() ' ایقاف مسار التنفیذ الحالي 
        End If 
        Console.Write("*"c) 
    Next 
End Sub 

  
 توقف عمل مسار التنفيذ وتنفيه من الحياة بشكل نهـائي، ولكـن توجـد      ()Abortالطريقة  

، والتي قد تحتاج    ()Suspendطريقة أخرى توقف عمل مسار التنفيذ بشكل مؤقت ولا تنهيه وهي            
  : لتكمل عمل مسار التنفيذ()Resumeإتباعها بالطريقة 

 
Dim newThread As New Thread(AddressOf DoSomething) 
… 
… 
 
 تشغیل مسار التنفیذ '
newThread.Start() 
… 
… 
 
 ایقافه بشكل مؤقت '
newThread.Suspend() 
… 
… 

)بعد الایقاف المؤقت(متابعة التنفیذ  '  
newThread.Resume() 
… 
… 
 
 ااء التنفیذ بشكل ائي '
newThread.Abort() 
… 
… 

  
لفتـرة  ) بشكل مؤقت (نفيذ   تمكنك من إيقاف عمل مسار الت      ()Sleepالمزيد أيضا، الطريقة    

  : ثانية0.001زمنية وحدتها 
 
Thread.CurrentThread.Sleep(500) ' لمدة نصف ثانیةایقاف مسار التنفیذ الحالي  
 

، والتي توقف عمل مسار التنفيذ الحالي بشكل مؤقت حتى    ()Joinمن الطرق أيضا الطريقة     
  :ينتهي تنفيذ مسار التنفيذ الآخر

 
Dim newThread As New Thread(AddressOf DoSomething) 
… 
… 
 
 تشغیل مسار التنفیذ '
newThread.Start() 
 
أیضاسیتم تنفیذ هذه الشیفرة  '  
… 
… 
 
newThread.Join () 
 لن یتم تنفیذ هذه الشیفرة حتى '
 يموت مسار التنفیذ الثاني '
… 
… 
 

يد التنفيذ بالاسـتعلام  اما الحديث عن الخصائص، فتستطيع معرفة ما اذا كان مسار التنفيذ ق        
  :()Join تطبيقها في الشيفرة السابقة عوضا عن الطريقة -مثلا–، ليمكنك IsAliveعن الخاصية 

 
 
Dim newThread As New Thread(AddressOf DoSomething) 
… 
… 
 
 تشغیل مسار التنفیذ '
newThread.Start() 

أیضاسیتم تنفیذ هذه الشیفرة  '  
… 
… 
 
 
Do While newThread.IsAlive() = True 
 
Loop 
 لن یتم تنفیذ هذه الشیفرة حتى '
 يموت مسار التنفیذ الثاني '
… 
… 

 
    ملاحظة

  
  : لمسار التنفیذ الحاليIsAliveمن الغباء الاستعلام عن الخاصیة 

 
If Thread.CurrentThread.IsAlive Then … 

  
  . دائما لأسباب منطقیةTrueفھي ستعود بالقیمة 

 
  

 قابلة للقراءة والكتابة، وفي الحقيقة ليس String خاصية حرفية من النوع     Nameة  الخاصي
لها أي تأثير عملي على مسار التنفيذ لحظـة التنفيـذ، ولكنهـا تفيـدك كثيـرا لحظـة التنقـيح                 

Debugging حيث يظهر اسم مسار التنفيذ بتلك الخاصية في شريط الأدوات ،Debug location 
  :)1-10شكل (

 
… 
… 
Thread.CurrentThread.Name = "مسار التنفیذ الرئیسي" 
newThread.Name = "مسار تنفیذ البحث عن الملفات" 

  

  
  .Debug Location ظهور أسماء مسارات التنفيذ في شريط الأدوات :1-10شكل 

 من معرفة حالة مسار التنفيذ، فالشرط التالي        ThreadStateالمزيد أيضا، تمكنك الخاصية     
  :الة مسار التنفيذ ما ان تم تشغيله أو لايختبر ح

 
If newThread.ThreadState = ThreadState.Unstarted Then … 

  
  :بقية القيم للحالات الأخرى التي يكون عليها مسار التنفيذ يعرضها الجدول التالي

  
  الحالة  القيمة

ThreadState.Aborted  انهي مسار التنفيذ بشكل نهائي.  
ThreadState.AbortRequested  تم طلب ايقاف لمسار التنفيذ.  

ThreadState.BackGround  مسار تنفيذ خلفيBackGround 

Thread.  
ThreadState.Running  يعمل.  

ThreadState.Suspended الإيقاف المؤقت.  
ThreadState.SuspendedRequested تم طلب إيقاف مؤقت لمسار تنفيذ.  

ThreadState.UnStarted بدأ بعدلم ي.  
ThreadState.WaitSleepJoin   تم استدعاء الطريقةJoin على مسار 

  .التنفيذ
  

  :Priorityالأسبقية خاصية 
، Multitasking المهام متعدد  هو نظام تشغيل Windowsأنت تعلم وانا اعلم ان نظام التشغيل  

ى النظام توزيـع     من تشغيل أكثر من برنامج في وقت واحد، الامر الذي يفرض عل            يمكنكبحيث  
الاصدار  (Windowsفي الإصدارات السابقة من نظام التشغيل       . المعالجة بين التطبيقات المختلفة   

Windows MEوقت المعالجة ، كانت تتم عملية توزيع ) وما قبلهProcessing Time   بـشكل
الأوامـر  ، ووقت المعالجة ما هي إلا كمية        Processesمتوازي ومتساوي لكافة البرامج العاملة      

 بـرامج  10فلو وجدت . والتعليمات التي يقوم المعالج بتنفيذها في فترة زمنية معينة لبرنامج واحد      
  .عاملة في الذاكرة، سيقوم المعالج بتنفيذ جزء معين من كل برنامج لفترة متساوية

د التوزيع السابق لوقت المعالجة عادل لدرجة كبيرة، ولكن صفة العدل التي من شمائله لم تع 
قد تكـون   ( البرامج العاملة في وقت واحد       -ان لم يكن مئات   –محبذة، وذلك بسبب وجود عشرات      

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


 NET Framework  358.إطار عمل : الثانيالجزء 
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

البرنـامج  ، الأمر الذي يسبب بطء كبير في عمليـة تنفيـذ   )معظمها غير نشطة أو غير مستخدمة  
  ).البرنامج الحالي ( Active Processالنشط 

الأسبقية  هو بإنتاج فلسفة     Windowsلذلك، كان الحل الذي اعتمدوه مطورو نظام التشغيل         
Priority ، والتي ظهرت مع الإصداراتWindows NT والتي تليها بحيث تعطي وقت معالجة 

لذلك، حتى لو وجدت مئات البرامج تعمل في الخلفية، فلن يعطيهـا نظـام              . أكثر للبرنامج النشط  
ئج ايجابية جدا للبرنامج الحالي التشغيل حقها الكافي من وقت المعالجة، الامر الذي يترتب عليه نتا     

  .مما يزيد من كفاءة تنفيذه
 فلسفة الأسبقية بشكل افتراضـي،      Windowsتطبق الإصدارات الجديدة من نظم التشغيل       

 Processing فـي القـسم   Programsويمكنك التحقق من ذلك بنفسك ان تم اختيـار الامـر   

Scheduling في صندوق الحوار Performance Optionلذي تصل اليـه عـن طريـق    ، وا
Control Panel -> System -> Advanced -> Performance -> Settings ) 10شكل-

2.(  
  

  
  . نظام التشغيل يعطي وقت معالجة أكثر للبرامج النشطة:2-10شكل 

يمكنك نظام التشغيل من زيادة أو إنقاص وقت المعالجة لكل برنامج عامـل، توجـد سـت       
، Low  ،BelowNormal: ابتداء من الأقل حتى الأكثـر     -لمعالجة، هي   مستويات لمقدار وقت ا   

Normal  ،AboveNormal  ،High   و ،Realtime .       يمكنك التحكم فيها لكل برنامج عامل عن
 والذي تصل اليه بالضغط على المفـاتيح  Windows Task Managerطريق صندوق الحوار 
[Alt]+[Ctrl]+[Del] ) تالية في أعلى الصفحة ال3 -10شكل.(  

  
    ملاحظة

  
تغیر مقدار وقت المعالجة للبرامج العاملة يدويا بنفسك، قد يـؤدي إلـى              

  .نتائج سلبیة غیر متوقعة
  

  
 في Threadsبعد هذه الفلسفة الطويلة حول الأسبقية، أردت ان أبين لك ان مسارات التنفيذ           

 لها مختلف من مسار تنفيذ       يمكن ان يكون وقت توزيع المعالجة      Processالبرنامج العامل الواحد    
، والتـي  Priorityإلى آخر، بل وتستطيع أيضا التحكم بمقدار وقت المعالجة عن طريق الخاصية        

،  ThreadPriority.BelowNormal:يمكنك إسـناد قيمــة لهـا مــن خمـس قـيم هـي       
ThreadPriority.Lowest ،ThreadPriority.Normal،ThreadPriority.AboveNormal 

  .ThreadPriority.Highest، أو 

  .غيير مقدار وقت المعالجة لبرنامج عاملت :3-10شكل 

  
دعني اريك هاتين الشيفرتين والتي قد توضح لك مدى تأثير توزيع وقـت المعالجـة بـين           

بشكل افتراضي قيمة (مسارات التنفيذ، الشيفرة الأول تكون فيها أسبقية كلا مسارات التنفيذ متساوية 
  ): لكل مسار تنفيذThreadPriority.Normal هي Priorityالخاصية 

 
Sub Main() 
    Dim counter As Long 
    Dim newThread As New Thread(AddressOf DoSomething) 
    newThread.Start() 
 
مقدار عداد الحلقةأو نقصان  زیادة إلىقد تحتاج  '      
 على حسب سرعة المعالج عندك '    
    For counter = 1 To 200000000 
    Next 
 
    Console.WriteLine("THREAD A HAS FINISHED") 
 
End Sub 

Sub DoSomething() 
    Dim counter As Integer 
 
    For counter = 1 To 100 
    Next 
    Console.WriteLine("THREAD B HAS FINISHED") 
End Sub 
 

المسارين، فانه من البديهي سينتهي تنفيذ حلقة مـسار التنفيـذ      بما ان وقت المعالجة متساوي لكلا       
، لتكون  100الثاني قبل الأول، وذلك لان عداد الحلقة التكرارية في مسار التنفيذ الثاني لا يتجاوز               

  :المخرجات بهذا الشكل
 
THREAD B HAS FINISHED 
THREAD A HAS FINISHED 

  
 وقـت المعالجـة لمـسار التنفيـذ الأول     -مثلا– ونزيد لنحاول تغيير أسبقية كلا مسارات التنفيذ،   

  :وننقصها من الثاني
 
Sub Main() 
    Dim counter As Long 
    Dim newThread As New Thread(AddressOf DoSomething) 
 
 تعدیل الاسبقیة '    
    Thread.CurrentThread.Priority = ThreadPriority.Highest 
    newThread.Priority = ThreadPriority.Lowest 
 
    newThread.Start() 
 
  زیادة مقدار عداد الحلقةإلىقد تحتاج  '    
 على حسب سرعة المعالج عندك '    
    For counter = 1 To 200000000 
    Next 
 
    Console.WriteLine("THREAD A HAS FINISHED") 
 
End Sub 
 
Sub DoSomething() 
    Dim counter As Integer 
 
    For counter = 1 To 100 
    Next 
    Console.WriteLine("THREAD B HAS FINISHED") 
End Sub 

عند تنفيذك للشيفرة السابقة، ستفاجئ إن اكتشفت أن مسار التنفيذ الأول قد أنهى حلقته التكراريـة                
ني، وذلك لأن وقت المعالجة المعطـى  قبل حلقة مسار التنفيذ الثا   )  مليون مرة  200والتي عدادها   (

 أكثر بكثير من وقت المعالجة المعطى لمـسار         ThreadPriority.Highestلمسار التنفيذ الأول    
  :، لتكون المخرجات بهذا الشكلThreadPriority.Lowestالتنفيذ الثاني 

 
THREAD A HAS FINISHED 
THREAD B HAS FINISHED 
 

    ملاحظة
  

خیرين استخلصتھا من تنفیذ الشیفرة على جھـازي        نتائج الاختبارين الأ  
، ولا اضمن لـك  Pentium 800 MHzالشخصي ذو معالج قديم من النوع 
  .نتائج متطابقة مع جھازك الشخصي

  
  

  :IsBackgroundخاصية الخلفية 
مـسارات  سيستمر برنامجك في العمل ما دامت مسارات تنفيذه على قيد الحياة، بغض النظر عن               

 حيث سيتم إنهائها بشكل تلقائي لحظة مـوت جميـع   Background Threadsلخلفية التنفيذ ا
  .مسارات التنفيذ العادية

 أو جعله خلفي عـن  Background Threadيمكنك معرفة ما اذا كان مسار التنفيذ خلفي 
  ):وهي قابلة للقراءة والكتابة (IsBackgroundطريق الخاصية 

 
Dim newThread As New Thread(AddressOf DoSomething) 
… 
… 
newThread.IsBackground = True 
… 
… 
If newThread.IsBackground Then 
    newThread.Abort() 
End If 

  
مسارات التنفيذ الخلفية هي نفس مسارات التنفيذ العادية، ويكمن الفرق الوحيـد بينهمـا ان            

 ان ماتت جميع مسارات التنفيذ الرئيسية في مسارات التنفيذ الخلفية لا تبقي البرنامج على قيد الحياة
وذلك بسبب وجود مسار التنفيـذ      (فمثلا، الشيفرة التالية ستبقي البرنامج على قيد الحياة         . البرنامج

  ):الثاني

Module Module1 
    Sub main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
        newThread.Start() 
    End Sub 
 
    Sub DoSomething() 
        Dim counter As Integer 
         
 سیتم تنفیذ كامل الحلقة '        
        For counter = 1 To 1000 
            Console.WriteLine(counter) 
        Next 
 
        Console.Read() 
    End Sub 
End Module 

  
، فسيتم إنهائه بمجـرد  Background Threadار التنفيذ الثاني من النوع الخلفي اما ان كان مس

  :انتهاء تنفيذ مسار التنفيذ الأول
 
Module Module1 
    Sub main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
 جعله مسار تنفیذ خلفي '        
        newThread.IsBackground = True 
        newThread.Start() 
    End Sub 
 
    Sub DoSomething() 
        Dim counter As Integer 
         
  تنفیذ كامل الحلقةلن یتم '        
        For counter = 1 To 1000 
            Console.WriteLine(counter) 
        Next 
 
        Console.Read() 
    End Sub 
End Module 

    ملاحظة
  

 عند تجربة الشیفرة السابقة، ArabicConsoleلا تحاول استخدام الكائن 
 يحتـوي علـى     ArabicConsoleوذلك لان البرنامج لن ينتھي بـسبب ان         

  .مسار تنفیذ خاص به يبقي البرنامج على قید الحیاة
 

  
  التعامل مع مسارات التنفیذ

 معرفة ان مسارات التنفيذ تعمل كالبرامج المستقلة، فيمكنك اعتبار ان كل مسار بادئ ذي بدء عليك
 في مسار تنفيـذ، فلـن تتـأثر         Exceptionتنفيذ هو برنامج عامل، لدرجة انه لو حدث استثناء          

  :مسارات التنفيذ الأخرى وستكمل عملها كان شيئا لم يحدث
 
Module Module1 
    Sub main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
      
        newThread.Start() 
        … 
        … 
 سیتم استكمال تنفیذ الشیفرة التالیة '        
 رغم وجود استثناء في مسار التنفیذ الثاني '        
        … 
        … 
    End Sub 
 
    Sub DoSomething() 
        Dim counter As Integer 
 
        Throw New Exception() 
        … 
        … 
    End Sub 
End Module 

  
، إلا ان مسار التنفيذ الرئيـسي  Throw Exceptionرغم ان مسار التنفيذ الثاني قد رمى استثناء 

 فـي  Catch Exceptionمع ذلك، أنصحك بشدة بتفادي الاسـتثناء  . سيكمل عمله بشكل طبيعي
مسار التنفيذ الثاني أو إيقاف عمل مسار التنفيذ الرئيسي حتى لا تنمو الشوائب والأخطاء في عمل                

  .البرنامج

    ملاحظة

  
لا تحاول تدارك الاستثناءات بـین مـسارات التنفیـذ المختلفـة فـذلك لـن         

  :يفیدك، فلو حاولت كتابة شیئا مثل
 
Dim newThread As New Thread(AddressOf DoSomething) 
 
Try 
    newThread.Start() 
Catch 
    … 
End Try 
 

 Try … End Tryفانك تتعب نفـسك دون جـدوى، حیـث وجـود التركیـب      
  .كعدمه ان كان الاستثناء قد رمي من مسار تنفیذ آخر

 
  

 تختلف من مسار Regional Settingsليس هذا فقط، بل حتى الاعدادات الإقليمية الحالية 
 وهي خاصية   CurrentCultureآخر، والدليل ان كل مسار تنفيذ يحتوي على الخاصية          تنفيذ إلى   
  : تمكنك من تحديد الاعدادات الإقليمية لمسار التنفيذ الحاليCultureInfoمن النوع 

 
Imports System.Globalization 
… 
… 
Module Module1 
    Sub main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
        Thread.CurrentThread.CurrentCulture = New _  
            CultureInfo("ar-SA") 
        newThread.CurrentCulture = New _  
            Globalization.CultureInfo("ar-EG") 
        newThread.Start() 
 
        MsgBox(100.ToString("C")) ' 100.00 رس 
    End Sub 
 
    Sub DoSomething() 
        MsgBox(100.ToString("C")) '  ج م 100.000 
    End Sub 
End Module 

    انظر أيضا
  

موضوع الاعدادات الإقلیمیة حديث ذو شـجون لمحـت إلیـه سـابقا فـي              
  .الفئات الأساسیةالفصل السادس 

 
  

 و ()Abortهائي والإيقاف المؤقت باستخدام الطـرق      على صعيد آخر، عمليات الإيقاف الن     
Suspend() نقطة آمنـة   لا تتم فورا لحظة استدعائها، وانما حتى يصل مسار التنفيذ إلىSafe 

Point والنقاط الآمنة ،Safe Points -  استنادا إلى مراجـع.NET Documentation-  هـي 
مثلا، عندما ينهي إجراء . تبدأ عملها ان Garbage Collectionلحظة من الوقت يمكن للمجموعة 

 Safe تحت العنوان NET Documentation.راجع فهرس مستندات (معين تنفيذه ويعود بقيمة 
Pointsلمزيد من التفاصيل حول النقاط الآمنة .(  
  

    ملاحظة
  

كانت أمنیتي عرض مثالا يوقف مسار تنفیذ وھو لیس في لحظة نقطـة          
 لست متأكدا حتى ھذه لحظة      –ق فضیلة   والاعتراف بالح -ولكني  . آمنة

لـذلك، دعنـي اكـرر كلمتـي الاخیـرة      . من مدى استیعابي للنقاط الامنة 
  . لعل وعسى تطلع منھا بشيءMSDNوأنصحك بالانتقال إلى مكتبة 

 
  

اما الحديث عن الأسبقية، فأنصحك بشدة عدم المحاولة بتعديلها وتغييرها بنفسك إلا ان دعت    
صفة عامة، يفضل إعطاء وقت معالجة اكبر لمسارات التنفيذ التي تتعامل مـع  ب. حاجة ماسة لذلك  

المخرجات الظاهرية على الشاشة، حتى تتمكن من تنفيذ شيفرات العرض على المستخدم بأسرع ما 
  .يمكن

  
  متى يموت مسار التنفيذ؟

، ()Abortالطريقة الأولى عندما يتم إيقافه باستخدام : يموت مسار التنفيذ في حالة من ثلاث حالات
، والثالث عنـد    )الذي يعتبر نقطة البداية لمسار التنفيذ     (الثانية عندما ينتهي عمل الإجراء المفوض       

  . في الإجراء المفوضExit Subتنفيذ العبارة 

 Background Threadsالحالات الثلاث السابقة تطبق أيضا على مسارات التنفيذ الخلفية 
 انها تموت إن ماتت جميع مسارات التنفيذ -كما أخبرتك–ة لها، وهي ولكن توجد حالة رابعة إضافي

  .العادية في البرنامج بشكل تلقائي
حالة غريبة حدثت لي عندما لعبت قليلا مع مسارات التنفيذ لم أجد تفسيرا لها حتـى هـذه                  

، )يـسي مسار التنفيذ الرئ  ( من أول مسار تنفيذ للبرنامج       ()Abortاللحظة، وهي استدعاء الطريقة     
  :فالشيفرة التالية

 
Module Module1 
    Sub main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
        newThread.Start() 
 
        Thread.CurrentThread.Abort() 
    End Sub 
 
    Sub DoSomething() 
        … 
        … 
    End Sub 
End Module 

  
ستنهي تنفيذ البرنامج بأكمله رغم وجود مسار تنفيذ آخر على قيد الحياة، فعندما جعلت مسار التنفيذ 

  :()Abortالرئيسي يموت دون استخدام 
 
Module Module1 
    Sub main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
 
        Thread.CurrentThread.Abort() 
    End Sub 
 
    Sub DoSomething() 
        … 
        … 
    End Sub 
End Module 

  
ولكن بعد عبث في أدوات التنقيح التابعة لبيئة        . تم إكمال عمل مسار التنفيذ الثاني دون أي مشاكل        

Visual Studio .NET   اكتشفت ان مسار التنفيذ الرئيسي لم يمت رغم عدم وجـود أي شـيفرة 
  !ة قيد التنفيذمصدري

أستطيع ان استنتج من هذه التجربة الحالة الخامسة التي تموت فيها مسارات التنفيذ الخلفيـة    
  .للبرنامج) الأول(وهي عندما يموت مسار التنفيذ الرئيسي ) الرابعة لمسارات التنفيذ العادية(

  
  مشاركة البیانات

، Free Threadingرات تنفيـذ حـرة    هي مساThreadمسارات التنفيذ التي أنشأناها من الفئة 
بحيث يمكن لكل مسار تنفيذ من مشاركة البيانات في نفس البرنامج العامل، وهذه الـشيفرة خيـر           

  :دليل
 
Public X As Integer ' متغير عام 
 
Sub Main() 
    Dim counter As Long 
    Dim newThread As New Thread(AddressOf DoSomething) 
 
    newThread.Start() 
 
    For counter = 1 To 1000 
        X += 1 
    Next 
    newThread.Join() 
 
    Console.WriteLine("X = " & X) ' 2000 
End Sub 
 
Sub DoSomething() 
    Dim counter As Integer 
 
    For counter = 1 To 1000 
        X += 1 
    Next 
End Sub 

  
ا مشكلة من مشاكل مشاركة البيانات بين مسارات التنفيذ المتعددة، ودعني           وضحت لك سابق  

أذكرك هنا بان المشكلة اخطر من كونها مشاركة المتغيرات العامة، بل قد تشمل كـل نـشاطات                 
الشيفرات المصدرية والكائنات الأخرى التي تستخدمها كمشاركة الملفات، الأجهزة والعتاد، مواقع           

حتـى تتقـي شـر      Synchronizationالتزامن   لذلك، ستضطر إلى استخدام      .الخ....الذاكرة،  
  .التعارضات التي قد تحدثها لك مسارات التنفيذ المتعددة

  المتغیرات المحلیة الدينامیكیة
) Lifetime وعمرهـا  Visibilityمن منطلق قابليـة رؤيتهـا       ( جميع المتغيرات بكافة أنواعها     
 انها قابلة للوصول من جميع -في هذا السياق–اعني بكلمة مشتركة مشتركة بين مسارات التنفيذ، و

 Localمع ذلـك، المتغيـرات المحليـة الديناميكيـة     . مسارات التنفيذ التي تعرفها في البرنامج

Dynamic Variables هي الوحيدة التي لا يتم تشاركها بين مسارات التنفيذ المتعددة، تحقق من 
  :هذه الشيفرة

 
Sub Main() 
    Dim newThread As New Thread(AddressOf DoSomething) 
    Dim newThread2 As New Thread(AddressOf DoSomething) 
 
    newThread.Start() 
    newThread2.Start() 
 
End Sub 
 
Sub DoSomething() 
 متغير محلي دینامیكي '    
    Dim X As Integer 
 متغير محلي دینامیكي '    
    Dim counter As Integer 
 
    Do 
        counter += 1 
        X += 1 
    Loop Until counter >= 100 
 
    Console.WriteLine(X) 
End Sub 

  
  :مخرجات الشيفرة السابقة ستكون دائما وأبدا

 
100 
100 

  
متعددة، لقام كلا المـسارين  لو كانت المتغيرات المحلية الديناميكية متشاركة بين مسارات التنفيذ ال        

.  مرة لكل مسار تنفيذ    100 بحيث لا يتم تنفيذ الحلقة       counterالسابقين بالتأثير على قيمة المتغير      
  :Static ستاتيكي counter السابق، وجعلت المتغير DoSomethingفلو حاولت تعديل الإجراء 

 
 
 
 
 

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


 NET Framework  370.إطار عمل : الثانيالجزء 
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

… 
… 
Sub DoSomething() 
 متغير محلي دینامیكي '    
    Dim X As Integer 
ستاتیكيمتغير محلي  '      
    Static counter As Integer 
 
    Do 
        counter += 1 
        X += 1 
    Loop Until counter >= 100 
 
    Console.WriteLine(X) 
End Sub 
 

القيمتـين  ) لقـة والذي اقصد به عدد مرات التكرار الحقيقية التي تمت للح (Xفسيظهر لنا المتغير   
  :التاليتين

 
100 
1 

 
إن سألتني عن سبب عدم تشارك المتغيرات المحلية الديناميكية بين مسارات التنفيذ المتعددة، 
فستكون إجابتي هي ان هذا النوع من المتغيرات يتم إنشائه لحظة التصريح عنها في داخل الإجراء 

، مما يعنـي ان     Stackل مؤقت في ذاكرة     عند استدعائه من قبل مسار التنفيذ، وسيتم حفظها بشك        
 نسخة خاصة من المتغيرات المحلية الديناميكية، لذلك لـن يـتم       -يدخل الإجراء -لكل مسار تنفيذ    

  .مشاركتها
  

  ThreadStatic Attributeالمواصفة 
 ThreadStatic Attributesإلى جانب المتغيرات المحلية الديناميكية، يمكنك استخدام المواصفة 

  :لمتغيرات العامة أو على مستوى الوحدة لمنع مشاركتها بين مسارات التنفيذ المتعددةعلى ا
 
 

Module Module1 
    Dim X As Integer 
    <ThreadStatic()> Dim Y As Integer 
    Sub Main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
        X = 100 
        Y = 100 
        newThread.Start() 
        newThread.Join() 

        Console.WriteLine(X) ' 200 
        Console.WriteLine(Y) ' 100 
 
    End Sub 
 
    Sub DoSomething() 
        X = 200 
        Y = 200 
    End Sub 
End Module 

  
 وتجنب عشرات المـشاكل  ThreadStatic Attributeقد يسرك كثيرا استخدام المواصفة 

والأخطاء التي تحدث بسبب التعارض على المتغيرات من قبل مسارات التنفيذ المتعـددة، ولكـن              
 لا تعمـل بـشكل صـحيح إلا مـع     ThreadStatic Attributeيؤسفني إخبارك ان المواصفة 

 أو في الفئات Modulesبرمجية اقصد هنا المتغيرات المعرفة في الوحدات ال(المتغيرات المشتركة 
Classes     باستخدام الكلمة المحجوزة Shared(      فعند تجربتي للشيفرة المصدرية التالية، أصـبت ،

  :Xبخيبة أمل كبيرة بعد رؤيتي لقيمة المتغير 
 

Class TestClass 
 لن یفیدك استخدام المواصفة هنا '    
    <ThreadStatic()> Public X As Integer 
    ' Shared ستعمل بكفائة على هذا الحقل لانه مشترك 
    <ThreadStatic()> Public Shared Y As Integer 
End Class 
 
Module Module1 
    Dim TestObject As New TestClass() 
    Sub Main() 
        Dim newThread As New Thread(AddressOf DoSomething) 
 
        TestObject.X = 100 
        TestObject.Y = 100 
 
        newThread.Start() 
        newThread.Join() 
 
        Console.WriteLine(TestObject.X) ' 200 
        Console.WriteLine(TestObject.Y) ' 100 
 
    End Sub 
 
    Sub DoSomething() 
        TestObject.X = 200 
        TestObject.Y = 200 
    End Sub 
End Module 

  TLSوحدة التخزين المحلیة 
 Thread Localوحدة تخزين محلية كل مسار تنفيذ تنشئه في شيفراتك المصدرية، يحمل معه 

Storage) TLS(             بحيث تمكن وحدة التخزين هذه مسار التنفيذ من حفظ بياناته الخاصة والغير ،
 Dataر بيانـات  جح تسمى TLSكل قيمة تحفظها في . رىقابلة للمشاركة مع مسارات تنفيذ أخ

Slot ) 10-4شكل(.  
  

  
  .Data Slots كل مسار تنفيذ يحتوي على مجموعة من جحور البيانات :10-4شكل 

  
 Data أسلوبين للتعامل مع جحـور البيانـات   NET Framework.يوفر لك إطار عمل 

Slotsجحور البيانات المسماة :  هماNamed data slots ،    وجحور البيانـات غيـر المـسماة
Unnamed data slots.  

، كل كائن أو قيمة تحفظها ترفق معهـا  Named data slotsفي جحور البيانات المسماة 
 يميزها ويكون كمعرف لها، يمكنك استخدام نفس هذا الاسم في كل            Stringاسما حرفيا من النوع     

 جحر بيانـات جديـد، اسـتدعي الطريقـة        وحتى تتــمكن من إنشاء   . مرة تود استرجاع القيمة   
AllocateNamedDataSlot():  

 
  جحر بیانات مسمىإنشاء '
Thread.AllocateNamedDataSlot("Name") 

، لذلك يفضل تعريف متغير من هذا LocalDataStoreSlotالطريقة السابقة تعود بكائن من النوع 
  :النوع حتى يسهل عليك التعامل مع جحر البيانات لاحقا

 
Dim namedDataSlot As LocalDataStoreSlot 
namedDataSlot = Thread.AllocateNamedDataSlot("Name") 

  
 الـسابق، اسـتدعي الطريقـة       Nameوحتى تتمكن من إسناد قيمة حقيقة لجحر البيانات المسمى          

SetData():  
 
Thread.SetData(namedDataSlot, "تركي العسيري") 

  
  : من قراءة جحر بيانات مسمى()GetDataريقة وعلى العكس، تمكنك الط

 
Console.WriteLine(Thread.GetData(namedDataSlot)) ' تركي العسيري 

  
 السابق، فلن تتمكن من تحديد جحر البيانـات       nameDataSlotوفي حال ما ان تم فقدان المؤشر        

Name إلا باستخدام الطريقة GetNamedDataSlot():  
 
Dim namedDataSlot2 As LocalDataStoreSlot 
namedDataSlot2 = Thread.GetNamedDataSlot("Name") 
 
Console.WriteLine(Thread.GetData(namedDataSlot2)) ' تركي العسيري 

  
اخيــرا، لا تنــسى إلغــاء جحــر البيانــات عنــد عــدم الحاجــة اليــه باســتدعاء الطريقــة 

FreeNamedDataSlot():  
 
Thread.FreeNamedDataSlot("Name") 
 
 

    ملاحظة
  

 إلـى  TLSإن تم إرسال اسم جحر بیانات غیر موجود في وحدة التخزين     
، فستقوم الطريقة بإنشاء جحـر بیانـات        ()GetNamedDataSlotالطريقة  

  .جديد

 Unnamed dataاما النوع الثاني من جحور البيانات هو جحور البيانات غير المـسماة  

slots)  تسمى أيضاAllocated data store slots(  فهي مثل جحور البيانات المسماة الـسابقة ،
 للكتابة والقراءة مـن وإلـى جحـر         Stringتماما، إلا انها لا تستخدم معرفات حرفية من النوع          

  : لتميز جحر البياناتLocalDataStoreSlotالبيانات، بل ستكتفي باستخدام مؤشر من النوع 
 
Dim dataSlot As LocalDataStoreSlot 
 
dataSlot = Thread.AllocateDataSlot() 
 
Thread.SetData(dataSlot, "تركي العسيري") 
 
Console.WriteLine(Thread.GetData(dataSlot)) ' تركي العسيري 

  
  تبادل البیانات بین مسارات التنفیذ

اشـل   بين مسارات التنفيذ يعتبر أسلوب برمجي فGlobal Variablesاستخدام المتغيرات العامة 
جدا بسبب مشكلة التعارضات التي تحدثنا عنها سابقا، كما ان استخدام وحدات التخزين المحليـة               

TLS  وذلك لان كل مسار تنفيذ له وحـدة تخـزين   ( مباشرة أمر مستحيل لمشاركة البياناتTLS 
  ).خاصة به

يمكنك مثلا الحل الذي يمكننا من تبادل البيانات بين مسارات التنفيذ المختلفة برمجي بحت، ف       
البدء بتشغيل مسار تنفيذ بحيث ينتمي إلى إجراء تابع لفئة معينة وإرسال كافة البيانات المطلوبة إلى 

  :تلك الفئة، بحيث نحصر عملية تنفيذ شيفرات مسار التنفيذ في فئة معينة
 
 

Class ThreadData 
    Public ThreadName As String 
    Public Data As Integer 
 
    Sub DoSomething() 
        Console.WriteLine(Me.ThreadName & " Data = " & Data) 
    End Sub 
End Class 

  
 نريد منها ان تكون مستقلة بمسار التنفيذ الذي ينفذها، وعند إنشائك لكائن من             ThreadDataالفئة  

بدء بتشغيل مسار التنفيذ مـن  هذه الفئة، يمكنك ارسال البيانات التي تودها إلى خصائص الفئة، وال         
  :()DoSomthingالطريقة 

Module Module1 
    Sub Main() 
        Dim One As New ThreadData() 
        Dim Two As New ThreadData() 
 
        One.ThreadName = "THREAD B" 
        One.Data = 10 
        Two.ThreadName = "THREAD C" 
        Two.Data = 20 
 
        With New Thread((AddressOf One.DoSomething)) 
            .Start() 
        End With 
        With New Thread((AddressOf Two.DoSomething)) 
            .Start() 
        End With 
        Console.WriteLine("Waiting...") 
    End Sub 
End Module 

  
  :عند تنفيذي للشيفرة السابقة، ظهرت المخرجات بهذا الشكل

 
Waiting… 
THREAD B Data = 10 
THREAD C Data = 20 
 
مع ذلك، فترتيب المخرجات لا يشترط ان يكون متوافقا عند تنفيذها على جهاز آخـر، حيـث ان               

  :لشكلالتسلسل قد يختلف على حسب مزاج المعالج، وقد تكون مخرجاتك بهذا ا
 
THREAD C Data = 20 
Waiting… 
THREAD B Data = 10 

  
 السابقة هو جعل مسار التنفيذ يعمل في كائن مستقل،          ThreadDataكان الغرض من الفئة     

ولكن توجـد  . بحيث لا تؤثر شيفراته المصدرية على بيانات مسارات التنفيذ الأخرى في البرنامج           
 التابعة للفئات، فلو أضفت الحدث      Eventsعلق بالأحداث   نقطة هامة عليك أخذها بعين الاعتبار تت      

  :ThreadDataالتالي في الفئة 

Class ThreadData 
    … 
    … 
    Event ThreadFinish() 
 
    Sub DoSomething() 
        … 
        … 
        RaiseEvent ThreadFinish() 
    End Sub 
End Class 

  
  :مسار التنفيذ الرئيسي للبرنامجوحاولت قنصه من 

 
Module Module1 
    Sub Main() 
        … 
        … 
        AddHandler One.ThreadFinish, AddressOf ThreadHasBeenFinished 
        … 
        … 
    End Sub 
     
    Sub ThreadHasBeenFinished() 
        … 
        … 
    End Sub 
End Module 
 

 السابق سيتم تنفيذه من قبل مسار ()ThreadHasBeenFinishedفتأكد وثق ثقة تامة ان الإجراء      
بالرغم مـن ان عمليـة      (التنفيذ الذي أدى إلى إطلاق الحدث وليس من قبل مسار التنفيذ الرئيسي             

ذ هذه لذلك، احرص على اخ).  تمت من قبل مسار التنفيذ الرئيسيAddHandlerالقنص باستخدام 
  .المسالة بعين الاعتبار عند إطلاق الأحداث

  Thread Synchronizationالتزامن 
في هذا القسم أحاول عرض لك بضعة أساليب تمكنك من تزامن مسارات التنفيذ على الـشيفرات                

  .المصدرية حتى تتقي شر التعارضات
  

  SyncLock … End SyncLockالتركیب 
جة مسارات التنفيذ المتعددة، فقد يأتيك شعور بان كـل سـطر مـن    ان كنت من المبتدئين في برم 

مـع الأسـف    –شيفرات البرنامج المصدرية سيتم تنفيذه بالكامل من قبل مسار تنفيذ معين وهـو              
  : شعور خاطئ تماما، فلو كان لدينا الشرط التالي-الشديد

 
If X = 10 Then 
    X = 0 
Else 
    X = 10 
End If 

  
قبل مسارين تنفيذين، فلا تعتقد ان مسار التنفيذ الأول سيتحقق من الشرط ومن ثـم               وتم تنفيذه من    

يقوم بتنفيذ جواب الشرط بينما مسار التنفيذ الثاني يتفرج عليه كالأطرش في الزفة، بـل سـتحدث     
  .التعارضات بين مسارات التنفيذ أيضا في اصغر جزء من أجزاء شيفراتك البرمجية

 والذي يمكنك من اعتبـار  SyncLock … End SyncLockيب من هنا يأتي دور الترك
جزء معين من شيفراتك المصدرية كوحدة واحدة بحيث يتم تنفيذها من قبل مسار تنفيذ واحد فقط،                
لتبقى مسارات التنفيذ الأخرى في قائمة الانتظار إن أرادت تنفيذ الشيفرة الموجودة في داخل هـذا          

  .التركيب
 فـي شـيفراتك   SyncLock … End SyncLockالتركيـب  كل ما تحتاجه لاسـتخدام  

 ويكـون  Nothing يحمل أي قيمـة لا تـساوي   Reference Typeالمصدرية، متغير مرجعي 
  :مشترك بين مسارات التنفيذ الأخرى

 
SyncLock y 
    If X = 10 Then 
        X = 0 
    Else 
        X = 10 
    End If 
End SyncLock 

ان الشيفرة الموجودة داخل التركيب السابق سيتم تنفيذها من قبل مسار تنفيذ واحد   والان اضمن لك    
فقط، وفي حال ما وجد مسار تنفيذ آخر يستخدم نفس الشيفرة، فسينتظر في طابور ريثمـا ينهـي                 

  .مسار التنفيذ الحالي إكمال الأجزاء المتبقية من الشيفرة
  : التركيب، راقب معي هذه الشيفرةحتى اريك تطبيق عملي ذو تأثير ظاهري على هذا

 
 

Module Module1 
    Sub main() 
        Dim counter As Integer 
 
        For counter = 1 To 9 
            With New Thread(AddressOf DoSomething) 
                .Name = CStr(counter) 
                .Start() 
            End With 
        Next 
        Console.Read() 
    End Sub 
 
    Sub DoSomething() 
        Dim counter As Integer 
 
 
        Console.WriteLine("From Thread " & _  
            Thread.CurrentThread.Name) 
 
        For counter = 1 To 1000 
            Console.Write(Thread.CurrentThread.Name) 
        Next 
        Console.WriteLine() 
    End Sub 
End Module 

  
مخرجات الشيفرة السابقة ستكون غير واضحة بشكل كبير، حيث سيتم تنفيذ الحلقة من قبل أكثـر                

  :من مسار تنفيذ في نفس الوقت وتكون المخرجات شيئا مثل
 
From Thread 1 
111111111111111111111111111111111111111 From Thread 2 
222222222222222222222222222221111111111111111111 From Thread 7 
77711111222222222221111111 
… 
… 

  
 … SyncLock ونجعل شيفراته داخل تركيـب  ()DoSomethingوالان لنعدل الإجراء 

End SyncLock:  

رات التنفیذمتغير مرجعي عام ومشترك بين كافة مسا '  
Public ThreadLocker As String = "Lock1" 
… 
… 
 
Sub DoSomething() 
    Dim counter As Integer 
 
    SyncLock ThreadLocker 
        Console.WriteLine("From Thread " & _  
            Thread.CurrentThread.Name) 
        For counter = 1 To 1000 
            Console.Write(Thread.CurrentThread.Name) 
        Next 
        Console.WriteLine() 
    End SyncLock 
End Sub 

  
 بان عمليـة تنفيـذ الـشيفرات    Visual Basic .NETبعد إضافة التركيب السابق، سيضمن لك 

ات التنفيذ ستظل فـي     الموجودة داخل هذا التركيب تتم من قبل مسار تنفيذ واحد فقط، وبقية مسار            
  :طابور انتظار، لذلك ستكون المخرجات أكثر تنظيما

 
From Thread 1 
111111111111111111111111111111111111111111111111111 … 
From Thread 5 
555555555555555555555555555555555555555555555555555 … 
From Thread 3 
333333333333333333333333333333333333333333333333333 … 
From Thread 2 
222222222222222222222222222222222222222222222222222 … 
… 
… 

  
  Synchronizationالمواصفة 

 عشرات المرات، يمكنك استخدام SyncLock … End SyncLockبدلا من استخدامك للتركيب 
 بداخلها يتم تنفيذها  في أعلى الفئة حتى تجعل جميع الشيفرات التي      Synchronizationالمواصفة  

  :في قبل مسار تنفيذ واحد فقط
 
<System.Runtime.Remoting.Contexts.Synchronization()> _ 
Class Test 
    ' ContextBoundObject لابد من اشتقاق الفئة 
    Inherits ContextBoundObject 
    … 
    … 
End Class 

ة التي تستخدمها لابد من تكـون مـشتقة          هو ان الفئ   Synchronizationيعيب المواصفة   
عيب آخر  .  مما يحرمك من إمكانية اشتقاق فئة أخرى       ContextBoundObjectوراثيا من الفئة    

، حيث يمكن Shared Members يتعلق بالأعضاء المشتركة Synchronizationفي المواصفة 
 من قبل مسارات التنفيذ ان تحدث التعارضات على الشيفرات المصدرية في أعضاء الفئة المشتركة

  .MethodImplمع ذلك، تستطيع الالتفاف حول هذه العيوب باستخدام المواصفة . المتعددة
  

 MethodImplالمواصفة 
 يمنع جميع مسارات التنفيذ من تنفيذ شيفرة الكـائن ان           Synchronizationاستخدامك للمواصفة   

ب غير مرن، حيث ما لفائـدة مـن تعـدد      كان مشغول بأحد مسارات التنفيذ، وهذا بحد ذاته أسلو        
  .مسارات التنفيذ إن كانت العمليات لا تتم إلا من قبل مسار تنفيذ واحد فقط في وقت واحد

 يمكنك من تحديد الأعضاء التي تود حكرها على مسار MethodImplاستخدامك للمواصفة 
هـذه  . Shared Membersتنفيذ واحد، مع العلم انها تعمل بنجاح أيضا على الأعضاء المشتركة 

  :طريقة استخدامها
 
  التالي لاختصار الشیفرةسماءاسترد مجال الأ '
Imports System.Runtime.CompilerServices 
… 
… 
Class Test1 
    <MethodImpl(MethodImplOptions.Synchronized)> _ 
    Sub AA(ByVal x As Char) 
        … 
        … 
    End Sub 
 
    <MethodImpl(MethodImplOptions.Synchronized)> _ 
    Shared Sub BB(ByVal x As Char) 
        … 
        … 
    End Sub 
 
    … 
    … 
End Class 

  
بودي توضيح بعض المسائل المترتبة على هذه المواصفة، حيث ان عملية الإقفـال لا تـتم     

 تشمل وتمتد إلى الطرق الأخـرى، فمـثلا لـو    على الطريقة التي عرفت فيها المواصفة فقط، بل      
  :استخدمت المواصفة في هاتين الطريقتين

 

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


 Threadingمسارات التنفیذ : العاشرالفصل   381  
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  
Class Test2 
    <MethodImpl(MethodImplOptions.Synchronized)> _ 
    Sub AA(ByVal x As Char) 
        … 
        … 
    End Sub 
    <MethodImpl(MethodImplOptions.Synchronized)> _ 
    Sub BB(ByVal x As Char) 
        … 
        … 
    End Sub 
 
    … 
    … 
End Class 

 
فاعلم ان الشيفرات المصدرية لكلا الطريقتين سيتم تنفيذها من قبل مسار تنفيذ واحد فقط وليس من             

ان  ان ك-كمزيد من التوضيح–قبل مسار تنفيذ واحد فقط للطريقة الأولى وآخر للطريقة الثانية، أي 
، فان مسار التنفيذ الثاني سيضطر إلى الانتظار        ()AAمسار التنفيذ الأول يجري عمليات الطريقة       
  .()BBرغم انه يود تنفيذ شيفرات الطريقة الآخر 

فيمكن ان يتم تنفيذها من قبل مسار تنفيذ   )  السابقة Test1كالفئة  (اما مع الأعضاء المشتركة     
الغيـر   (Instance Membersت احد الأعضاء العاديـة  آخر ان كان مسار التنفيذ يجري عمليا

  ).مشتركة
  

  فئات أخرى
 و Synchronization والمواصـفتين  SyncLock … End SyncLockإلى جانب التركيـب  

MethodImpl توجد مجموعة إضافية من الفئات والخاصة بتزامن عمل مسارات التنفيذ المتعددة ،
ــي     ــة ه ــرامج الجدي ــب الب ــي اغل ــستخدم ف ، Monitor ،Interlocked ،Mutex: ت

ReaderWriterLock  ،ManualResetEvent   و ،AutoResetEvent .   يمكنك البحـث عـن
 ان كنت مهتما في هذا الموضوع، وذلك NET Documentation.تفاصيل استخدامها في مراجع 

تيـة   بشكل مبسط حيث أنها تمثل البنيـة التح        Monitorلأني لن اعرض لك هنا إلا الفئة الأولى         
  . السابق ذكرهSyncLock … End SyncLockللتركيب 

 انك تحدد الـشيفرة المـصدرية وقـت    SyncLock … End SyncLockيعيب التركيب 
 حصر مجموعة من الشيفرة المصدرية في كتلة واحدة         -مثلا–التصميم وليس التنفيذ، فلن تستطيع      
 والتي تقوم بنفس عمـل  Monitorة ومن هنا يأتي دور الفئ. ان تم تحقق شرط معين لحظة التنفيذ      

  . ولكن في وقت التنفيذ وليس التصميمSyncLock … End SyncLockالتركيب 

 لاستخدامها، حيث ان جميـع طرقهـا        Monitorلست بحاجة لإنشاء كائن جديد من الفئة        
 لتحديد نقطة البداية للـشيفرة المـراد   ()Enter، استدعي الطريقة Shared Methodsمشتركة 
 لتحديد نقطة النهاية، مع العلم ان كلا الطريقتين تتطلبان متغير عـام             ()Exit، والطريقة   حصرها

Global Variables مرجعي Reference قيمته لا تساوي Nothing – تماما كمتطلبات التركيب
SyncLock … End SyncLock-فلو كان لدينا هذا التركيب :  

 
Public X As String = "Lock1" 
… 
SyncLock X 
    … 
 الشیفرة المراد حصرها على مسار تنفیذ واحد فقط '    
    … 
End SyncLock 

  
  : بهذا الشكلMonitorيمكنك تحويله إلى الفئة 

 
Public X As String = "Lock1" 
… 
Monitor.Enter(X) 
    … 
 الشیفرة المراد حصرها على مسار تنفیذ واحد فقط '    
    … 
Monitor.Exit(X) 

 
  :، يمكنك تنفيذه داخل جملة شرطيةMonitor من المرونة التي توفرها لك الفئة وكنوع

 
Public X As String = "Lock1" 
… 
If Y = 0 Then Monitor.Enter(X) 
… 
 الشیفرة المراد حصرها على مسار تنفیذ واحد فقط '
… 
Monitor.Exit(X) 
 
 

    ملاحظة
  

 سـیظھر رسـالة     ()Enter دون الطريقـة     ()Exitاستدعائك للطريقة   
لذلك، علیك تصحیح الشائب في الشیفرة السابقة في حال . خطأ

  :ان لم تكن قیمة الشرط صحیحة
  

If Y = 0 Then Monitor.Exit (X) 

 سـتجعل كـل   ()Exit و ()Enterالمزيد أيضا، الشيفرة المحصورة بين فكي الاستدعائين    
مسار التنفيذ الحالي تنفيذ الشيفرة، مع ذلك قد مسارات التنفيذ الأخرى في قائمة الانتظار حتى ينهي 

تود من مسارات التنفيذ الأخرى الانتظار لفترة من الوقت تحددها كوسيطة ثانيـة إلـى الطريقـة       
TryEnter():  

 
Public X As String = "Lock1" 
… 
ظر فترة نصف ثانیة ثم نفذ الشیفرات حتىأنت '  
  يجري التنفیذآخرلو كان مسار تنفیذ  '
Monitor.TryEnter(X, 500) 
    … 
 الشیفرة المراد حصرها على مسار تنفیذ واحد فقط '    
 لفترة اقل من نصف ثانیة '    
    … 
Monitor.Exit(X) 

  
  ThreadPoolالفئة 

 إلى إضعاف كفاءة التنفيذ، -بلا شك–إنشاء عشرات مسارات التنفيذ في برنامج عامل واحد يؤدي          
سـتجرى آلاف   ) Threadباسـتخدام الفئـة     (ئ فيها مسار تنفيذ جديـد       كما انك في كل مرة تنش     

التعليمات والخاصة بنظام التشغيل لخلق المسار، ولحظة موت مسار التنفيـذ تـؤدي إلـى آلاف                
  . لقتل المسار-والخاصة بنظام التشغيل أيضا–العمليات 

، حيث تمكنك Thread Poolبركة مسارات التنفيذ يمكن زيادة كفاءة التنفيذ بالاعتماد على 
يمكنك استعارة مـسار تنفيـذ      .  واستخدامه في برنامجك   -بشكل سريع –من استعارة مسار التنفيذ     

، والتي تتطلب من الإجراء المفوض لها ان يحتوي ()QueueUserWorkItemباستدعاء الطريقة 
  :Objectعلى وسيطة واحدة من النوع 

 
ThreadPool.QueueUserWorkItem(AddressOf DoSomething) 
… 
… 
' Object لابد ان يحتوي الاجراء المفوض على وسیطة من النوع 
Sub DoSomething(ByVal x As Object) 
    … 
    … 
End Sub 

  
، انك تستطيع إرسال أي قيمة له تـستفيد         ThreadPoolالميزة في الإجراء المفوض بمشيد الفئة       
  :Threadمنها، وهي ميزة حرمتنا منها سابقا الفئة 

ThreadPool.QueueUserWorkItem(AddressOf DoSomething, 100) 
… 
… 
Sub DoSomething(ByVal x As Object) 
    … 
    Console.WriteLine (X) ' 100 
End Sub 

 
 سيؤدي إلى إنشاء بركة مسارات تنفيذ       ()QueueUserWorkItemأول استدعاء للطريقة    

، يمكنك معرفة العدد Priorityنفيذ متساوية الأسبقية  مسار ت 25 على   -بشكل افتراضي –تستوعب  
الأقصى من مسارات التنفيذ الممكن استعارتها في بركة مسارات التنفيذ بالاسـتعلام عنهـا فـي                

، كما تستطيع معرفة عدد مسارات التنفيذ المتوفرة فـي البركـة            ()GetMaxThreadsالطريقة  
، كلا الطـريقتين تـستقبل وسـيطتين     ()GetAvailableThreadsلاستعارتها باستدعاء الطريقة    

  :ByRefبالمرجع 
 
Dim x, y, z As Integer 
 
ThreadPool.GetMaxThreads(x, z) 
ThreadPool.GetAvailableThreads(y, z) 
 
Console.WriteLine(x) ' 25 
Console.WriteLine(y) ' 23 
 

    ملاحظة
  

ــريقتین     ــلا الطـــــ ــة لكـــــ ــیطة الثانیـــــ  و ()GetMaxThreadsالوســـــ
GetAvailableThreads()          يسند لھا قیمـة تمثـل عـدد مـسارات التنفیـذ 

 لتفاصیل أكثر حول MSDNراجع مكتبة . IOالخاصة بدخل وخرج الملفات  
  .ھذا الموضوع

 
  

نقطة هامة جدا جدا، مسارات التنفيذ المستعارة من بركة مسارات التنفيذ هي مسارات تنفيذ              
، لذلك كل مسار تنفيذ سيواجه Free Threading حرة  وليسApartment Threadingمتباعدة 

  :نسخة خاصة به من المتغيرات العامة، الشيفرة المسطورة في أعلى الصفحة المقابلة خير دليل

Module Module1 
    Dim x As Integer 
    Sub main() 
        x = 100 
        ThreadPool.QueueUserWorkItem(AddressOf DoSomething, Nothing) 
 
        Thread.Sleep(1000) 
        Console.WriteLine(x) ' 100 
        Console.Read() 
    End Sub 
 
    Sub DoSomething(ByVal x As Object) 
        x = 1000 
        Console.WriteLine(x) ' 1000 
    End Sub 
End Module 

  
 لن يتم إنشاء نسخة لهـا  Global Reference Variablesامة المرجعية مع ذلك، المتغيرات الع

  :Managed Heapلكل مسار تنفيذ، وذلك لأنها تحفظ في نفس ذاكرة 
 
Class TestClass 
    Public x As Integer 
End Class 
 
Module Module1 
    Public TestObject As New TestClass() 
    Sub main() 
        TestObject.x = 100 
        ThreadPool.QueueUserWorkItem(AddressOf DoSomething, Nothing) 
 
        Thread.Sleep(1000) 
        Console.WriteLine(TestObject.x) ' 1000 
        Console.Read() 
    End Sub 
 
    Sub DoSomething(ByVal x As Object) 
        TestObject.x = 1000 
    End Sub 
End Module 

  
أخيرا وليس آخرا، يمكنك معرفة ما اذا كان مسار التنفيذ الحالي يعمل في بركة مـسارات                

  :IsThreadPoolThreadالتنفيذ بالاستعلام عن الخاصية 
 
If Thread.CurrentThread.IsThreadPoolThread Then 
    … 
    … 
End If 

  :مسارات التنفيذهذا مثال كامل لاستخدام بركة 
 

Module Module1 
    Sub main() 
        Dim counter As Integer 
 
        ThreadPool.QueueUserWorkItem(AddressOf DoSomething, 1) 
        ThreadPool.QueueUserWorkItem(AddressOf DoSomething, 2) 
        ThreadPool.QueueUserWorkItem(AddressOf DoSomething, 4) 
        Console.Read() 
    End Sub 
 
    Sub DoSomething(ByVal x As Object) 
        Dim counter As Integer 
        For counter = 1 To 100000 
            Console.Write(x) 
        Next 
    End Sub 
End Module 

  
 أو استعارة مسار تنفيذ     Threadرة باستخدام الفئة    ايهما أفضل، إنشاء مسارات التنفيذ مباش     

 ستكون معتمدة على    -كل العادة -؟ اجابتي   ThreadPoolمن برك مسارات التنفيذ باستخدام الفئة       
متطلباتك، ولكن دعني اذكرك هنا بان استعارة مسارات التنفيذ من البركة لا تقوم بتنفيذ الاجـراء              

، اضف إلى ذلـك انهـا مـسارات تنفيـذ خلفيـة      Threadالمفوض لها فورا كما تفعل مع الفئة   
Background Thread ــة ــا عادي ــة (، وان جعلته ــناد القيم ــية Falseباس ــى الخاص  إل

IsBackground (       فلن تتمكن من قتلها باستخدام الطريقةAbort() .    باختصار، أنشي مـسارات
عمر البرنامج دون توقف،   ان كانت المسارات تعمل في فترة طويلة من          Threadالتنفيذ من الفئة    

وعود نفسك على الاستعارة من بركة مسارات التنفيذ ان كنت تنوي إنشاء وقتل الكثير من مسارات 
  .التنفيذ مرات عديدة لإنجاز مهام معينة في فترات معينة من حياة البرنامج

  
  Timersالمؤقتات 

:  هـي Timersالمؤقتـات   ثـلاث انـواع مـن    NET Framework.يوفر لك إطار عمـل  
Timers.Timer ،Threading.Timerو ،Windows.Forms.Time  تجدها في مجال الأسماء 

في هذا القسم اعرض لك المؤقتين الأول والثاني، اما الاخير فأرى انـه مـن         . Systemالجذري  
  .Controlsالأدوات المناسب تأجيله حتى الفصل الرابع عشر 

  System.Timers.Timerالمؤقت 
باسـتخدام   (Elapsed، عليـك قـنص حدثـه        Timers.Timer لكائن من النـوع      عند إنشائك 

WithEvents   أو AddHandler (           والذي يتم تفجيره كل فترة من الوقت تحددها في الخاصـية
Intervalأو إرسالها كمشيد :  

 
 ثانیة '
Dim Timer1 As New Timers.Timer (1000) 
 
 نصف ثانیة '
Dim Timer2 As New Timers.Timer () 
Timer2.Interval = 500 

  
 Elapsed ان اردت تفجير الحـدث       AutoReset إلى الخاصية    Falseيمكنك اسناد القيمة    

 وإيقافـه فـي أي وقـت       ()Startمرة واحدة فقط، كما تستطيع تشغيل المؤقت باستدعاء الطريقة          
  .()Stopبالطريقة 

  :System.Timers.Timerهذا مثال لاستخدام المؤقتات من النوع 
 

Module Module1 
    Sub main() 
        Dim MyTimer As New System.Timers.Timer() 
        AddHandler MyTimer.Elapsed, AddressOf DoSomething 
 
        MyTimer.AutoReset = True 
        MyTimer.Interval = 1000 
        MyTimer.Start() 
 
        Console.Read() 
    End Sub 
 
    Sub DoSomething(ByVal sender As Object, ByVal e As 
Timers.ElapsedEventArgs) 
        Console.WriteLine(Now.ToString("hh:mm:ss")) 
    End Sub 
End Module 

    ملاحظة
  

 والتابعة للمؤقتات مـن النـوع       Elapsedالإجراءات التي تم قنصھا للحدث      
System.Timers.Timer     ــك ــذ، يمكن ــسارات التنفی ــة م ــي برك ــل ف  تعم

  :التحقق من ذلك
  

Console.WriteLine(Thread.CurrentThread.IsThreadPoolThread) ' True 
 

 

  
  System.Threading.Timerالمؤقت 

 ولكنه يختلـف فـي   Timers.Timer كعمل المؤقت السابق  Threading.Timerيعمل المؤقت   
ولـيس   (Delegate إجراء مفوض    Threading.Timerمه، حيث يتطلب المؤقت     طريقة استخدا 

  .يتم تنفيذه كل فترة معينة) Eventحدث 
، عليك إرفاق جميع البيانات المطلوبة      Threading.Timerإن أردت إنشاء كائن من الفئة       

وسيطة والذي يشترط ان يحتوي على –الوسيطة الأولى تمثل الإجراء المفوض     . كوسيطات لمشيده 
  :Objectمن النوع 

 
Dim MyTimer As New Threading.Timer(AddressOf DoSomething, …) 
 
Sub DoSomething(ByVal x As Object) 
    … 
    … 
End Sub 

  
  :يمكنك ارسال أي قيمة إلى هذا الإجراء المفوض، بذكرها كوسيطة ثانية لمشيد فئة المؤقت

 
 ارسل القیمة 100 '
Dim MyTimer As New Threading.Timer( …, 100, …) 
 
Sub DoSomething(ByVal x As Object) 
    Console.WriteLine(x) ' 100 
End Sub 

  
اما الوسيطة الثالثة فتحدد فيها المدة التي تود بدء عملية تنفيذ الإجراء، والوسيطة الرابعة تحدد فيها    

  :الفترة الزمنية لتكرار تنفيذ الإجراء

  ثواني لمرة واحدة فقط5لاجراء بعد نفذ ا '
Dim MyTimer As New Threading.Timer( …, …, 5000, 0) 
 
  ثواني، وثم كرر التنفیذ كل نصف ثانیة10نفذ الاجراء بعد  '
Dim MyTimer As New Threading.Timer( …, …, 10000, 500) 
 

  
 التي تمكنك من ()Changeريقة لا توجد خصائص تمكنك من تعديل القيم السابقة، ولكن توجد الط    

اخيرا، تستطيع إيقاف عمل المؤقت باستدعاء الطريقة       . تعديل قيمة الوسيطة الثالثة والرابعة للمشيد     
Dispose().  

  :Threading.Timerهذا مثال لاستخدام المؤقت 
 

Module Module1 
    Sub main() 
        Dim MyTimer As New Threading.Timer(AddressOf DoSomething, _  
        Nothing, 1, 2000) 
 
        Console.Read() 
    End Sub 
 
    Sub DoSomething(ByVal x As Object) 
        Console.WriteLine(Now.ToString("hh:mm:ss")) 
    End Sub 
End Module 
 

    ملاحظة
  

ــوع    ــن الن ــات م ــراءات المفوضــة للمؤقت  System.Threading.Timerالإج
  .تعمل أيضا في بركة مسارات التنفیذ

، إن  Threadingأكاد لا اصدق أني انتهيت فعلا من كتابة هذا الفصل والخاص بمسارات التنفيذ              
-Multiكان الشرح غير واضح ولم تفهم منه شيئا، فتذكر ان برمجة مسارات التنفيـذ المتعـددة           

Threading     برمجية والتي شهدتها أنامل المبرمجين علـى مـر          من اعقد وأصعب المواضيع ال
أما الصفحات المقبلة ستأخذك إلى موضوع جديد كليا وحديث العهد، اذ انه ظهر مع إطار . التاريخ
  .Assembliesالمجمعات  فقط وهو NET Framework.عمل 

 Assembliesالمجمعات 

  
، أو أي Class Librariesت فئات  قياسية، أو مكتباWindowsسواء كنت تنوي تطوير تطبيقات 

مجمـع  نوعية أخرى من البرامج، عليك معرفة ان التطبيق الذي تطوره وتنوي توزيعـه يـسمى     
Assembly.  

في هذا الفصل النظري أتحدث عن مجموعة كبيرة من المواضيع المتفرقة والتي تظهر لك              
 الربط والترجمة لتمكنك مـن      البنية التحتية لتركيبة المجمعات، كما سنستخدم مجموعة من أدوات        

ولكن قبل ذلك، دعنا نبدأ من الأساس .  مختلفة في برنامج واحدNET.دمج شيفرات مكتوبة بلغات 
  .Managed Modulesوتوضيح الوحدات المدارة 

  
  Managed Modulesالوحدات المدارة 

، )اخـرى  NET.أو أي لغة  (Visual Basic .NETعندما تقوم بترجمة برنامجك المكتوب بـ 
 أو EXE ملـف  في، والتي قد تكون Managed Moduleوحدة مدارة فالنتيجة النهائية تسمى 

DLL)     فالملف التنفيذي    "في"لاحظ حرف الجر ،EXE     هذه الوحدة   ) قد يشمل أكثر موحدة مدارة ،
رة الوحدة المدا. المدارة يمكن ان تعمل بشكل مستقل أو تعتمد على وحدات مدارة اخرى ليتم تنفيذها

إذاً، فـالمجمع قـد   . Assemblyمجمع الأو مجموعة الوحدات المدارة المترابطة تكون ما يسمى  
  .يحتوي على وحدة مدارة واحدة أو مجموعة من الوحدات المدارة

  :تتكون الوحدة المدارة من اربعة اقسام هي
1( Windows PE File Header :    وهو رأس للملف خـاص بنظـام التـشغيل

Windowsله جميـع التطبيقـات التـي تعمـل تحـت بيئـة        والذي تشمMicrosoft 

Windows.  
2( .NET Framework File Header:     وهـو أيـضا رأس للملـف خـاص 

 يحتوي على معلومات حـول  NET Framework.بالتطبيقات الموجهة إلى إطار عمل 

انـات  الوحدة المدارة الحالية كنقطة بداية التنفيذ، مؤشرات إلى شيفرات التنفيذ، أو إلى بي            
Metadataوغيرها ... 

3( Metadata:           يحتوي هذا القسم على وصف لأنواع البيانات الموجودة في الوحدة 
، التركيبـات مـن النـوع       Modules، الوحدات البرمجيـة     Classesكالفئات  (المدارة  

Structure   أو Enum   الواجهات ،Interfaces... يستخدم إطار عمل    ). الخ.NET  هذا 
اع البيانات المرسلة بين المجمعات أو الوحدات المدارة المختلفة، كما   القسم للتحقق من أنو   

 Reflectionيمكنك الاستعلام عن هذه البيانات باستخدام مجموعة من فئات الانعكـاس  

Classesكما سترى لاحقا في الفصل القادم . 

 Visual Basicتعـرف علـى   ذكرت في الفـصل الأول  كما  :MSILشيفرة  )4
.NET  الترجمة على الفور    ن فقرة    وبالتحديد عJIT      حيث قلت ان شيفراتك المصدرية ،

 Microsoft تسمى Assemblyلحظة الترجمة سيتم تحويلها إلى لغة شبيهه بلغة التجميع 

Intermediate Language) MSIL أو IL .( شيفرةMSIL  تمثل شيفرة كل أجـزاء 
 .الوحدة المدارة بعد الترجمة

  
 من القسم الاول، الثاني، والرابع الا ان كنت تنوي التخصص في            تطبيقيا، لن تستفيد الكثير   

 NET. الخاصة بإطـار عمـل   CRL، ولغة التنفيذ المشتركة Windowsالبنية التحتية لتطبيقات   
 فسترى في الفصل    Metadataوهي مواضيع متقدمة جدا وخارج نطاق الكتاب، اما القسم الثالث           

 كيف يمكنك الاستفادة منها والاستعلام عن البيانات Reflection Classesفئات الانعكاس القادم 
 من معرفة جميع الفئات -مثلا–، لتتمكن Visual Basic .NETالموجودة في الوحدة المدارة بـ 

  .والاستعلام عن جميع أعضائها من برنامج خارجي اخر

  
 Assembliesالمجمعات 

، أي بعبارة اخـرى     NET. لغات    هو اصغر وحدة تمثل تطبيق منجز بأحد       Assemblyالمجمع  
المجمع هو التطبيق  الذي أنشأته، بحيث يكون له اسم خاص، رقم إصدار، اسم المصمم، الشركة                 

 تستخدمها في برامج DLL، أو مكتبة EXEقد يكون المجمع في ملف قابل للتنفيذ مباشرة       . الخ...
). الـخ  ...EXE ،DLL ،DOC ،HTML ،BMP(اخرى، أو مجموعة مترابطة من الملفـات        
 تـسمى الوحـدات     -كما ذكرت في الفقرة الـسابقة     –يحتوي المجمع على مجموعة من العناصر       

 Nonexecutableليس هذا فقط، بل قد يحتوي المجمع على عناصر اخرى ليست تنفيذية  . المدارة

، صـور  Texts، ملفـات نـصوص   HTML، صـفحات  Resource Filesكملفات المصادر 
Pictures ... الخ.  

 فيما بينهم، فإنهم يستخدموا المصطلح أنجزنا مجمعات        NET.لك، عندما يتباهى مبرمجو     لذ
  .عوضا عن المصطلحات برامج أو تطبيقات

  
  المجمعات الأحادية والمتعددة الملفات

لا يشترط ان تكون جميع الوحدات المدارة والتابعة لمجمع معين مدمجة في ملف واحد، بل يمكن                
 -نظريـا –ة من الملفات المستقلة، مع ذلك جميع هذه الملفات مترابطـة     ان يكون المجمع مجموع   

فلو كان للمجمع صلاحيات معينـة، فـان هـذه    . )1-11شكل  (وكأنها ملف واحد لتكون المجمع      
الصلاحيات تشمل جميع الوحدة المدارة التابعة له، وان كان للمجمع رقم إصدار خاص فـستحمل               

  .رقمكافة الوحدات المدارة نفس هذا ال
  

  
   المجمعات أحادية ومتعددة الملفات:1-11شكل 

  
معظم التطبيقات التي أنجزناها في هذا الكتاب تحتوي على وحدة مدارة واحدة، وبعد عملية              

 مجمع أحادي الملف، تمكنا في الفصول السابقة من إنجاز هذه EXEالترجمة يمثل الملف التنفيذي   
 وبالتحديد بعد الضغط علـى المفتـاح   Visual Studio .NETالمجمعات باستخدام بيئة التطوير 

[F5] .                اما ان كان لديك أكثر من وحدة مدارة وأردت دمجها في ملف واحد، فعليك استخدام الأداة
AL.EXEوالتي سيأتيك تفصيلها لاحقا .  

اعيد واكرر نقطة هامة عليك أخذها دائما في الاعتبار، المجمعات سـواء كانـت أحاديـة               
 DLL أو EXE كملـف    -منطقيـا –متعددة الملفات هي برنامج واحد، ويمكنك تخيله        الملفات أو   

  . متعدد الملفات-فيزيائيا–واحد رغم انه 

  أسالیب تنفیذ المجمعات
في -مهما كان غرضها ومهما كانت طرق إنجازها، يتم تنفيذها          ) البرامج والتطبيقات (المجمعات  

 Windows، تطبيـق  Libraryمكتبـة  :  بطريقة واحدة من ثلاثـة طـرق هـي   -NET.عالم 

Application أو تطبيق ،Console Application.  
، DLL فيكون امتداد الملف النهائي للمجمع في اغلب الأحوال    Librariesبالنسبة للمكتبات   

 حتى  Referenceلن تستطيع تشغيلها مباشرة لاستخدامها، وانما عليك إضافتها في خانة المراجع            
  . الأخرىNET.فادة من فئاتها في برامجك الأخرى والتي تنجزها بلغات تتمكن من الاست

 يمكـن تـشغيلها   Console Application و Windows Applicationاما التطبيقات 
يكـون  ( تمثل الإجراء الابتدائي  وهي.Entry Pointنقطة إدخال مباشرة، لذلك هي تتطلب منك 

Sub Main() في اغلب تطبيقات Visual Basic .NET (     الذي يـتم اسـتدعائه لحظـة تنفيـذ
  .البرنامج

 متشابهة إلى حد كبير، Console Application و Windows Applicationالتطبيقات 
 والخـاص بعـرض     Consoleويكمن الفرق بينهما ان النوع الثاني يمكنك من استخدام الكـائن            

  .المخرجات، اما الاول فلا يعتمد عليه
 Project فـي صـندوق الحـوار    Output Typeمن خانـة  يمكنك تحديد نوع المجمع 

Property Pages ) 2-11شكل(.  
  

  
  . تحديد نوع المجمع:2-11شكل 

  المجمعات الخاصة والمشتركة
 Private Assembliesالمجمعـات الخاصـة   يوجد نوعان من المجمعات يمكنك إنجازها هما 

ة هي مجمعات تضع ملفاتها في المجمعات الخاص. Shared Assembliesالمجمعات المشتركة و
ليس هذا فقط، بل إن اسـتخدامها محـصور         . نفس مجلد البرنامج أو المجلدات الفرعية التابعة له       
  :فلو إفترضنا هذه المجمعات. فهي خاصة به–للبرنامج الذي في نفس المجلد أو المجلد الأبوي لها 

 
C:\Folder\AAA.DLL 
C:\BBB.EXE 
C:\Folder\CCC.EXE 
C:\Folder\SubFolder\DDD.EXE 

  
 فقـط،  CCC.EXE و BBB.EXE من خلال التطبيقـين  AAA.DLLفيمكنك استخدام المجمع  

مع ذلك، عليك استخدام الوسم . وذلك لانهما في نفس مجلد المجمع الخاص أو في المجلد الأبوي له   
<probing> في ملف التهيئة للتطبيق BBB.EXEي يحتوي  لتحدد فيهما اسم المجلد الفرعي والذ

 لطريقة استخدام هـذا     MSDNراجع مكتبة    (Probingبالإيجاد  على المجمع لانجاز ما يعرف      
  .الوسم

  
    ملاحظة

  
عندما تقوم بإضافة مجمـع مـشترك فـي برنامجـك عـن طريـق صـندوق          

– Visual Studio .NET، سـتقوم بیئـة التطـوير    Add Referencesالحوار 
مـع  . د برنامجـك الرئیـسي     بنسخ ملف ھذا المجمع إلى مجل      -مشكورة

ذلك، عندما تحذف المجمع من صـندوق الحـوار الـسابق، سـتقوم بیئـة        
 بحـذف ملـف المجمـع مـن مجلــد     -غیـر مـشكورة ھـذه المــرة   –التطـوير  

  !دون سابق إنذار البرنامج مباشرة 
  

  
 اخـرى،   NET.فهي مجمعات قابلة للاستخدام من قبل مجمعات         اما المجمعات المشتركة  

الوحدة العامـة   ، يعرف هذا المجلد بالاسم      X:\Windows\assemblyا في المجلد    وتضع ملفاته 
ك رؤية المجمعات المشتركة بجهازك يمكن. )Global Assembly Cache) GACللمجمعات 

ــام    ــشف النظ ــن مستك ــد    Windows Explorerم ــى المجل ــال إل ــك بالانتق  وذل
X:\Windows\assembly ) بالصفحة التالية3-11شكل (.  

  .GAC المجمعات المشتركة في وحدة :3-11كل ش

  
 لايريك الشكل الفيزيـائي  Windows Explorerمن الضروري التنبيه هنا بان مستكشف النظام 

، Command DOS Promptالحقيقي لملفات المجمعات المشتركة، فلو تستخدم موجه الأوامـر  
المجلـدات يحتـوي علـى    ستكتشف ان المجمعات موزعة على شكل مجلدات، كل مجلد من هذه    

  .مجلدات فرعية تمثل الإصدارات المختلفة للمجمع
يمكنك حذف مجمع مشترك بالنقر بز الفأرة الأيمن على رمز المجمـع واختيـار الامـر                

Delete               من القائمة المنبثقة، اما إن اردت إضافة مجمع مشترك فتستطيع نقل ملفه إلى GAC  أو 
  .سيأتي ذكرها لاحقا في هذا الفصل والتي GACUTIL.EXEاستخدام الأداة 

  
    ملاحظة

  
لا يمكنك حذف أو إضافة مجمع مشترك في الوحـدة العامـة للمجمعـات      

GAC   إلا إذا كانـت لــديك صــلاحیة المـستخدم Administrator مــن قبــل 
  .نظام التشغیل

  Strong Namesالأسماء القوية 
ينها على حسب موقع ملفاتها في القرص،  يتم التفريق بPrivate Assembliesالمجمعات الخاصة 

لذلك إن وجدت عشرات المجمعات المتشابه في أسمائها وأرقام إصداراتها، فستنحصر المسؤولية            
  .عليك كمبرمج من وضع ملفات المجمعات في مجلدات منفصلة

، فجميع المجمعات توضع فن Shared Assembliesاما الحديث عن المجمعات المشتركة 
، وان تشابهت اسماء المجمعات، فسيتم انشاء مجلدات فرعيـة     GAC الرئيسي للوحدة    نفس المجلد 

 له اصدارين، فسيتم حفظه فـي   AAAبارقام الإصدارات المختلفة للمجمع، فلو تخيلنا ان المجمع         
GACبهذا الشكل :  

 
X:\Windows\assembly\GAC\AAA\1.2.3432423 
X:\Windows\assembly\GAC\AAA\2.4.12334 
… 
… 

  
مع ذلك، الاعتماد على رقم الإصدار غير كافي وذلك لأنه قد ينشئ أكثر من مبرمج مجمع يحمل                 

  .نفس الاسم ونفس رقم الإصدار وبالتالي يتم التعارض
حجمه ( هو الحل لمشكلة التعارض السابقة، وهو رقم كبير جدا Public Keyالمفتاح العام 

تضمن عدم تعارض مفتاحين عامين فـي الكـرة         يتم إنشائه بخوارزميات معقدة بحيث      )  بت 128
  !الأرضية

 المجمع، مفتاحه العام، رقم اصـداره، اسـم    اسم: النقطة الهامة التي اريد ان اصل بك هي       
 Assembly'sبالاسم القوي للمجمـع  الخ تسمى مجتمعة ...الشركة المنتجة، اعداداته الإقليمية 

Strong Name.  
 والتـي  SN.EXE لمجمعاتك عن طريق الأداة Strong Nameتستطيع تسجيل اسم قوي 

  .سيأتي ذكرها لاحقا
  

  Assemblyالمواصفة 
 إن اردت إسناد معلومات حول المجمع، كرقم الاصـدار،          Assemblyيمكنك استخدام المواصفة    

  :الخ...اسم المجمع، الشركة، حقوق التأليف، 
 
Imports System.Reflection 
Imports System.Runtime.InteropServices 

<Assembly: AssemblyTitle("نظام المحاسبة")>  
<Assembly: AssemblyDescription(" برنامج حسابي لادارة العملیات والحسابات
  <("للشركات
<Assembly: AssemblyCompany("شبكة المطورون العرب")>  
<Assembly: AssemblyCopyright("الحقوق محفوظة لشبكة المطورون العرب ©")>  
<Assembly: AssemblyVersion("1.0.2321432")>  
… 
… 
… 
 

 
 حتى تزيد رقم المراجعة بشكل تلقائي في كل مرة تنفذ *بالنسبة للاصدار، فيمكنك استخدام النجمة    

  :وتجرب المجمع
 
<Assembly: AssemblyVersion("1.0.*")> 

  
 تقوم باضافة الشيفرات Visual Studio .NETاخيرا، المشاريع التي تنجزها بيئة التطوير 

  .، ويفضل لك إتباع نفس الأسلوبAssemblyInfo.vbالسابقة في ملف مستقل يحمل الاسم 

  
  Configuration Filesملفات التھیئة 

يمكنك التحكم أكثر في إدارة المجمعات المتعددة وتغيير معظم أساليب تنفيذها أو البحث عنها عن               
هذه الملفات ما هي الا ملفات نصية منسقة بصيغ . Configuration Filesطريق ملفات التهيئة 

XML   يمكنك تحريرها بأبسط برامج التحرير كالمفكرة Notepad لتغيير مجموعة من الاعدادات 
  .التي تتعلق بالمجمعات والربط بينها

  
    ملاحظة

  
)  وما قبلھا  x.3الإصدارات  ( المخضرمین   Windowsان كنت من مبرمجي     

لفات التھیئة ھي مشابھة إلى حد كبیـر بملفـات التھیئـة     فالفكرة من م  
  . والتي ترفق مع التطبیقات المختلفةini.*السابقة ذو الامتداد 

  أنواع ملفات التھیئة
 توجد ثلاث أنواع من ملفات التهيئة عبارة عن مـستويات لتحديـد الاعـدادات               NET.في عالم   

، ملف تهيئـة  Application Configuration Fileملف تهيئة التطبيق :  هيNET.للمجمعات 
 Machine Configuration، وملف تهيئة الجهاز Publisher Configuration Fileالناشر 

File .وفيما يلي تفاصيلها:  
  

  :Application Configuration Fileملف تهيئة التطبيق 
ت انشاء ملـف تهيئـة   ملف التهيئة هذا يكون موجه وخاص بتطبيق أو مجمع واحد فقط، وان ارد       

التطبيق بنفسك، فلابد ان يكون في نفس مجلد ملف التنفيذ الرئيسي للمجمع و يحمل نفس اسم ملف                
فلو كان الاسم الكامل لملف .  إلى امتداد الملف الرئيسيconfig.التنفيذ الرئيسي مع اضافة الامتداد 
  :التنفيذ الرئيسي للمجمع بهذا الشكل

 
C:\MyProg\application.exe 

  
  :لابد ان يكون ملف تهيئة التطبيق بهذا الاسم

 
C:\MyProg\application.exe.config 

  
  :Publisher Configuration Fileملف تهيئة الناشر 

 بحيث يشمل Shared Assembliesهذا النوع من ملفات التهيئة يستخدم مع المجمعات المشتركة 
يمكنك الاسـتفادة مـن     . التي تستخدم هذا المجمع   تأثيره على المجمع المشترك وجميع التطبيقات       

علـى  ) لنقل مكتبة فئـات   ( بتوزيع احد المجمعات المشتركة      -مثلا–ملفات تهيئة الناشر ان قمت      
المستخدمين، وبعد فترة اكتشف خطأ في هذه المكتبة وأردت من المستخدمين إعادة تثبيتها، لذلك قد  

جيه جميع التطبيقات التي استخدمت المجمع المشترك       ترفق ملف تهيئة من هذا النوع حتى تعيد تو        
  .إلى الاصدار الجديد منه

–شروط تسمية ملف تهيئة الناشر هي مثل شروط تسمية ملف تهيئة التطبيق، ولكنه موجه               
  . للمجمعات المشتركة-كما قلت

  :Machine Configuration Fileملف تهيئة الجهاز 
 Administrator Configurationبملف تهيئة المـشرف  يسمى ايضا (اما ملف تهيئة الجهاز 

File ( يشمل تأثيره على جميع تطبيقات وبرامج.NET التي تعمل تحت إصدار إطار عمل .NET 

Framework    اسم هذا الملف  .  معين في الجهاز الحاليmachine.config   وتجده فـي المجلـد 
C:\WINDOWS\Microsoft.NET\Framework\vxxx\CONFIG\) ث تمثل   حيxxx  رقم 

، ففي جهازي الشخصي هـذا هـو   ) المثبت في الجهازNET Framework.اصدار إطار عمل 
  : الاسم الكامل لملف تهيئة الجهاز

 
C:\WINDOWS\Microsoft.NET\Framework\v1.0.3705\CONFIG\machine.config 

  
  تغییر الاعدادات

  :XML بصيغ <configuration>الشكل العام لملفات التهيئة يحمل الوسم 
 
<?xml version="1.0" encoding="UTF-8"?> 
    <configuration> 
        … 
        … 
        … 
    </configuration> 

  
توجد مئات الاعدادات والوسوم الإضافية التي تتحكم في سلوك تنفيذ المجمعات يمكنك إضافتها بين 

 الاعدادات خاصة بملفات تهيئـة التطبيقـات        بعض هذه .  السابق <configuration>فكي الوسم   
Application configuration files  ــاز ــة الجه ــف تهيئ ــة بمل ــضها خاص ــط، وبع  فق

machine.configفقط، والبعض منها يشمل جميع الأنواع المختلفة لملفات التهيئة .  
يق ومـن  في حالة تعارض الاعدادات بين ملفات التهيئة، فالأسبقية ستكون لملف تهيئة التطب  

، أي ان عملية قراءة ملفات التهيئة تبدأ بملـف تهيئـة            machine.configثم ملف تهيئة الجهاز     
 .الجهاز ومن ثم ملف تهيئة التطبيق وأي تعارض بين الاعدادات سيطغي عليه ملف تهيئة التطبيق

  
  اعدادات لملفات التھیئة

ض بعض هذه الاعـدادات، امـا إن   احاول في الفقرات التالية والصفحات القادمة لهذا الفصل عر    
 NET.أردت معرفة جميع الاعـدادات الأخـرى، فلـست بحاجـة إلـى تـذكيرك بمراجـع        

Documantation:  

  :<requiredRuntime>الوسم 
 الـذي  NET Framework.عندما تنوي تنفيذ مجمعك على جهاز اخر، فان إصدار إطار عمل 

مع ذلك، تستطيع تغيير الاصدار المطلوب . ج فيهيتطلبه هو نفس الإصدار الذي تم ترجمت البرنام     
 من <startup> والذي يشترط وضعه في القسم الفرعي   >requiredRuntime<باستخدام الوسم   

  :ملف التهيئة
 
<?xml version="1.0" encoding="UTF-8"?> 
    <configuration> 
        <startup> 
            <requiredRuntime version="v1.0.3705" /> 
        </startup> 
    </configuration> 
 

  
    ملاحظة

  
-caseأي انھا ( تفرق بین الحروف الصغیرة والكبیرة XMLتذكر ان وسوم   

sensitive .لذلك، اكتبھا بنفس حالة الأحرف التي تراھا.  
  

  
 يختلف عن رقم الإصدار     >requiredRuntime<إن كان رقم الاصدار المحدد في الوسم        

، سيتم حفظ معلومات إضافية حول      )أي الإصدار الذي تم ترجمة البرنامج عليه      (الأصلي للبرنامج   
، وحتى لا تكثر هذه المعلومات في كـل  Windows Registryهذا الاختلاف في مسجل النظام 

 فــي الوســم safemode للخاصــية Trueمجمــع تقــوم بتغييــره، يمكنــك إســناد القيمــة 
>requiredRuntime<:  

 
<?xml version="1.0" encoding="UTF-8"?> 
    <configuration> 
        <startup> 
            <requiredRuntime version="v1.0.3705" safemode="true" /> 
        </startup> 
    </configuration> 

  
  :<gcConcurrent>الوسم 

كرة في مسار تنفيذ مستقل، مما يضمن  بعملية تفريغ الذاGarbage Collectionتقوم المجموعة 
عدم إيقاف مسارات التنفيذ الأخرى والتي بحاجة لعملية الاستمرار خاصة ان كانت خاصة لعرض    

 بإلغاء هذه الطريقة وإيقـاف عمـل كافـة        MSDNمع ذلك، تنصح مستندات     . واجهات رسومية 

عمل في أجهـزة الخـوادم      مسارات التنفيذ لحظة تفريغ الذاكرة في ان كنت تنوي تطوير برامج ت           
Servers يمكنك استخدام الوسم .  ولا تحتوي على أية واجهات رسومية>gcConcurrent<  فـي 

 لوقف جميـع    enabled لخاصيته   False لملف التهيئة وإسناد القيمة      >runtime<القسم الفرعي   
  : للذاكرةGarbage Collectionمسارات التنفيذ لحظة تفريغ المجموعة 

 
<?xml version="1.0" encoding="UTF-8"?> 
    <configuration> 
        <runtime> 
            <gcConcurrent enabled="false"/> 
        </runtime> 
    </configuration> 
 
 

    انظر أيضا
  

 وعملیة تفريغ Garbage Collectionلمزيد من التفاصیل حول المجموعة 
  .الفئات والكائناتالذاكرة، راجع الفصل الثالث 

  
  

  :<add>الوسم 
كمـسار  ( لوضع قيم خاصة بك قد يحتاجها برنامجـك          >appSettings<تستطيع استخدام الوسم    

، يمكنك )الخ...ملفات قواعد البيانات، اعدادات واجهة الاستخدام، بيانات اخرى خاصة ببرنامجك،     
  :لملف التهيئة <appSettings> في داخل القسم الفرعي <add>استخدام الوسم 

 
<?xml version="1.0" encoding="UTF-8"?> 
    <configuration> 
        <appSettings> 
            <add key="Site" value="www.dev4arabs.com" /> 
            <add key="DBPath" value="C:\Folder\Data.MDB" /> 
            <add key="Show Startup Window" value="False" /> 
        </appSettings> 
    </configuration> 

  
تستطيع قراءة هذه الاعدادات والخاصة بك في أي وقت من شيفراتك المـصدرية لحظـة تنفيـذ                 

 System.Configuration.ConfigurationSettings.AppSettingsالبرنامج باستخدام الفئة    
  :بهذا الشكل

Imports System.Configuration.ConfigurationSettings 
 
Module Module1 
    Sub main() 
        ' www.dev4arabs.com 
        ArabicConsole.WriteLine(AppSettings("Site")) 
 
        ' C:\Folder\Data.MDB 
        ArabicConsole.WriteLine(AppSettings("DBPath")) 
 
        ' False 
        ArabicConsole.WriteLine(AppSettings("Show Startup Window")) 
    End Sub 
End Module 

  
   NET Framework Configuration.استخدام الأداة

 NET.ان كنت شخص لا يفضل تحرير ملفـات التهيئـة يـدويا، يمكنـك الاسـتعانة بـالأداة       

Framework Configuration ) يار الرمـز  ، والتي تصل إليها باخت)4-11شكلMicrosoft 

.NET Framework Configuration من المجموعة Administrative Tools   فـي لوحـة 
  .Control Panelالتحكم 

  

  
  .NET Framework Configuration. الأداة :4-11شكل 

    ملاحظة

  
 ھي برنامج يسمى في عـالم  NET Framework Configuration.الأداة 

Windows  بــ Microsoft Management Console snap-in .   وھـذا النـوع
 Windows NT ،Windowsمــن البــرامج لا يعمــل إلا تحــت الإصــدارات 

2000 ،Windows XPوما بعدھا ،.  
  

  
 السابقة، >gcConcurrent<الوسم وحتى اعرض لك مثالا لاستخدامها، دعني اكرر فقرة 

 فـي ملـف تهيئـة الجهـاز     Garbage Collectionيمكنك التحكم في طريقة تنفيذ المجموعة 
machine.config بالضغط بزر الفأرة الأيمن على الرمز My Computer ،في الشجرة اليسرى 

شـكل  ( My Computer Properties ليظهر لك صندوق الحـوار  Propertiesومن ثم اختيار 
11-4(.  
  

  
  .Garbage Collection تغيير سلوك تنفيذ المجموعة :4-11شكل 

 فـي   Applicationوتصحيح ملفات تهيئة التطبيقات بالضغط على الرمـز         يمكنك إضافة   
وبالنسبة للمجمعات المشتركة، فستجد ضـالتك بالـضغط علـى الرمـز     . الشجرة اليسرى للنافذة  

Configured Assemblies.  

  
  أدوات الترجمة، الربط، والتسجیل

بط الشيفرات المـصدرية  في هذا القسم اعرض لك مجموعة من الأدوات التي تستخدم لترجمة ور        
والتـي   (Visual Studio .NETوالوحدات المدارة والمجمعات، يمكنك استخدام بيئة التطـوير  

لإنجاز معظم المهام المطلوبة، أو استخدام هذه الأدوات مباشرة من          ) تستخدم بدورها هذه الأدوات   
  .Command Promptموجه الأوامر 

 Commandوات عن طريق موجه الأوامـر  قد تستغرب مدى الجدوى من استخدام الأد

Prompt عوضا عن بيئة التطوير Visual Studio .NET السبب في ذلك هو ان بعض المهام ،
لا يمكن إنجازها بالاعتماد على بيئة التطوير فقط، من أمثلة هذه المهام إنجاز مجمع متعدد الملفات، 

  .رها من الأمور اخرى، وغيNET.أو ترجمة شيفرات مصدرية من لغات برمجة 
 corvars.bat فلا تنسى تنفيذ الملف Command Promptان قمت بتشغيل موجه الاوامر 

ــد    ( ــي المجل ــده ف ــذي تج  X:\Program Files\Microsoft Visual Studioوال

.NET\FrameworkSDK\Bin (     حيث يقوم بتحميل مساراتPaths     هذه الأدوات ويسهل عليك 
 إلى تنفيذ هذا الملف ان كنت قد شغلت نافذة موجه الأوامر من             الوصول لها، مع ذلك لست بحاجة     

 الموجود في المجموعة البرمجيـة  Visual Studio .NET Command Promptخلال الرمز 
Microsoft Visual Studio .NET بقائمة Start ) 6-11شكل(.  

  

  
  .Visual Studio .NET Command Promptتشغيل موجه الأوامر من خلال : 6-11شكل 

    ملاحظة
  

 بتوفیر ھذه الخدمات الإضافیة من خلال Microsoftلا اعلم لماذا لم تقم  
 نفسھا، بدلا من الرجوع إلى أسالیب Visual Studio .NETبیئة التطوير 

  ! واستخدام موجه الاوامرVisual Programmingما قبل البرمجة المرئیة 
  

  
 لمزيد من NET Documentation.تندات فيما يلي عرض سريع لهذه الأدوات، راجع مس

  .التفاصيل حول طرق استخدامها
  

  VBC.EXEالمترجم 
 والذي يقوم بترجمة Visual Basic Compiler هو مترجم اللغة الأساسي VBC.EXEالمترجم 

باسـم  ) باستخدام المفكـرة مـثلا  (أنشئ ملف . ILالشيفرات المصدرية إلى شيفرات مكتوبة بلغة      
test.vbيه هذه الشيفرة المبسطة واكتب ف:  

 
' test.vb في الملف 
Imports System 
 
Module Module1 
    Sub Main 
        Console.WrilteLine ("Welcome") 
    End Sub 
End Module 

  
 من موجـه  VBC.EXEيمكنك ترجمة الملف السابق وتحويله إلى وحدة مدارة باستخدام المترجم   

  :الأوامر بهذه الصيغة
 
C:\>vbc test.vb 
 

    ملاحظة
  

في الحقیقـة الأمـر الـسابق يـؤدي إلـى إنـشاء مجمـع أحـادي الملـف،            
ولكني استخدمت المصطلح وحدة مدارة لأبین لـك انـه فـي حـال كـون              
المجمع يحتوي على وحدة مدارة واحدة فقـط، فسیـصبح المـصطلحین             

  .مجمع ووحدة مدارة مترادفین تقريبا

  : أكثر من ملف، فستستطيع إرفاقها بكتابة أسمائها جميعاان كان برنامجك يحتوي على
 
C:\>vbc test.vb test2.vb test3.vb 

  
، وان اردت تغييـر  test.exeسيكون اسم ملف الوحدة المدارة مثل اسم الملف المـصدري الاول      

  ):لترمز كافة الملفات* والتي تمكنك من استخدام النجمة  (out/الاسم استخدم الوسيطة 
 
C:\>vbc /out:myprog *.vb 

  
 Console يمثـل مجمـع مـن النـوع     myprog.exeالامر السابق سينشئ ملف تنفيذي باسم 

Application .  مع ذلك، تستطيع تحديد نوع المجمع اما تطبيـقWindows Application أو 
  ):t/يمكنك اختصارها إلى  (target/ باستخدام الوسيطة Libraryمكتبة 

 
( winprog.exe قتطبی ) 
C:\>vbc winprog.vb /target:winexe 
 
( mylib.exe مكتبة) 
C:\>vbc mylib.vb /t:library 

  
في نفـس   ولكنها   في الأمثلة السابقة لإنشاء وحدات مدارة        VBC.EXEاستخدمنا المترجم   

، مع ذلك تـستطيع  Single File Assembly مضمونه ومشمولة في مجمع أحادي الملفالوقت 
 لتستفيد منها لاحقا باستخدام الـرابط  دون تضمينها في مجمع   لى وحدات مدارة    ترجمة الشيفرات إ  

AL.EXE يمكنك عمل ذلك بتحديد النوع ،module مع الوسيطة /target:  
 
C:\>vbc test.vb /target:module 

  
ولا يـشمل  ( يمثل شيفرات الوحدة المدارة فقط  test.netmoduleالأمر السابق سينشئ ملف باسم      

، وان كانت الوحدة المدارة تعتمد أو تستخدم انواع بيانات لوحدات مدارة اخرى، عليـك               )المجمع
  :addmodule/اضافة هذه الوحدات المدارة باستخدام الوسيطة 

 
C:\Test>vbc test.vb /target:module /addmodule:other.netmodule 

  
مر في القراءة وتابع الفقرة التالية وحتى ترى كيف يمكنك الاستفادة من ملفات الوحدات المدارة، است

  .AL.EXEالرابط 
  

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


 NET Framework  408.إطار عمل : الثانيالجزء 
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  AL.EXEالرابط 
، وهو الوسيلة الوحيدة التي تمكنـك  Assembly Linker هو رابط المجمعات AL.EXEالرابط 

الدور الرئيسي الذي يقوم به . Multiple File Assembliesمن انجاز مجمعات متعددة الملفات 
  . مختلفة وتكوين مجمعNET.جزة بلغات هو ربط الوحدات المدارة والمن

 تحديد الوحدات المدارة التي ترغب في ربطها لتكوين مجمع AL.EXEيتطلب منك الرابط    
  :متعدد ملفات

 
C:\>al file1.netmodule file2.netmodule /out:myass.dll 

  
، file1.netmoduleمر السابق يقوم بإنشاء مجمع متعدد الملفات يحتوي على ثـلاث ملفـات              الا

file2.netmodule   و ،myass.dll )  لا تنسى ان هذا المجمع مجمع متعدد الملفات   . )7-11شكل
 لابـد ان    file2.netmodule و file1.netmodule ملفات الوحدات المدارة     -بعبارة أخرى –أي  

  .myass.dllتكون في نفس مجلد الملف 
  

  
  .مجمع متعدد الملفات: 7-11شكل 

  
    ملاحظة

  
 بمعرفـة كافیـة، الملـف    NET Documentation. مـستندات  حتى تواجـه 

myass.dll  مرشــد  الــسابق ھــو الملــف الرئیــسي للمجمــع ويــسمى
  .Assembly's Manifestالمجمع 

  
  

، تستطيع تكوين مجمعات  Libraryالمجمع السابق هو مجمع أسلوب تنفيذه من النوع مكتبة          
 أو t:exe/ بإرسال الوسيطات Windows Application أو Console Applicationمن النوع 

/t:winعلى التوالي :  

( Console Application ) 
C:\>al file.netmodule /t:exe /out:myprog.exe 
 
( Windows Application ) 
C:\>al file.netmodule /t:win /out:myprog.exe 

  
، وذلـك لان  Entry Pointمع ذلك، ستظهر لك رسالة خطأ مفادها عدم تحديد نقطة بداية التنفيذ 

 لابد ان يكون لهـا  Windows Application و Console Applicationالمجمعات من النوع 
 الذي تود جعله نقطة البداية للمجمع بإرسال ()Sub Mainيمكنك تحديد اسم الإجراء . نقطة بداية
  :main/الوسيطة 

 
C:\test>al file.netmodule /t:exe /out:myprog.exe /main:Module1.main 

  
    ملاحظة

  
 ساعات متواصلة من التجـارب الغیـر ناجحـة          4معلومة كلفتني أكثر من     
 وھـي ان اسـم إجـراء    NET Documentation.لن تجدھا في مستندات 

 يتعامل معـه بحـساسیة    main/نقطة البداية الذي تحدده مع الوسیطة       
ــة البرمجــة  case-sensitiveللأحــرف   Visual Basic .NET رغــم ان لغ

  .case insensitiveلیست حساسة 
  

  
  :مثال تطبيقي

والان سأريك تطبيق عملي لإنشاء مجمع متعدد الملفات يشمل وحدات مدارة مترجمة من شيفرات              
أنـشئ ملفـين   . Visual Basic .NET و Visual C# .NETمصدرية بلغتين مختلفين همـا  

file1.vb و file2.vbواكتب بهما هذه الشيفرة :  
 ' file.vb في الملف 

Public Module MainModule 
    Public Sub Main () 
        Sub2() 
    End Sub 
End Module 
 
' file2.vb في الملف 
Module mdlFile2 
    Sub Sub2 () 
        System.Console.WriteLine ("VB code works.") 
        file3.clsfile3.Sub3() 
    End Sub 
End Module 

 والذي يعرض رسالة مفادهـا ان  ()Sub2 السابق يقوم باستدعاء الإجراء ()Sub Mainالإجراء 
، وبعد ذلك يقوم الإجـراء باسـتدعاء الإجـراء    Visual Basic .NETهذه شيفرة مكتوبة بلغة 

Sub3() وهو إجراء سنكتبه بلغة C#:  
 // file3.cs في الملف 

namespace file3 
{ 
    class clsfile3 
    { 
        static public void Sub3() 
        { 
            file4.clsfile4.Sub4(); 
        } 
    } 
} 
 
 
// file4.cs في الملف 
namespace file4 
{ 
    class clsfile4 
    { 
        static public void Sub4() 
        { 
            System.Console.WriteLine ("C# code works."); 
            System.Console.Read(); 
 
        } 
    } 
} 
 
 

    ملاحظة
  

، كمـا  )C Sharpاختـصار  . (cs تكون امتداداتھا #Cالملفات المنجزة بلغة 
  .CSC.EXEان مترجم اللغة الخاص بھا ھو 

  
  

) C#) CSC.EXE ، وابدأ فورا باستخدام مترجمة لغـة C:\Testاحفظ الملفات الأربعة في مجلد    
 لأننا نريد   target:module/، ولا تنسى استخدام الوسيطة      file4.csc و   file3.csلتترجم الملفات   

  :وحدة مدارة دون مجمع
 

C:\Test>csc /target:module *.cs 

 و file.vbوالان حان دور الملفين . file3.netmoduleالامر السابق سينشئ ملف الوحدة المدارة 
file2.vb رجمتهما باستخدام المترجمة   وتVBC.EXE    ولكن من الـضروري إضـافة مرجـع ،

 file2.vb، وذلك لان الشيفرة المكتوبة في الملف        addmodule/الوحدة المدارة السابقة باستخدام     
  : والذي يتبع لوحدة مدارة خارجية()Sub3تستدعي الإجراء 

 
C:\Test>vbc /target:module *.vb /addmodule:file3.netmodule 

  
 واربط كلا الوحدات المـدارة      AL.EXEتوكل على االله الذي لا تضيع ودائعه، واستخدم الرابط          

 ()Main، نقطـة بدايتـه الإجـراء    )Console Application) /t:exeلتكون مجمع من النوع 
)/main:MainModule.Main( واسم ملفه الرئيسي ،)/out:myprog.exe:(  

 
C:\test>al file.netmodule file3.netmodule  /t:exe /out:myprog.exe 
/main:MainModule.Main 

  
تم إنشاء المجمع المنجز بلغتي برمجة، يمكنك الان كتابة اسمه لتنفيذه كما كنا نفعل       ! مبروك

  :، لتكون المخرجات بهذا الشكلMS-DOSقبل عشرات السنين تحت أنظمة 
 
C:\test>myprog 
 
VB code works. 
C# code works. 
 
C:\test> _ 

  
  . استيعاب الخطوات السابقة على)8-11الشكل (ان لم تفهم شيئا من هذا المثال، عسى ان يساعدك 

  .myprog.exe، و File.netmodule ،file3.netmodule مجمع متعدد الملفات :8-11شكل 

  
  SN.EXEالمسجل 

اتك، الخطوة الاولى التي  لمجمعStrong Names لتسجيل اسماء قوية SN.EXEيستخدم المسجل 
  ):k–باستخدام الوسيطة ( وحفظه في ملف منفصل Partial keyمفتاح مركب شاء تحتاجها هي إن

 
C:\>sn –k mykey.snk 

  
 والمفتـاح الخـاص   Public Keyوهو يشمل المفتاح العام (الامر السابق سينشئ مفتاح مركب 

Private Keyوالذي يمثل توقيع المنتج أو الشركة .(  
، عليك  AL.EXE أو الرابط    VBC.EXEان كنت تود ترجمة البرنامج باستخدام المترجم        

  : لحظة الترجمة أو الربطkeyfile/ارسال الملف مع المدخل 
 
C:\>vbc /out:myprog *.vb /keyfile:mykey.snk 

  
، فأضف الشيفرة التالية في الملـف  Visual Studio .NETاما ان كنت في داخل بيئة التطوير 

AssemblyInfo.vb:  
 
<Assembly: AssemblyKeyFile("mykey.snk") 

  GACUTIL.EXEالمسجل 
 باســتخدام الأداة Shared Assemblyيمكنـك تــسجيل المجمــع وجعلـه مجمعــا مــشتركا   

GACUTIL.EXE              واستخدامها سهل جدا، فكل ما تطلبه منك اسم ملف المجمع المراد تسجيله ،
  :i/ مع إرسال الوسيطة GACفي الوحدة العامة للمجمعات 

 
C:\>gacutil /i mylib.dll 

  
  :l/ بإرسال الوسيطة GACيمكنك عرض جميع المجمعات في الوحدة العامة 

 
C:\>gacutil /l 

  
 وليس اسم ملفـه  بكتابة اسم المجمع  GACأخيرا، تستطيع حذف مجمع مشترك من وحدة        

  :u/مع إرفاقه بالوسيطة 
 
C:\>gacutil /u mylib 

  
 باختلاف إصداراتها، mylib السابق، انه يحذف كافة المجمعات التي تحمل الاسم u/المدخل يعيب 

  ):u/ وليس u–استخدم الوسيطة (لذلك يفضل تحديد المجمع الذي تود حذفه بذكر إصداره 
 
C:\>gacutil –u mylib, ver=1.2.3432432 

ذه الأدوات في مشاريعك الكبيرة، حيث عليـك       لا تعتمد على هذا الفصل ان كنت تنوي استخدام ه         

 وقراءة التفاصيل الدقيقة حول المجمعات والمترجمات وطريقة التعامل         MSDNالعودة إلى مكتبة    
معها، كما عليك استخدام الأدوات السابقة بحذر شديد فهي تعتمد على نوعيـة الـشيفرات التـي                 

. ى شيفرات مبسطة هدفها توضيح الفكرة فقطتستخدمها في برامجك، اما الأمثلة السابقة فليست سو  
 يمكنك من الاسـتعلام والحـصول علـى    Reflection Classesفئات الانعكاس الفصل التالي 

  .معلومات حول المجمعات والوحدات المدارة التي أنشأتها

  الثاني عشرالفصل  12

 

  

  
 Reflectionفئات الانعكاس 

Classes  
  

ات مشمولة فـي مجـال الاسـماء     هي مجموعة من الفئReflection Classesفئات الانعكاس 
System.Reflection        غرضها التعامل مـع المجمعـات Assemblies     الوحـدات المـدارة ،
Managed Modules  الفئات والكائنـات ،Classes and Objects  الواجهـات ،Interfaces ،
الحقول، الطرق، الخصائص، (، وأعضاء البيانات Enums او Structuresالتركيبات سواء كانت 

 ).والأحداث

بعـد اسـم الكـائن، سـتلاحظ ان قائمـة           "." في محرر الشيفرات، بعد ان تكتب نقطـة         
IntelliSense    تظهر لك كافة الأعضاء التابعة للكائن، كيف تمكنـت بيئـة التطـوير Visual 

Studio .NETمن معرفة هذه الأعضاء؟ الجواب هو ما يفصله لك هذا الفصل .  
  

    ملاحظة
  

ــات الان  ــا ان فئـــ ــماء   بمـــ ــال الاســـ ــي مجـــ ــشمولة فـــ ــاس مـــ عكـــ
System.Reflection     فمـــن البـــديھي ســـتقوم اســـتیراده لاختـــصار ،
  :الشیفرات البرمجیة

 
Imports System.Reflection 

  

  
  التعامل مع المجمعات والوحدات المدارة

. Module و Assembly: في هذا القسم اعرض لك فئتين تمثل المجمعات والوحدات المدارة هما
 NET.البحث عن كافة تفاصـيل خـصائص وطـرق هـذه الفئـات فـي مـستندات        يمكنك 

Documentationحيث لن تجد هنا الا عرض سريع ومختصر لأبرز أعضائها .  

  Assemblyالفئة 
لا يمكنك انشاء كائن مباشرة     .  مجمع مستقل  -كما هو واضح من اسمها    – تمثل   Assemblyالفئة  

 لن تنشئ مجمع جديد، بل الذي ستفعله هو العودة بمرجع  من هذه الفئة، وذلك لأنكNewباستخدام  
  :()GetExecutingAssemblyللمجمع الحالي باستدعاء الطريقة المشتركة 

 
Dim ass As [Assembly] 
 
ass = [Assembly].GetExecutingAssembly() 

  
  :()LoadFromأو العودة بمرجع لمجمع عن طريق اسم ملفه بإرساله إلى الطريقة المشتركة 

 
Dim ass As [Assembly] 
 
ass = [Assembly].LoadFrom("C:\myLib.dll") 

  
  .MSDNأو مجموعة من الطرق الأخرى والتي تجد تفاصيلها في مكتبة 

  
    ملاحظة

  
 Visual للغـة البرمجـة   keyword ھي كلمة محجوزة Assemblyالكلمة 

Basic .NET من ھذه  عند تصريح متغیر ] و [، لذلك استخدمت الأقواس
 :الفئة

 
Dim ass As [Assembly] 

  
  

بعد إنشاء كائن يمثل مجمع، تستطيع الاستعلام عن مجموعة كبيرة من محتوياتـه، كاسـم               
) Public Keyيشمل الاسم، رقم الاصدار، الاعدادات الإقليمية، والمفتاح العـام  (المجمع الكامل 

ن المجمع قادم من الوحدة العامـة       ، كما يمكنك معرفة ما اذا كا      FullNameعن طريق الخاصية    
 التي  Location، والخاصية   GlobalAssemblyCache عن طريق الخاصية     GACللمجمعات  

  :تعود بمسار ملف المجمع
 
Dim ass As [Assembly] 
 
ass = [Assembly].GetExecutingAssembly 
ArabicConsole.WriteLine(ass.FullName) 
ArabicConsole.WriteLine(ass.GlobalAssemblyCache) ' False 
ArabicConsole.WriteLine(ass.Location)  ' C:\test.dll 

  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  

 والتـي تعـود بكـائن مـن النـوع         EntryPointمن الخصائص الاخـرى، الخاصـية       
Refelection.MethodInfoتمثل إجراء نقطة البداية للمجمع :  

 
Dim ass As [Assembly] 
Dim mthd As MethodInfo 
 
ass = [Assembly].GetExecutingAssembly 
mthd = ass.EntryPoint 
 
ArabicConsole.WriteLine(mthd.Name) ' main 

  
 ان كان المجمـع مـن النـوع مكتبـة           Nothing السابقة بالقيمة    EntryPointستعود الخاصية   

Libraryوذلك لان هذا النوع من المجمعات لا يحتوي على نقطة بداية ،.  
  

    انظر أيضا
  

لمعرفــة مــاذا يقــصد  Assembliesالمجمعـات   الفــصل الــسابق راجـع 
 فـسأتحدث  MethodInfo، وبالنـسبة للفئـة   Entry Pointبنقطة البدايـة  

  .عنھا لاحقا في ھذا الفصل
 

  
 تمثل جميـع البيانـات المعرفـة        Type تعود بمصفوفة من النوع      ()GetTypesالطريقة  

  :مجمعفي ال) الخ...الفئات، الواجهات، التركيبات، (
 
Dim ass As [Assembly] 
Dim t As Type 
ass = [Assembly].LoadFrom("C:\myLib.dll") 
 
For Each t In ass.GetTypes() 
    ArabicConsole.WriteLine(t.Name) 
Next 

  
بما فيها التـي لا     ( السابقة تعود بجميع الأنواع المعرفة في المجمع         ()GetTypesالطريقة  

، لـذلك يفـضل اسـتخدام     )Private أو   Friend كالمعرفة بمحدد الوصول     تستطيع الوصول لها  
 والتـي   Public والتي تعود بالأنواع المعرفة بمحدد الوصول        ()GetExportedTypesالطريقة  

  :يمكنك الوصول لها

Dim ass As [Assembly] 
Dim t As Type 
ass = [Assembly].LoadFrom("C:\myLib.dll") 
 
For Each t In ass.GetExportedTypes() 
    ArabicConsole.WriteLine(t.Name) 
Next 

  
 والذي تعرفنا عليه في الفصول الأولى مـن هـذا   Late Bindingاخيرا، الربط المتأخر 

 لتتمكن من إنشاء ()CreateInstanceالكتاب يستخدم فئات الانعكاس أيضا لإنجازه، لديك الطريقة 
  :رسال اسمه البرمجيكائن بالربط المتأخر وذلك با

 
Dim ass As [Assembly] 
ass = [Assembly].LoadFrom("C:\test\test.dll") 
Dim obj As Object = ass.CreateInstance("LibNamespace.PublicClass") 
 
 اثبات انه تم انشاء الكائن '
ArabicConsole.WriteLine(obj.GetType.FullName) ' 
LibNamespace.PublicClass 

  
  Moduleالفئة 

 والتابعة للمجمع، يمكنـك  Managed Modules فهي تمثل الوحدات المدارة Moduleاما الفئة 
، أو Assembly التابعـة للفئـة    ()GetModuleإنشاء كائن من هذه الوحدة باستدعاء الطريقـة         

  : والتي تعود بمصفوفة تمثل كافة الوحدات المدارة في المجمع()GetModulesالطريقة 
 
Dim ass As [Assembly] 
ass = [Assembly].LoadFrom("C:\test\test.dll") 
 
Dim mdl As [Module] 
 
For Each mdl In ass.GetModules 
    ArabicConsole.WriteLine(mdl.Name) 
Next 

  
 ()GetTypes على الطريقة    Module، تحتوي الفئة    Assemblyوكما هو الحال مع الفئة      

  . الانواع المعرفة في الوحدة المدارة وليس كامل المجمعولكنها تعود بجميع

  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  

 التعامل مع أنواع البیانات
في هذا القسم اعرض لك المزيد من الطرق والخصائص التي تمكنك من الاستعلام عن الطـرق                 

ولكن قبل ذلك، احتاج إلى تعريفـك  . والخصائص وكافة أعضاء البيانات التي تريدها لحظة التنفيذ    
  . والتي اذكر أننا استخدمناها سابقا في مواقع متفرقة من هذا الكتابSystem.Typeفئة اكثر بال

  
  System.Typeالفئة 

 ما عـدا  System. Reflection مشمولة في مجال الاسماء -كما ذكرت–جميع فئات الانعكاس 
ن  مSystem.Typeمع ذلك، تصنف الفئة . System فهي في مجال الاسماء الجذري  Typeالفئة  

 Primitiveفئات الانعكاس ايضا، فهي تمثل نوع معين من البيانات، قد يكون هذا النوع أولـي  

Type)   كـInteger  ،Long  ،String  ،Boolean... أو نـوع معـرف كفئـة         )الخ ،Class ،
  .Interface، واجهة Enum أو Strucuterتركيب 

ا ستحاول اسناد مرجع إلـى   مباشرة، وانم Newلن تنشئ كان جديد من هذه الفئة باستخدام         
 والتابعة  ()GetTypes باستدعاء الطريقة    -مثلا– بأساليب عديدة، منها     Typeالكائنات من النوع    

  : السابقةAssemblyللفئة 
 
Dim ass As [Assembly] 
Dim t As Type 
ass = [Assembly].LoadFrom("C:\myLib.dll") 
 
For Each t In ass.GetTypes() 
    ArabicConsole.WriteLine(t.Name) 
Next 

  
 Visual والتابعة للغـة البرمجـة   ()GetTypeأو تسهل علي نفسك المهمة اكثر باستخدام الدالة 

Basic .NET:  
 
Dim t As Type 
 
t = GetType(Integer) 
ArabicConsole.WriteLine(t.FullName) ' System.Int32 

  
 الفئـة   انالفئـات الأساسـية   الـسادس    لو لم تخونك الذاكرة فستذكر في بدايـة الفـصل         

System.Object)   والتي ترث منها جميع فئات إطار عمـل.NET Framework الأخـرى  (

 يمثل نوع الكائن الحـالي      Type والتي تعود بكائن من النوع       ()GetTypeتحتوي على الطريقة    
  :Late Bindingحتى لو كان الكائن منشئ بالربط المتأخر 

 
Dim t As Type 
Dim obj As Object = New TestClass() 
 
t = obj.GetType() 
 
ArabicConsole.WriteLine(t.FullName) ' MyNamespace.TestClass 

  
 فـي الفئـة   Overloads السابقة تم إعـادة تعريفهـا       ()GetTypeالمزيد ايضا، الطريقة    

System.Type بحيث تمكنك من ارسال قيمة حرفية من النوع Sring:  
 
Dim t As Type 
t = Type.GetType("System.Double") 
 
Console.WriteLine(t.FullName) 

  
-Caseدعني الفت انتباهك بان الاسم الحرفي الذي ترسله للطريقـة حـساس لحالـة الاحـرف                

Sensetive كما عليك استخدام الاسماء الأصلية لإطار عمل ،.NET Framework)  كـInt32 ،
Int64 ،DateTime.... عوضا عن الأسماء الخاصة بلغة البرمجة ) الخVisual Basic .NET 

  ).الخ ...Integer ،Long ،Dateكـ (
 والتي ترسل معها    ()GetTypeArray، لديك الطريقة    ()GetTypeبالإضافة إلى الطريقة    

  :مصفوفة لكائنات، لتعود بنوع كل كائن من هذه المصفوفة على حده
 
Dim X() As Object = {"3.5 ,100 ,"عباس} 
Dim t() As Type = Type.GetTypeArray(X) 
 
 
ArabicConsole.WriteLine(t(0).FullName) ' System.String 
ArabicConsole.WriteLine(t(1).FullName) ' System.Int32 
ArabicConsole.WriteLine(t(2).FullName) ' System.Double 

  
  :COM  المكوناتخاص لمبرمجي

 المخضرمين، فدعني اهمس في اذنك وأخبرك ان الفئـة          COMالمكونات  ان كنت من مبرمجي     
Type    يمكن لها ان تحمل كائناتها انواع بيانات لفئات منجزة بتقنية COM   تستطيع عمـل ذلـك ،

برمجـة   ( ProgIDمعـرف    وارسـال قيمـة      ()GetTypeFromProgIDباستدعاء الطريقة   
  ):تاب موضوع أقدم من ان يذكر في هذا الكCOMالمكونات 

' Microsoft Word استخدام احد فئات 
Dim word As Type = Type.GetTypeFromProgID("Word.Application") 
 
ArabicConsole.WriteLine(word.FullName) ' System.__ComObject 

  
  خصائص إضافیة

د توجد عشرات الخصائص للقراءة فقط التي تمكنك من الحصول على تفاصيل اكثر للنوع المـسن             
 التي تعود بالاسـم البرمجـي   Name، من هذه الخصائص الخاصية      Typeإلى الكائن من النوع     

 التي تشمل مجال الاسماء في الاسم البرمجـي للنـوع، الخاصـية             FullNameللنوع، الخاصية   
Module التي تعود بالوحدة المدارة Managed Module   التي عرف بها النـوع، والخاصـية 

Assembly بالمجمع الذي عرف في النوع التي تعود:  
 
Dim t As Type = GetType(String) 
 
ArabicConsole.WriteLine(t.Name) ' String 
ArabicConsole.WriteLine(t.FullName) ' System.String 
ArabicConsole.WriteLine(t.Module) ' CommonLanguageRuntimeLibrary 

 
، IsClass ،IsInterface:  هـي Boolean خصائص منطقية أخرى تعود بقيم من النـوع    

IsEnum  ،IsValueType          والتي تكتشف فيها ما اذا كان النـوع يمثـل فئـة Class  واجهـة ،
Interface تركيب من النوع ،Enum نوع ذات قيمة ،Value Type    أو تركيـب مـن النـوع 
Structureعلى التوالي :  

 
Dim t As Type = GetType(TestClass) 
 
ArabicConsole.WriteLine(t.IsClass)  ' True 
ArabicConsole.WriteLine(t.IsValueType) ' False 

  
يمكنك معرفة ايضا محدد الوصول الذي أستخدم لتعريف النـوع عـن طريـق الخاصـية       

IsPublic  ــة ــود بالقيم ــي تع ــول  True الت ــدد الوص ــان مح ــية Public ان ك ، الخاص
IsNestedPrivate   التي تعود بالقيمة True     ان كـان محـدد الوصـولPrivate  والخاصـية ،

IsNestedAssembly     التي تعود بالقيمة True        ان كان محـدد الوصـول Friend   الخاصـية ،
IsNestedFamily     التي تعود بالقيمة True       ان كان محدد الوصـول Protected   والخاصـية ،

IsNestedFamORAssem التي تعود بالقيمة True    ان كـان محـدد الوصـول Protected 

Friend.  

وبالنسبة للعلاقات الوراثية بين الفئات، فيمكنك معرفة ما اذا كان النوع غير قابل للوراثـة               
NotInheritable      ان كانت قيمة الخاصيةIsSealed   هي True      وعلى العكس تعود الخاصـية ،

IsAbstract بالقيمة True ان كان النوع MustInherit.  
  . ان اردت معرفة اكبر قدر من الخصائصMSDNنصيحتي لك بان تراجع مكتبة 

  
 التعامل مع الأعضاء

في القسم السابق عرضت لك طرق وخصائص تمكنك من الاستعلام عن انواع البيانات، وفي هذا               
) حقول، خصائص، طرق، وأحداث(القسم سنفصل في هذه الاستعلامات بحيث تعرض لك أعضاء         

في العـودة إلـى مراجـع       )  بانها أصبحت مملة   التي أحس (أعيد واكرر نصيحتي السابقة     . النوع
MSDNللحصول على كافة الطرق والخصائص .  

  
  MemberInfoالفئة القاعدية 

فـي  ) حقل، خاصية، حدث، أو طريقة    (  سواء كان     Member تمثل عضو    MemberInfoالفئة  
، FieldInfo ،PropertyInfo: النوع، وهي بدورها تعتبر الفئة القاعدية لاربع فئات اخرى هـي       

EventInfo و ،MethodInfo ) 1-12شكل(.  
  

  
  MemberInfo الفئة القاعدية :1-12شكل 

حتى تتمكن من استيعاب الفئات الاربعة المشتقة، سيسهل عليك كثير إن أتقنت الفئة القاعدية        
MemberInfo      ولكي تتعامل مع كائنات الفئة القاعدية ،MemberInfo  فلن تتمكن من إنـشائها ،

 وانما عليك الحصول إلى مرجع العضو في النوع، يتم ذلك باسـتدعاء الطريقـة               Newم  باستخدا
GetMembers()والتي بكافة الأعضاء التابعة للنوع :  

 
Dim T As Type = GetType(Integer) 
Dim m As MemberInfo 
 
For Each m In T.GetMembers 
    ArabicConsole.WriteLine(m.Name) 
Next 

  
  :Integerشيفرة السابقة ستكون أعضاء النوع  المخرجات 

 
MaxValue 
MinValue 
ToString 
GetTypeCode 
ToString 
CompareTo 
GetHashCode 
Equals 
ToString 
ToString 
Parse 
Parse 
Parse 
Parse 
GetType 

  
 ركزت في المخرجات السابقة، ستلاحظ ان بعض الأعضاء قد تكررت، والسبب في ذلـك ان                ان

 أكثر من مرة، لذلك حتى تتجنب     Overridesضاء تابعة لفئات مشتقة أو تم إعادة قيادتها         هذه الأع 
 ما هي إلا تركيب مـن  ()GetMembersالتكرار تستطيع إرسال مجموعة من القيم إلى الطريقة        

  :، والذي يمكنك من حصر الأعضاء على عدة أوجه، منهاBindingFlags باسم Enumالنوع 
ــة  § ــة  BindingFlags.Publicالقيم ــة، والقيم ــضاء العام ــال للأع ــصر المج  تح

BindingFlags.NonPublicللأعضاء الغير عامة . 

، والقيمة Shared Members تعود بالأعضاء المشتركة BindingFlags.Staticالقيمة  §
BindingFlags.Instance للأعضاء التقليدية Instance Members. 

د بالأعضاء المعرفة فـي النـوع        تعو BindingFlags.DeclaredOnlyالقيمة   §
  .اي لا تشمل أعضاء الفئات القاعدية–الحالي فقط 

  
، على المستوى العام Instance Membersفمثلا، الشيفرة التالية ستعود بالأعضاء التقليدية 

Publicوتشمل الأعضاء المصرحة في النوع ومتجاهلة أعضاء الفئات القاعدية ،:  
 
Dim T As Type = GetType(Integer) 
Dim m As MemberInfo 
 
For Each m In T.GetMembers(BindingFlags.Public _ 
    Or BindingFlags.Instance Or _  
    BindingFlags.DeclaredOnly) 
 
    ArabicConsole.WriteLine(m.Name) 
Next 

  
  :مخرجات الشيفرة السابقة ستكون شيئا مثل

 
ToString 
GetTypeCode 
ToString 
CompareTo 
GetHashCode 
Equals 
ToString 
ToString 

  
عليك معرفة ان تكرار بعض الأعضاء في المخرجات السابق ناتج عن كونهـا معـاد تعريفهـا                 

Overloadsفي نفس النوع، ولا تتبع للفائت القاعدية .  
  

  :الطرق والخصائص
الاستعلام حـول    على عشرات الطرق والخصائص التي تمكنك من         MemberInfoتحتوي الفئة   

 التي تعود باسم العضو، الخاصية Nameالأعضاء والخاصية بالنوع المرسل، لديك مثلا الخاصية 
MemberType      حقل  ( التي تعود بنوع العضوField   طريقة ،Method    خاصـية ،Property 

  . المستخدمة مع العضوAttribute التي تعود بالمواصفة Attributes، والخاصية )الخ...
خصائص التي تمكنك من معرفة محدد الوصول المستخدم لتعريف العضو الخاصـية            من ال 

IsPublic     التي تعود بالقيمة True      ان كان محدد الوصول Public   الخاصية ،IsPrivate   التـي 

 التي تعود بالقيمة    IsAssembly، والخاصية   Private ان كان محدد الوصول      Trueتعود بالقيمة   
True    ل   ان كان محدد الوصوFriend   الخاصية ،IsFamily     التي تعود بالقيمة True    ان كـان 

 ان True التي تعود بالقيمـة  IsFamilyORAssembly، والخاصية Protectedمحدد الوصول   
  .Protected Friendكان محدد الوصول 

 أو لا عـن طريـق الخاصـية         Sharedاخيرا، يمكنك معرفة ما اذا كان العضو مشترك         
IsStatic.  

  
  ل مع الحقولالتعام
 على مجموعة من    FieldInfo تابع لنوع بيانات، تحتوي الفئة       Field تمثل حقل    FieldInfoالفئة  

ولكي تتعامل مع   . MemberInfoالخصائص والطرق الخاصة بها، أو المشتقة من الفئة القاعدية          
إلى مرجع   وإنما عليك الحصول     New، فلن تتمكن من إنشائها باستخدام       FieldInfoكائنات الفئة   

 والتي تعود بكافة الحقول التابعة للنوع ()GetFieldsالعضو في النوع، يتم ذلك باستدعاء الطريقة   
  ): التي تحتوي على اربعة حقولPersonبافتراض ان لدينا الفئة (

 
Class Person 
    Public Name As String 
    Public BirthDate As Date 
    Public Salary As Decimal 
    Public Address As String 
End Class 
… 
… 
… 
Dim T As Type = GetType(Person) 
Dim f As FieldInfo 
 
For Each f In T.GetFields() 
    ArabicConsole.WriteLine(f.Name) 
Next 

  
  :مخرجات الشيفرة السابقة ستكون

 
Name 
BirthDate 
Salary 
Address 

    ملاحظة
  

 وھـي   BindingFlags تستقبل وسیطة مـن النـوع        ()GetFieldsالطريقة  
  .()GetMembersتماثل وسیطة الطريقة 

 
  

  :الطرق والخصائص
 ان كـان    False التي تعود بالقيمة     IsNotSerialized الخاصية   FieldInfoمن خصائص الفئة    
 Type التي تعود بقيمة من النـوع        FieldType، والخاصية   Serializableالحقل قابل للتسلسل    

  :تمثل نوع الحقل
 
Dim T As Type = GetType(Person) 
Dim F As FieldInfo = T.GetField("BirthDate") 
 
ArabicConsole.WriteLine(F.FieldType.ToString) ' System.DateTime 
 
 

 و  ()SetValue تحتـوي علـى الطـرقتين        FieldInfoوللحديث عن الطـرق، فالفئـة       
GetValue()نك من إسناد أو قراءة قيمة إلى الحقل بأسلوب الربط المتـأخر   اللتان تمكناLate 

Binding.  
  

  التعامل مع الخصائص
 PropertyInfo تابعة لنوع بيانات، تحتوي الفئـة        Property تمثل خاصية    PropertyInfoالفئة  

. MemberInfoعلى مجموعة من الخصائص والطرق الخاصة بها، أو المشتقة من الفئة القاعدية 
 وإنما عليك New، فلن تتمكن من إنشائها باستخدام      PropertyInfoولكي تتعامل مع كائنات الفئة      

 والتي تعود   ()GetPropertiesالحصول إلى مرجع العضو في النوع، يتم ذلك باستدعاء الطريقة           
  : بكافة الخصائص التابعة للنوع

 
Dim T As Type = GetType(String) 
Dim P As PropertyInfo 
 
 
 
For Each P In T.GetProperties() 
    ArabicConsole.WriteLine(P.Name) 
Next 

  :مخرجات الشيفرة السابقة ستكون
 
Chars 
Length 
 
 

    ملاحظة
  

 BindingFlags تــستقبل وســیطة مــن النــوع ()GetPropertiesالطريقــة 
  .()GetMembersوھي تماثل وسیطة الطريقة 

 
  

  :الطرق والخصائص
 ان كانـت  True التي تعـود بالقيمـة    CanRead الخاصية   PropertyInfoص الفئة   من خصائ 

 ان كانت قابلة للكتابة، كما تحتوي على الخاصـية   CanWriteالخاصية قابلة للقراءة، والخاصية     
PropertyType التي تعود بقيمة من النوع Typeتمثل نوع الخاصية :  

 
Dim T As Type = GetType(String) 
Dim P As PropertyInfo = P.GetProperty("Length") 
 
ArabicConsole.WriteLine(P.PropertyType.ToString) ' System.Int32 
 

  
 و  ()SetValue تحتوي ايضا علـى الطـريقتين        FieldInfoوالحديث عن الطرق، فالفئة     

GetValue()تأخر  اللتان تمكناك من إسناد أو قراءة قيمة إلى الخاصية بأسلوب الربط المLate 

Binding .  والطريقتينGetGetMethod()   و GetSetMethod()       اللتان تعودان بكـائن مـن 
وإسناد القيمـة للخاصـية   ) Property Get( يمثلنا اجراء استرجاع القيمة MethodInfoالنوع 

)Property Set (–فكما تعلم ان الخصائص تحتوي على اجرائين:  
 
Property xxxx() As yyyy 
    Get 
 اجراء استرجاع القیمة '        
    End Get 
 
    Set(ByVal Value As yyyy) 
 اجراء اسناد القیمة '        
    End Set 
End Property 

  
  

PDF created with pdfFactory Pro trial version www.pdffactory.com

http://www. dev4arabs.com


 NET Framework  428.إطار عمل : الثانيالجزء 
  ــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــــ

  التعامل مع الطرق
 تابعة لنوع بيانات، تحتوي الفئة على مجموعة مـن          Method تمثل طريقة    MethodInfoالفئة  

ولكي تتعامل مع   . MemberInfoالمشتقة من الفئة القاعدية     الخصائص والطرق الخاصة بها، أو      
 وانما عليك الحـصول علـى   New، فلن تتمكن من إنشائها باستخدام MethodInfoكائنات الفئة  

 والتي تعود بكافة الطـرق      ()GetMethodsمرجع العضو في النوع، يتم ذلك باستدعاء الطريقة         
  :التابعة للنوع

 
Dim T As Type = GetType(Double) 
Dim M As MethodInfo 
 
For Each M In T.GetMethods 
    ArabicConsole.WriteLine(M.Name) 
Next 

  
  :مخرجات الشيفرة السابقة ستكون

 
ToString 
GetTypeCode 
ToString 
CompareTo 
GetHashCode 
Equals 
ToString 
IsInfinity 
IsPositiveInfinity 
IsNegativeInfinity 
IsNaN 
ToString 
Parse 
Parse 
Parse 
Parse 
TryParse 
GetType 
 
 

    ملاحظة
  

 وھي BindingFlags تستقبل وسیطة من النوع ()GetMethodsالطريقة 
  .()GetMembersتماثل وسیطة الطريقة 

  :الطرق والخصائص
 التي تعود بقيمة مـن النـوع        ReturnType الخاصية   MethodInfoمن خصائص الفئة    

Type    إن كانت  (القيمة التي تعود بها الطريقة       تمثل نوعFunction  والخاصـية  ) بكـل تأكيـد ،
IsAbstract     التي تعود بالقيمة True         ان كانت الطريقة مصرحة على انهـا MustOverride ،

  .Overridable ان كانت الطريقة مصرحة بـ True تعود ايضا بالقيمة IsVirtualوالخاصية 
 والتي تقـوم    ()Invoke تحتوي على الطريقة     MethodInfoوللحديث عن الطرق، فالفئة     

  .Late Bindingبتنفيذ الطريقة من خلال الربط المتأخر 
  

    ملاحظة
  

 مدعومــة أيــضا فــي الفئــة  MethodInfoمعظــم خــصائص وطــرق الفئــة 
PropertyInfoالسابقة والعكس صحیح .  

 

  
  التعامل مع الأحداث

 على مجموعـة مـن      EventInfoة، تحتوي الفئة     تابع لفئ  Event تمثل حدث    EventInfoالفئة  
ولكي تتعامل مع   . MemberInfoالخصائص والطرق الخاصة بها، أو المشتقة من الفئة القاعدية          

 وانما عليك الحصول على مرجع New، فلن تتمكن من انشائها باستخدام EventInfoكائنات الفئة 
 والتي تعود بكافة الأحـداث التابعـة   ()GetEventsالعضو في النوع، يتم ذلك باستدعاء الطريقة  

  ): التي تحتوي على ثلاثة أحداثTestClassبافتراض ان لدينا الفئة (للنوع 
 
Class TestClass 
    Event XXX() 
    Event YYY() 
    Event ZZZ() 
End Class 
… 
… 
… 
Dim T As Type = GetType(TestClass) 
Dim E As EventInfo 
 
For Each E In T.GetEvents 
    ArabicConsole.WriteLine(E.Name) 
Next 

  :مخرجات الشيفرة السابقة ستكون
 
XXX 
YYY 
ZZZ 
 
 

    ملاحظة
  

 وھـي  BindingFlags تستقبل وسیطة من النـوع  ()GetEventsالطريقة  
  .()GetMembersتماثل وسیطة الطريقة 

 
  

  :الطرق والخصائص
ص إضافية خاصة بهـا غيـر المـشتقة مـن الفئـة              على خصائ  EventInfoلا تـحتوي الفئة    

MemberInfo.  
 ()GetAddMethod تحتوي على الطريقـة      EventInfoاما الحديث عن الطرق، فالفئة      

 تمثل الاجراء الذي قام بقنص حدث الكـائن باسـتخدام   MethodInfoالتي تعود بكائن من النوع     
AddHandler()    بينما الطريقة ،GetRemoveMethod()   بالإجراء الذي أوقـف عمليـة     تعود
 فتعود بكـائن مـن      ()GetRaiseMethodاما الطريقة   . ()RemoveHandlerالقنص باستخدام   

اي الإجراء الذي سيتم تنفيذه     ( ايضا ولكنه يمثل الإجراء الذي يقنص الحدث         MethodInfoالنوع  
  ).لحظة وقوع الحدث

  
 Parametersالوسیطات 
 GetParameters تحتوي علـى الخاصـية       PropertyInfo و   MethodInfoالفئات من النوع    

.  يستقبلها الإجـراء   Parameter تمثل وسيطة    ParameterInfoوالتي تعود بمصفوفة من النوع      
 على مجموعة من الطرق والخصائص والخاصة بهذه الوسيطات،         ParameterInfoتحتوي الفئة   

 التـي تعـود بنـوع       ParameterType التي تعود باسم الوسيطة، والخاصية       Nameكالخاصية  
  .الوسيطة

 فهـذا يعنـي ان الوسـيطة اختياريـة          True بالقيمـة    IsOptionalان عادت الخاصية    
Optional               ويمكنك معرفة القيمة الافتراضية للوسـيطة الاختياريـة عـن طريـق الخاصـية ،

DefaultValue.  

التابعـة للفئـة     و ()MySubالشيفرة التالية توضح طريقة الاستعلام عن وسيطات الطريقة         
TestClass:  

 
Module MainModule 
 
    Class TestClass 
        Sub MySub(ByVal x As Integer, ByRef y As String) 
 
        End Sub 
    End Class 
 
    Sub main() 
        Dim T As Type = GetType(TestClass) 
        Dim P As ParameterInfo 
 
        For Each P In T.GetMethod("MySub").GetParameters() 
            ArabicConsole.WriteLine(P.Name & " As " & _  
                P.ParameterType.ToString) 
        Next 
 
    End Sub 
End Module 

  
  :مخرجات الشيفرة السابقة ستكون شيئا مثل

 
x As System.Int32 
y As System.String 

  
  التعامل مع الكائنات

، String ،Integer ،Double ،Personفي الأقسام السابقة، تعاملنا مع انـواع البيانـات كــ            
TestClass...            اما . الخ، وكان كل ما فعلناه هو الاستعلام عنها والحصول على معلومات حولها

 -مثلا– الكائنات، لنتمكن     ونتعامل مع  Reflectionفي هذا القسم سنحاول الاستفادة من الانعكاس        
  .أو استدعاء طريقة من اسناد قيمة لخاصية

  
  ReflectionExampleالفئة 

في هذا القسم سنجري مجموعة كبيرة من التجارب للقراءة والكتابة إلى حقول وخصائص الكائن،              
بالاضافة إلـى اسـتدعاء طرقـه وإرسـال وسـيطات لهـا، لـذلك فـضلت عـرض الفئـة                     

ReflectionExample  كما بالصفحة التالية–والتي سنجري عليها كافة التجارب:  

Class ReflectionExample 
    Public ExampleField As String 
 
    Private m_ExampleProperty As Integer 
    Public Property ExampleProperty() As Integer 
        Get 
            Return m_ExampleProperty 
        End Get 
        Set(ByVal Value As Integer) 
            m_ExampleProperty = Value 
        End Set 
    End Property 
 
    Public Sub ExampleMethod(ByVal x As Integer, ByVal y As Integer) 
        ArabicConsole.WriteLine(-x) 
        ArabicConsole.WriteLine(-y) 
    End Sub 
 
    Public Sub ExampleMethod2(Optional ByVal y As Integer = -1) 
        ArabicConsole.WriteLine(y) 
    End Sub 
End Class 

  
أحب ان انوه هنا بان الأمثلة في الفقرات التالية مباشرة لا تستخدم اي طرق ووسائل إضافية 

، وذلـك   )الـخ ...ن قابلية القراءة والكتابة، محددات الوصـول،        كالتحقق م (للتحقق من الأعضاء    
  !وعلي أنا ايضا–لاختصار وتسهيل الشيفرات البرمجية عليك 

  
   الحقولقراءة قیم/إسناد

 أو بـأي  ()GetField أو ()GetFieldsحاول الحصول على مرجع الحقل باسـتدعاء الطريقـة     
 والتي ()GetValue لتستدعي طريقته FieldInfoطريقة اخرى، واسند القيمة إلى كائن من النوع 

  :تعود بقيمة تمثل قيمة الحقل
 

Dim obj As New ReflectionExample() 
Dim F As FieldInfo = obj.GetType.GetField("ExampleField") 
 
obj.ExampleField = "***" 
 
 التحقق من ان الحقل تم قرائة بشكل صحیح '
ArabicConsole.WriteLine(F.GetValue(obj)) ' *** 

 لاسناد قيمة للحقل، تتطلب هذه الطريقة وسيطة        ()SetValueفي المقابل، استدعي الطريقة     
  :اضافية تمثل القيمة المراد اسنادها للحقل

 
Dim obj As New ReflectionExample() 
Dim F As FieldInfo = obj.GetType.GetField("ExampleField") 
 
F.SetValue(obj, "$$$") 
 
 التحقق من نجاح عملیة اسناد القیمة '
ArabicConsole.WriteLine(obj.ExampleField) 

  
   الخصائصقراءة قیم/إسناد

 ()GetProperty أو   ()GetPropertiesحاول الحصول على مرجع الخاصية باستدعاء الطريقة        
 ـ PropertyInfoأو باي طريقة اخرى، واسند القيمة إلى كائن مـن النـوع              ستدعي طريقتـه    لت

GetValue()            والتي تعود بقيمة تمثل قيمة الخاصية، من الضروري ارسـال القيمـة Nothing 
  : وسيطات نرسلها اليهاأيةكوسيطة ثانية لهذه الطريقة، وذلك لان هذه الخاصية لا تحتوي على 

 
Dim obj As New ReflectionExample() 
Dim P As PropertyInfo = obj.GetType.GetProperty("ExampleProperty") 
 
obj.ExampleProperty = 111 
  بشكل صحیحتها تم قرائالخاصیةالتحقق من ان  '
ArabicConsole.WriteLine(P.GetValue(obj, Nothing)) ' 111 
 

 سأريك مثالا لارسال وسيطات إلى الإجـراء، امـا الان           الطرقاستدعاء  في الفقرة التالية    
 ()SetValue ونحاول إسناد قيمة لها بالطريقة     ExampleProperty1ة  فلنكمل عملنا مع الخاصي   

  ): مع الوسيطة الثانيةNothingلا تنسى ارسال القيمة (
 

Dim obj As New ReflectionExample() 
Dim P As PropertyInfo = obj.GetType.GetProperty("ExampleProperty") 
 
P.SetValue(obj, 999, Nothing) 
من نجاح عملیة اسناد القیمةالتحقق  '  
ArabicConsole.WriteLine(obj.ExampleProperty) ' 999 

  
  استدعاء الطرق

، يمكنك في اي وقت MethodInfoبعد حصولك على مرجع للطريقة وإسنادها في كائن من النوع 
لى  والتي تتطلب وسيطات الطريقة، ان كانت الطريقة لا تحتوي ع          ()Invokeاستدعائها بالطريقة   

 عـدد  Object، اما إن وجدت فأرسل مصفوفة من النـوع    Nothingأية وسيطات فأرسل القيمة     
  :عناصرها يماثل عدد وسيطات الطريقة المستدعاة

 
Dim obj As New ReflectionExample() 
Dim M As MethodInfo = obj.GetType.GetMethod("ExampleMethod") 
Dim params() As Object = {-5, 5} 
 
M.Invoke(obj, params) 

  
  :مخرجات الشيفرة السابقة ستكون

 
5 
-5 

  
ــة  ــيطات اختياري ــة وس ــة Optional Parametersان كــان للطريق ، فأرســل القيم

Type.Missing           في المصفوفة المراد إرسالها كقيم لوسيطات الطريقة، في الشيفرة التالية قمـت 
 قيمة للوسـيطة الاختياريـة،      لل لا يرس  ، الأو ()ExampleMethod2بعمل استدعائين للطريقة    

  :10والثاني يرسل القيمة 
 

Dim obj As New ReflectionExample() 
Dim M As MethodInfo = obj.GetType.GetMethod("ExampleMethod2") 
Dim params() As Object = {Type.Missing} 
Dim params2() As Object = {10} 
 
M.Invoke(obj, params) 
M.Invoke(obj, params2) 

  
  :لتكون المخرجات

 
-1 
10 

  مواضیع أخرى
كما يفعل الربط (في هذا القسم اختم الفصل بعرض طريقة الإنشاء الديناميكي للكائنات لحظة التنفيذ 

  .، وطريقة معرفة الإجراءات المستدعيةLate Bindingالمتأخر 
  

 الإنشاء الدينامیكي للكائنات
 تمكنا من استدعاء طرق وخصائص الكائن، ولكن هذا الكائن منشأ من شيفراتنا في الأقسام السابقة،

  :المصدرية لحظة التصميم
 
Dim obj As New ReflectionExample() 
 

 هي امكانية انشاء الكائنات Reflectionمع ذلك، فالفائدة الحقيقية التي تجنيها من الانعكاس     
لذي تكون صيغته مبتدئـة باسـم المجمـع ثـم النـوع      لحظة التنفيذ بذكر الاسم النصي للكائن وا   

AssemblyName.TypeName .    يمكنك عمل ذلك، باسـتدعاء الطريقـةCreateInstance() 
  : تمثل النوعType والتي تتطلب وسيطة من النوع System.Activatorالتابعة للفئة 

 
Dim T As Type = Type.GetType("MyAssembly.MyClass") 
Dim obj As Object = Activator.CreateInstance(T) 

  
 فعليك إرسال وسيطات ذلك المشيد مع الطريقـة         Constructorان كان النوع يحتوي على مشيد       

CreateInstance() على شكل مصفوفة من النوع Object:  
 
Dim T As Type = Type.GetType("MyAssembly.MyClass") 
Dim params() As Object = {-5, 5} 
Dim obj As Object = Activator.CreateInstance(T, params) 

  
، System.Activatorطريقة اخرى تمكنك من إنشاء الكائن ديناميكيا دون استخدام الفئـة            

ولكنها طويلة بعض الشيء تتطلب التوغل في تفاصيل مشيد الفئة لإرساله إلى كائن مـن النـوع                 
ConstructorInfo ) وهو مشتق من الفئةMemberInfo:(  

 
Dim T As Type = Type.GetType("MyAssembly.MyClass") 
 
Dim types() As Type = { GetType(Integer), GetType(Integer) } 
 
Dim con As ConstructorInfo = T.GetConstructor (types) 
 
Dim params() As Object = {-5, 5} 
 
Dim obj As Object = con.Invoke(params) '  الكائنإنشاء  

 معرفة الإجراءات المستدعیة
 التابعـة لكـائن الاسـتثناء    StackTrace علمنا ان الخاصية     اكتشاف الأخطاء في الفصل السابع    

Exception             تعود بقيمة حرفية تمثل طابور الإجراءات المتعلقة بالاستثناء، كيف تمكنك هذه الفئة 
 System.Diagnostics الأسماء   من معرفة ذلك؟ والجواب بفضل مجموعة من الفئات في مجال         

  . ايضاReflectionتصنف من فئات الانعكاس –
 الإجراءات المعرفة في طابور الاستدعاءات، تشمل الإجراءات في StackTraceتمثل الفئة 

هذا الطابور الإجراءات المستدعية في قائمة الانتظار ليعود التنفيذ لها والمستدعاة التـي يجـري               
، وعليك استدعاء الطريقة New باستخدام StackTraceك إنشاء كائن من الفئة      تنفيذها حاليا، يمكن  

GetFrame()        التي تعود بكائن من النـوع StackFrame      يمثـل القـسم المـستدعي، الكـائن 
StackFrame       بدوره يتطلب منك استدعاء طريقته GetMethod()     الذي يعود بكائن من النـوع 
MethodBase)       يتبع إلـى مجـال الاسـماء System.Collection       واسـتخدامه شـبيه بــ 
MethodInfo( الشيفرة التالية تعود بكافة الإجراءات في طابور الاستدعاءات ،Stack Trace:  

 
Dim ST As New StackTrace() 
Dim counter As Integer 
 
For counter = 0 To ST.FrameCount - 1 
    Dim SF As StackFrame = ST.GetFrame(counter) 
    Console.WriteLine(SF.GetMethod.Name & " ()") 
Next 

  
ارفق لك هذا المثال التطبيقي الذي يظهـر لـك طـابور الاسـتدعاءات مـن الاجـرائين         

CallerSub() و CalleeSub():  
 

Imports System.Reflection 
Imports System.Diagnostics 
 
Module Module1 
    Sub main() 
        CallerSub() 
    End Sub 
 
    Sub CallerSub() 
        Dim ST As New StackTrace() 
        Dim counter As Integer 
 
        ArabicConsole.WriteLine("** CallerSub *********") 

        For counter = 0 To ST.FrameCount - 1 
            Dim SF As StackFrame = ST.GetFrame(counter) 
            ArabicConsole.WriteLine(SF.GetMethod.Name & " ()") 
        Next 
 
        CalleeSub() 
    End Sub 
 
    Sub CalleeSub() 
        Dim ST As New StackTrace() 
        Dim counter As Integer 
 
        ArabicConsole.WriteLine("** CalleeSub *********") 
        For counter = 0 To ST.FrameCount - 1 
            Dim SF As StackFrame = ST.GetFrame(counter) 
            ArabicConsole.WriteLine(SF.GetMethod.Name & " ()") 
        Next 
    End Sub 
End Module 

  
 وذلك لأنه اول إجراء يدخل طـابور  ()Mainي مخرجات الشيفرة السابقة ستشمل الإجراء الرئيس     

  :الاستدعاءات
 
** CallerSub ********* 
CallerSub () 
main () 
 
** CalleeSub ********* 
CalleeSub () 
CallerSub () 
main () 

  :أسلوب أفضل
  من كتابة شيفرات الحلقة التكرارية في كل مرة تود معرفة طابور الاستدعاءات، لما لا تعرف    بدلا

 تمثل الإجراءات   String يعود بمصفوفة حرفية من النوع       ()GetStackإجراء خاص بك بالإسم     
  :في طابور الاستدعاءات

Function GetStack() As String() 
    Dim counter As Integer 
    Dim stackArray() As String 
    Dim ST As New StackTrace() 
 
    For counter = 1 To ST.FrameCount - 1 
        Dim SF As StackFrame = ST.GetFrame(counter) 
        ReDim Preserve stackArray(counter) 
        stackArray(counter - 1) = SF.GetMethod.Name 
    Next 
 
    Return stackArray 
End Function 
 
 

    ملاحظة
  

 السابق تبدأ بواحد، وذلك لاني ()GetStackلاحظ ان الحلقة في الإجراء 
  . نفسه في المصفوفة()GetStackاريد تجاھل العودة باسم الإجراء 

 
 

  
  
  
  
  
  
  

 هو أسلوب يمكنك من الاستفادة من أنواع البيانات لحظة التنفيـذ ولـيس              Reflectionالانعكاس  
يعيب الانعكاس انه   . ياالتصميم، بحيث تتمكن من إنشاء كائنات بمجرد الحصول على أسمائها نص          

يتوجب عليك التحقق من كافة التفاصيل الدقيقة المتعلقة بالبيانات وأعضائها قبل استخدامها، ولكنه             
 NET.يمكنك من إنجاز مهام لا تخطر على بال احد، ولا تنسى ان البنية التحتية لإطـار عمـل   

Framework ائنـات  لعـل أبرزهـا عمليـات تسلـسل الك    ( تستخدم فئات الانعكاسObject 

Serialization .( لنودع الآن المشاريع من النوعConsole Application وننتقل إلى مرحلة ،
  . عنوان الجزء الثالث من هذا الكتابWindowsتطوير تطبيقات  وهي -طالما انتظرناها–جديدة 

